



M.Th.A.M. Vijftigschild

BASIC computerspellen



Kluwer
Software-reeks

BASIC-computerspellen

M.Th.A.M. Vijftigschild

BASIC-computerspellen



Kluwer Technische Boeken B.V. Deventer-Antwerpen

ISBN 90 201 1601 0
D/1982/0108/272

© 1982/1983 Kluwer Technische Boeken B.V. Deventer

1e druk 1982
2e oplage 1983
3e oplage 1983

Niets uit deze uitgave mag worden verveelvoudigd en/of openbaar gemaakt door middel van druk, fotokopie, microfilm of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van de uitgever.

No part of this book may be reproduced in any form, by print, photoprint, microfilm or any other means without written permission from the publisher.

Ondanks alle aan de samenstelling van de tekst bestede zorg, kan noch de redactie noch de uitgever aansprakelijkheid aanvaarden voor eventuele schade, die zou kunnen voortvloeien uit enige fout, die in deze uitgave zou kunnen voorkomen.

Woord vooraf

Dit is het eerste Nederlandstalige computerspellenboek. Alle spellen die in dit boek staan, zijn er speciaal voor ontwikkeld. Bij de ontwikkeling van de programma's is ervan uitgegaan, dat de programma's op elke computer met een BASIC-interpreter of -compiler moeten kunnen draaien. Hier en daar zullen misschien kleine aanpassingen noodzakelijk zijn. Als men het boek doorbladert, zal het opvallen dat de meeste programma's volgens een zeer strakke modulaire structuur zijn opgebouwd. Voor deze opbouw is heel bewust gekozen. De bedoeling is niet alleen de gebruiker een aantal kant-en-klare spelprogramma's ter beschikking te stellen, maar hem tevens in staat te stellen op eenvoudige wijze wijzigen en aanpassingen aan te brengen. Om de gebruiker op weg te helpen, zijn bij een aantal programmabeschrijvingen suggesties toegevoegd. Deze suggesties hebben tot doel de gebruiker een stukje op weg te helpen om eventuele wijzigingen aan te brengen. Bij de opzet van dit boek is ervan uitgegaan dat een beeldscherm als uitvoereenheid wordt gebruikt en het toetsenbord als invoereenheid. Bewust is geen gebruik gemaakt van achtergrondgeheugen, omdat de communicatie tussen de computer en het achtergrondgeheugen niet uniform is geregeld.

Voor wat het beeldscherm betreft, is ervan uitgegaan dat er 64 karakters op een regel gaan en dat er minstens 14 regels op het scherm kunnen worden gezet. Bij de programmatuur is afgezien van het gebruik van PEEK- en POKE-opdrachten. De plaats, waar het scherm gedefinieerd staat bij de verschillende merken computers, is overal anders.

Bij de meeste programma's zit een run-voorbeeld, zodat men enig idee heeft wat het programma gaat doen.

Sommige programma's worden ingeleid met een uitgebreid verhaal om de spelsituatie te verduidelijken.

In de programmatuur zal men regelmatig twee specifieke TRS-80 statements tegenkomen. Het eerste is het CLS-statement. Het CLS-statement zorgt ervoor dat het scherm in één keer wordt gewist. Bij de Commodore-computers is dit bijvoorbeeld CLEAR/HOME. Het tweede is het RANDOM-statement. Dit zorgt ervoor dat de random-generator op een willekeurige waarde start. In andere computers komt dit statement voor met benamingen als RANDOMIZE of RAND. Heeft men een computer waarbij dit statement niet voorkomt dan kan men het zonder bezwaar weglaten.

Een ander probleem vormt de functie RND(0). Het blijkt dat deze functie door de verschillende merken microcomputers anders wordt geïnterpreteerd. In dit boek is de functie RND(0) een getal met een waarde die ligt tussen 0 en 1. Achterin het boek is een vrij uitgebreide appendix opgenomen, waar de BASICS van de verschillende computers naast elkaar staan. Men kan dan opzoeken wat een bepaald statement doet.

In het eerste hoofdstuk wordt uitgelegd hoe de programma's die in dit boek staan, zijn ontwikkeld. Meestal wordt dit soort programma's recht toe recht aan geschreven, dat is in dit geval niet gebeurd. De gebruikte methode is ook toepasbaar voor het ontwikkelen van andere applicatie-programmatuur.

Inhoud

1. Inleiding	9
2. Standaardspellen	11
2.1 Mijn computer	11
2.2 31 Fiches	13
2.3 Dobbelsteen	17
2.4 Sul en Xantippe	21
2.5 Eenentwintigen	26
2.6 Hoog-Laag	32
2.7 Gokken	36
2.8 Letters	40
2.9 Galgje	44
2.10 Mastermind	50
2.11 Vier op een rij	55
3. Science fiction spellen	61
3.1 Maanlander	61
3.2 Ruimtevaartuig	66
3.3 De Blijde	74
3.4 De Compusaurus	80
3.5 De Zandplaneet Silico	88
3.6 Rigel Zuiderkwartier	95
3.7 De slag bij de Shrixheuvels	102
4. Simulatie spellen	111
4.1 De kolenmijn	111
4.2 Boerderij Zeldenrust	119
4.3 De stadhouder van Flipflopland	126
Appendix	135

1. Inleiding

Wanneer men het boek doorbladert, zal het opvallen dat de meeste programma's volgens eenzelfde structuur zijn opgebouwd. Dit is niet zomaar gedaan, maar zeer bewust. De gevolgde methode heeft als groot voordeel dat elk programma in logische eenheden is op te delen. De verschillende eenheden worden via een stuurmodule met elkaar verbonden. In de programmatuur zitten nog meer standaards ingebouwd, maar die zullen alleen bij nauwkeurige bestudering opvallen.

Daar dit het eerste Nederlandstalige spellenboek in BASIC is, zal wat dieper worden ingegaan op het ontwikkelen van een spel tot een computerprogramma.

Zoals al werd opgemerkt, zijn de meeste programma's in dit boek via een gestructureerd modulaire manier opgebouwd. Dit heeft tot gevolg dat de meeste programma's ook uiterlijk sterk op elkaar lijken. Dat laatste is ook de bedoeling. Als men de opbouw van één programma door heeft, kent men de opbouw van alle andere programma's. De programma's beginnen allemaal met één of meer NON-EXECUTABLE-statements. Dit zijn statements die voor de uitvoering van een BASIC-programma noodzakelijk zijn, maar waarvan de gebruiker niet merkt dat ze ook inderdaad worden uitgevoerd. Het zijn dus statements zoals, DATA, DIM en RANDOM. Direct na deze statements volgt een GOTO naar een hoog regelnummer. Dit is de enige GOTO die in het programma voorkomt. De sprongopdracht zorgt ervoor dat de stuurmodule wordt gestart, deze houdt het gehele programma onder controle. Via subroutines worden de verschillende programmamodules aangeroepen. Het voordeel van deze methode is dat de noodzakelijke programmamodules door elkaar kunnen worden ontwikkeld.

Elk spelprogramma bestaat uit een invoer-, bewerkings- en uitvoermodule. Deze modules zijn ook terug te vinden in de programma's, vaak met deze aanduidingen. Behalve de reeds genoemde modules komen ook de instructiemodule en de initialisatiemodule voor. De volgorde waarin deze modules in het programma komen te staan is niet van belang, omdat de volgorde in de stuurmodule wordt bepaald.

Een idee voor een spel wordt meestal eerst op papier uitgewerkt. Meestal kan men dan de spelregels al bepalen. Is men zo ver, dan kan de instructiemodule met de spelregels worden gemaakt. Is de module ingetikt dan kan deze worden getest.

Zo, het eerste stukje programma is klaar. Waarschijnlijk heeft u al begrepen dat elke module als een programmaatje wordt beschouwd. Zo'n programma-module heeft een logisch begin en een logisch eind. Bij de ontwikkeling van deze programmatuur is ervan uitgegaan dat de module wordt aangeroepen met het regelnummer waarop het eerste statement van de module staat. De module wordt bij het hoogste regelnummer van de module verlaten; dit is bij de RETURN-opdracht. Op deze manier bereikt men dat er slechts één begin en één einde in de module is. Men voorkomt op deze manier de zogenaamde 'spaghetti-structuur' die helaas maar al te vaak in BASIC-programma's wordt aangetroffen. De bedoeling van al deze modules is te zorgen dat fouten in de programmatuur zeer snel kunnen worden gelokaliseerd en ook weer snel zijn opgelost. Het zal duidelijk zijn dat in het algemeen elke module maar één functie heeft.

De instructiemodule is al aan de orde geweest. Handig is het om daarna met de uitvoermodule te beginnen. De reden is dat men op die manier automatisch de variabelen kan bepalen die men in het programma nodig heeft. Tevens kan men dan bepalen welke variabele via welke module komt.

Nadat de uitvoermodule is ingebracht, kan hij worden getest. Zeker wanneer er hele plaatjes moeten worden opgebouwd, zal dit vrij veel testwerk geven om alles netjes op zijn plaats te krijgen. Werkt de module dan hoeft men er verder niet meer naar om te kijken.

De module die logisch volgt op de uitvoermodule is de invoermodule. In deze module worden de gegevens opgehaald die van het toetsenbord komen. Zou men een spel ontwerpen met joysticks, dan haalt men de gegevens op van de joysticks. In deze module worden niet alleen de gegevens opgehaald, maar wordt tevens nagegaan of de waarden aan de gestelde eisen voldoen. Afhankelijk van het soort programma kan men als het fout gaat direct om nieuwe invoer vragen of men kan opnieuw een plaatje opbouwen en dan de invoermodule weer aanroepen. Dit hangt af van hoe men de stuurmodule wil structureren. Een aantal variabelen dient in een spelprogramma een vaste startwaarde te krijgen, dit wordt gedaan in de initialisatiemodule. Deze module wordt slechts eenmaal aangeroepen aan het begin van het programma.

Er blijft nog één module over om te maken: de bewerkingsmodule. Deze module vormt het hart van elk programma. Hierin worden de verschillende berekeningen uitgevoerd en de verschillende strings opgebouwd die in de uitvoermodule naar het beeldscherm worden gestuurd.

Als de bewerkingsmodule klaar is, kan er worden getest. Treden er fouten op dan zit de fout meestal in de berekeningsmodule.

Deze methode heeft behalve het voordeel van de overzichtelijkheid ook het voordeel dat later vrij eenvoudig kleinere berekeningsmodules kunnen worden bijgevoegd. Deze kunnen dan óf via de stuurmodule óf via de grote berekeningsmodule worden aangeroepen. Wanneer men later modules toevoegt, moet men zich wel aan de gestelde eisen houden.

Nu alle modules klaar zijn, moet de zaak nog aan elkaar worden geknoopt. Dit gebeurt via de stuurmodule. Om een inzicht te krijgen hoe zoiets er uitziet, is het programma 'De Zandplaneet Silico' een goed voorbeeld.

Het veelvuldig gebruik van REM-statements bevordert, zeker bij het ontwikkelen, de leesbaarheid van een programma. De programmatuur in dit boek is er dan ook rijkelijk mee voorzien. De gebruiker kan aan de hand van de REM-statements al direct een goed inzicht krijgen in de werking van een programma en wat de verschillende variabelen voorstellen. De REM-statements achter een BASIC-statement kunnen het beste worden weggelaten als men het programma overtikt. Eventueel kunnen alle REM-statements worden weggelaten. Laat men de regels die met een REM-statement beginnen weg, dan wordt het wel erg ingewikkeld om de afbakening tussen de verschillende modules te vinden.

2. Standaardspellen

2.1 Mijn computer

Het programma 'Mijn computer' is een vrij eenvoudig demonstratieprogramma. Er wordt om de voornaam van de speler gevraagd. Aan de hand van de ingevoerde naam wordt een golvend display opgebouwd.

Na de golvende display wordt om de geboortedatum gevraagd. Deze geboortedatum wordt met de formule van 'Zeller' omgezet in de geboortedag.

Programma

```
10 REM *** MIJN COMPUTER ***
20 RANDOM
30 DATA ZONDAG, MAANDAG, DINSDAG, WOENSDAG, DONDERDAG
40 DATA VRIJDAG, ZATERDAG
50 CLS
60 PRINT "GEEF EENS EEN GETAL ";
70 INPUT A
80 FOR I = 1 TO A
90 A = RND(0)
100 NEXT I
110 PRINT
120 PRINT "IK BEN DE NIEUWE COMPUTER VAN ....."
130 PRINT "VOLGENS ..... BEN IK ZEER KNAP."
140 PRINT "AF EN TOE LAAT IK WEL EENS EEN STEEK VALLEN."
150 PRINT
160 PRINT "HOE HEET JIJ ";
170 INPUT A$
180 FOR J = 1 TO 2
190 FOR I = 1 TO 360 STEP 4
200 X = ABS(INT(COS(I * 22/7/180)*40))+1
210 Y = ABS(INT(SIN(I * 22/7/180)*40))+1
220 IF X < Y THEN 260
230 Z = X
240 X = Y
250 Y = Z
260 PRINT TAB(X)" HALLO "; A$;TAB(Y)" HALLO ";A$
270 NEXT I
280 NEXT J
290 PRINT
300 B$ = "ABCDEFGHIJKLMNOPQRSTUVWXYZ"
310 X = INT(RND(0) * 26 + 1)
320 B$ = LEFT$(B$,X)
330 B$ = RIGHT$(B$,1)
340 C$ = RIGHT$(A$,LEN(A$)-1)
350 C$ = B$ + C$
360 PRINT "HOE WAS JE NAAM OOK AL WEER ";C$;
370 INPUT B$
380 IF A$ = B$ THEN 410
```

```

390 PRINT "JE MOET NIET JOKKEN ";A$;"!!!!"
400 PRINT "IK HEB JE WEL DOOR."
410 PRINT "IK MAAK OOK WELEENS FOUTEN, ZOALS JE ZIET !!"
420 PRINT "WANNEER BEN JE GEBOREN ";B$;" <D,M,J>";
430 INPUT D,M,J
440 H = 0
450 IF (D < 0) OR (D > 31) THEN 480
460 IF (M < 1) OR (M > 12) THEN 480
470 IF (J > 1900) AND (J < 2000) THEN 500
480 H = 1
490 PRINT "DE DATUM, DIE JE OPGEEFT KLOPT NIET ! EEND"
500 IF H = 1 THEN 420
510 REM ** BEPALING DAG VAN DE WEEK **
520 IF M > 2 THEN 550
530 J = J - 1
540 M = M + 12
550 E = INT(J/100)
560 Y = J - E * 100
570 A = INT(2.6001*(M-2)-0.2)+D+Y+INT(Y/4)+INT(E/4)-2*E
580 A = A - INT(A/7)*7 +1
590 REM ** BIJZOEKEN VAN DE NAAM VAN DE DAG **
600 RESTORE
610 FOR I = 1 TO A
620 READ B$
630 NEXT I
640 PRINT "VOLGENS MIJ BEN JE GEBOREN OP EEN ";B$;" ."
650 PRINT "KLOPT HE ";
660 INPUT A$
670 IF LEFT$(A$,1) <> "N" THEN 700
680 PRINT "VRAAG HET DAN NOG MAAR EENS AAN JE MOEDER !"
690 PRINT
700 PRINT "NOG VEEL PLEZIER MET DE REST VAN DE PROGRAMMA'S."
710 END

```

2.2 31 Fiches

Het programma 31 fiches is een van de kleinere programma's. Men tikt het binnen een korte tijd in. Het is een leuk spel, zeker wanneer men niet doorheeft hoe het werkt. De werking van het programma is overigens vrij simpel.

De speler begint met fiches van de stapel te pakken, minimaal 1 en maximaal 4. De computer doet hetzelfde. Degene die het laatste fiche moet pakken heeft verloren.

Het spel is zodanig in elkaar gezet dat de speler altijd verliest. Wanneer men het programma volgt, zal men er snel achter komen waarom.

Suggestie 1: Veel mensen zullen het niet leuk vinden dat de computer iedere keer wint. Daar is natuurlijk best wel wat aan te doen. Men zou bijvoorbeeld de speler óf de computer als eerste kunnen laten beginnen. De keuze kan men aan de speler overlaten of men kan via een willekeurig (random) getrokken getal bepalen wie er begint.

Suggestie 2: Degenen die een computer hebben met meer grafische mogelijkheden, kunnen echte stapeltjes maken in plaats van de vakjes met sterretjes.



Programma

```
10 REM *** 31 FICHES ***
20 RANDOM
30 GOTO 800
40 :
50 REM *** SPELREGELS ***
60 CLS
70 PRINT "*** 31 FICHES ***";
80 PRINT TAB(40) "- SPELREGELS -"
90 PRINT
100 PRINT "DE SPELER EN DE COMPUTER MOGEN OM DE BEURT FICHES"
110 PRINT "VAN DE STAPEL AFNEMEN. MAXIMAAL ZIJN DIT VIER"
120 PRINT "FICHES EN MINIMAAL IS DAT EEN FICHE. "
130 PRINT "DEGENE DIE HET LAATSTE FICHE MOET PAKKEN HEEFT"
140 PRINT "VERLOREN."
150 PRINT
160 PRINT "DE SPELER MAG BEGINNEN !"
170 PRINT
180 PRINT "HEEFT MEN DIT GELEZEN , DRUK DAN DE <RETURN-TOETS> I
190 INPUT A$
200 RETURN
210 :
220 REM *** INITIALIZEREN ***
230 A$ = "!----!----!----!----!----!"
240 C$ = ""
250 S = 31 :REM AANTAL FICHES
260 H = 3 :REM TEKST OVERSLAAN
270 GOSUB 320
280 H = 0 :REM SPELER BEGINT
290 RETURN
300 :
310 REM *** UITVOER ***
320 CLS :REM SCHERM WISSEN
330 PRINT "*** 31 FICHES ***";
340 PRINT
350 PRINT TAB(10) A$
360 FOR I = 6 TO 0 STEP -1
370 PRINT TAB(10) "!";
380 FOR J = 1 TO 5
390 Y = I * 5 + J
400 B$ = " * "
410 IF Y <= S THEN 430
420 B$ = " "
430 PRINT B$;"!";
440 NEXT J
450 PRINT
460 NEXT I
470 PRINT TAB(10) A$
480 IF H = 3 THEN 560
490 PRINT D$; "NEEMT ";K;"FICHE(S). "
500 IF H = 2 THEN 560
510 PRINT "ER ZIJN NOG ";S; "FICHES OVER."
520 IF H = 0 THEN 560
530 REM ** WACHTLUS **
```



```

540 FOR I = 1 TO 700
550 NEXT I
560 RETURN
570 :
580 REM *** ZET SPELER ***
590 PRINT "HOEVEEL FICHES GAAN ERAF 1-2-3-4 ";
600 INPUT K
610 IF (K < 1) OR (K > 4) THEN 590
620 K = INT(K)
630 S = S - K
640 H = 1
650 D$ = "DE SPELER "
660 RETURN
670 :
680 REM *** ZET COMPUTER ***
690 K = 5 - K
700 S = S - K
710 H = 0
720 D$ = "DE COMPUTER "
730 IF S <> 1 THEN 760
740 H = 2
750 C$ = "IK HEB GEWONNEN; ER IS NOG EEN FICHE OVER !!"
760 RETURN
770 :
780 REM ***** STUURMODULE *****
790 REM *** SPELREGELS ***
800 GOSUB 60
810 REM *** INITIALIZEREN ***
820 GOSUB 230
830 REM *** ZET SPELER ***
840 GOSUB 590
850 REM *** SPEELBORD ***
860 GOSUB 320
870 REM *** ZET COMPUTER ***
880 GOSUB 690
890 REM *** SPEELBORD ***
900 GOSUB 320
910 IF H <> 2 THEN 840
920 PRINT C$
930 END

```

Voorbeeld

*** 31 FICHES ***

```

  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *
  * | * | * | * | *

```

HOEVEEL FICHES GAAN ERAF 1-2-3-4 4

*** 31 FICHES ***

*	*			
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*

DE SPELER NEEMT 4 FICHE(S).
ER ZIJN NOG 27 FICHES OVER.

*** 31 FICHES ***

*				
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*

DE COMPUTER NEEMT 1 FICHE(S).
ER ZIJN NOG 26 FICHES OVER.
HOEVEEL FICHES GAAN ERAF 1-2-3-4 2

*** 31 FICHES ***

*	*	*	*	
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*
*	*	*	*	*

DE SPELER NEEMT 2 FICHE(S).
ER ZIJN NOG 24 FICHES OVER.

2.3 Dobbelsteen

Dit programma is vrij eenvoudig. Het is als het ware een basis voor de ontwikkeling van andere bordspellen in de computer. Een voorbeeld is het programma 'Sul en Xantippe'. Het programma laat zien hoe met één tot maximaal vier dobbelstenen kan worden gegooid. Tevens geeft dit programma er een aardig beeld van hoe men binnen de primitieve mogelijkheden van een niet-grafisch beeldscherm toch nog een aanvaardbaar beeld kan opbouwen.

Suggestie: Men zou het rollen van de dobbelstenen kunnen nabootsen wanneer men over PEEK- en POKE-instructies beschikt. Tijdens het rollen worden de ogen in het kader gE-POKEd. Men moet overigens niet vergeten de ogen van de vorige worp te wissen. Is de dobbelsteen 'uitgerold' dan geeft men het totaal aantal ogen.

Programma

```
10 REM *** DOBBELSTEEN ***
20 RANDOM
30 DIM A$(3), A(9,4)
40 GOTO 1030
50 :
60 REM *** INITIALIZEREN ***
70 H = 0
80 C$ = " !-----! "
90 RETURN
100 :
110 REM *** STRINGS WISSEN ***
120 FOR I=1 TO 3
130 A$(I) = ""
140 NEXT I
150 A$ = ""
160 S = 0 :REM SOM AANTAL OGEN
170 RETURN
180 :
190 REM *** INVOER ***
200 IF H > 0 THEN 290
210 H = 1
220 CLS
230 PRINT "*** DOBBELSTEEN ***"
240 PRINT
250 PRINT "MET HOEVEEL DOBBELSTENEN WIL JE GOOIEN <MAX.4>";
260 INPUT X
270 IF ( X < 0 ) OR ( X > 4 ) THEN 220
280 PRINT "HOEVEEL KEER MOETEN DE DOBBELSTENEN ROLLEN ";
290 PRINT " <STOP = 0>";
300 INPUT K
310 RETURN
320 :
330 REM *** ROLLEN VAN DOBBELSTEEN ***
340 FOR I = 1 TO ABS(K) :REM BEVEILIGING
350 Y = RND(0) :REM ROLLEN
360 NEXT I
```

```

370 RETURN
380 :
390 REM *** UITVOER OPBOUWEN ***
400 FOR Y = 1 TO X           :REM LUS = AANTAL DOBBELST.
410 A$ = A$ + C$
420 FOR J = 1 TO 3
430 A$(J) = A$(J) + " !"
440 FOR I = ( J - 1 ) * 3 + 1 TO ( J - 1 ) * 3 + 3
450 B$ = " "
460 IF A(I,Y) <> 1 THEN 480
470 B$ = " 0"
480 A$(J) = A$(J) + B$
490 NEXT I
500 A$(J) = A$(J) + " ! "
510 NEXT J
520 NEXT Y
530 RETURN
540 :
550 REM *** DOBBELSTENEN OPBOUWEN ***
560 FOR J = 1 TO X
570 FOR I = 1 TO 9
580 A(I,J) = 1
590 NEXT I
600 REM ** POSITIES WORDEN NOOIT GEBRUIKT **
610 A(4,J) = 0
620 A(6,J) = 0
630 REM ** DOBBELSTEEN ROLLEN **
640 G = INT(RND(0)*6 + 1)
650 S = S + G           :REM SOMMEREN AANTAL WORPEN
660 REM ** DOBBELSTEEN MAKEN **
670 IF G <> 6 THEN 690
680 A(5,J) = 0
690 IF G > 5 THEN 720
700 A(2,J) = 0
710 A(8,J) = 0
720 IF G <> 4 THEN 740
730 A(5,J) = 0
740 IF G > 3 THEN 820
750 A(3,J) = 0
760 A(7,J) = 0
770 IF G <> 2 THEN 790
780 A(5,J) = 0
790 IF G <> 1 THEN 820
800 A(1,J) = 0
810 A(9,J) = 0
820 NEXT J
830 RETURN
840 :
850 REM *** UITVOER ***
860 CLS
870 PRINT "*** DOBBELSTEEN ***"
880 PRINT
890 PRINT A$
900 FOR I = 1 TO 3
910 PRINT A$(I)

```

```

920 NEXT I
930 PRINT A$
940 PRINT
950 PRINT "TOTAAL GEGOOID ";S;" OGEN"
960 PRINT
970 IF K <> 0 THEN 990
980 PRINT "LAATSTE WORP !!"
990 RETURN
1000 :
1010 REM **** STUURMODULE ****
1020 REM *** INITIALIZEREN ***
1030 GOSUB 70
1040 REM *** STRINGS WISSEN ***
1050 GOSUB 120
1060 REM *** INVOER ***
1070 GOSUB 200
1080 REM *** ROLLEN DOBBELSTEEN ***
1090 GOSUB 340
1100 REM *** DOBBELSTENEN OPBOUWEN ***
1110 GOSUB 560
1120 REM *** UITVOER OPBOUWEN ***
1130 GOSUB 400
1140 REM *** UITVOER ***
1150 GOSUB 860
1160 IF K <> 0 THEN 1050
1170 END

```

Voorbeeld

*** DOBBELSTEEN ***

MET HOEVEEL DOBBELSTENEN WIL JE GOOIEN <MAX.4> 3

HOEVEEL KEER MOETEN DE DOBBELSTENEN ROLLEN <STOP = 0> 2

*** DOBBELSTEEN ***

```

|-----| |-----| |-----|
|         | |   0   | |   0 0  |
|    0    | |   0   | |   0 0  |
|         | |   0   | |   0 0  |
|-----| |-----| |-----|

```

TOTAAL GEGOOID 8 OGEN

HOEVEEL KEER MOETEN DE DOBBELSTENEN ROLLEN <STOP = 0> 0

*** DOBBELSTEEN ***

-----	-----	-----
0 0	0 0 0	0 0
0		0
0 0	0 0 0	0 0
-----	-----	-----

TOTAAL GEGOOID 16 OGEN

LAATSTE WORP !!

2.4 Sul en Xantippe

Sul en Xantippe is een oud dobbelspelletje. Het is de bedoeling om in een zo klein mogelijk aantal worpen een Sul en een Xantippe te gooien. Een Sul is een doublet; de twee dobbelstenen hebben een gelijk aantal ogen. Een Xantippe is een worp waarbij het totaal aantal ogen zeven is.

Het programma is gebaseerd op de randomgenerator, die in praktisch alle microcomputers aanwezig is. Wanneer men het programma bekijkt, zal het opvallen dat modules uit het programma Dobbelsteen zijn gebruikt. Het blijkt dat wanneer men de modules goed ontwerpt, ze ook in andere programma's zijn te gebruiken. In dit boek zijn meer van dit soort voorbeelden te vinden.

Suggestie: Het programma gaat ervan uit dat men tegen de computer speelt. Het programma is vrij eenvoudig aan te passen, zodat het met meer personen te spelen is.



Programma

```
10 REM *** SUL EN XANTIPPE ***
20 RANDOM
30 DIM A$(3), A(9,2),B(2),C(2,2),D(2)
40 GOTO 1530
50 :
60 REM *** INITIALIZEREN ***
70 H = 1 :REM STUURVARIABLE SPELER/COMP
80 FOR I = 1 TO 2
90 B(I) = 0 :REM RESULTAAT EEN WORP
100 C(I,1) = 0 :REM SUL SCHOON
110 C(I,2) = 0 :REM XANTIPPE SCHOON
120 D(I) = 0 :REM AANTAL WORPEN
130 NEXT I
140 X = 2 :REM AANTAL DOBBELSTENEN
150 C$ = " !-----! "
160 RETURN
170 :
180 REM *** SPELREGELS ***
190 CLS
200 PRINT "*** SUL EN XANTIPPE ***"
210 PRINT
220 PRINT "OM DE BEURT GOOIT MEN MET TWEE DOBBELSTENEN."
230 PRINT "DE SPELER BEGINT EN DAARNA GOOIT DE COMPUTER."
240 PRINT "WIE HET EERST EEN SUL EN EEN XANTIPPE GOOIT, "
250 PRINT "OF OMGEKEERD HEEFT GEWONNEN."
260 PRINT
270 PRINT "EEN SUL IS EEN DOUBLET D.W.Z. 2 GELIJKE "
280 PRINT "AANTALLEN OGEN."
290 PRINT "EEN XANTIPPE IS EEN WORP, WAARBIJ HET TOTAAL"
300 PRINT "AANTAL OGEN GELIJK AAN 7 IS."
310 PRINT
320 PRINT "HEEFT MEN DIT GELEZEN DRUK DAN DE <RETURN> IN";
330 INPUT A$
340 RETURN
350 :
360 REM *** STRINGS WISSEN ***
370 FOR I=1 TO 3
380 A$(I) = ""
390 NEXT I
400 A$ = ""
410 E$ = ""
420 F$ = ""
430 D(Q) = D(Q) + 1 :REM WORPEN-TELLER
440 S = 0 :REM SOM AANTAL OGEN
450 RETURN
460 :
470 REM *** INVOER-SPELER ***
480 PRINT "HOEVEEL KEER MOETEN DE DOBBELSTENEN ROLLEN ";
490 INPUT K
500 RETURN
510 :
520 REM *** ROLLEN VAN DOBBELSTEEN ***
530 FOR I = 1 TO ABS(K) :REM BEVEILIGING
```

```

540 Y = RND(0)                                :REM ROLLEN
550 NEXT I
560 RETURN
570 :
580 REM *** UITVOER OPBOUWEN ***
590 REM ** MATRIX A$ OPBOUWEN **
600 FOR Y = 1 TO X                            :REM LUS = AANTAL DOBBELST.
610 A$ = A$ + C$
620 FOR J = 1 TO 3
630 A$(J) = A$(J) + " !"
640 FOR I = ( J - 1 ) * 3 + 1 TO ( J - 1 ) * 3 + 3
650 B$ = " "
660 IF A(I,Y) <> 1 THEN 680
670 B$ = " 0"
680 A$(J) = A$(J) + B$
690 NEXT I
700 A$(J) = A$(J) + " ! "
710 NEXT J
720 NEXT Y
730 A$(2) = A$(2) + E$
740 RETURN
750 :
760 REM *** DOBBELSTENEN OPBOUWEN ***
770 FOR I = 1 TO 9
780 A(I,J) = 1
790 NEXT I
800 REM ** POSITIES WORDEN NOOIT GEBRUIKT **
810 A(4,J) = 0
820 A(6,J) = 0
830 REM ** DOBBELSTEEN ROLLEN **
840 G = INT(RND(0)*6 + 1)
850 S = S + G                                :REM SOMMEREN AANTAL WOPREN
860 B(J) = G                                :REM AANTAL OGEN ONTHOUDEN
870 REM ** DOBBELSTEEN MAKEN **
880 IF G <> 6 THEN 900
890 A(5,J) = 0
900 IF G > 5 THEN 930
910 A(2,J) = 0
920 A(8,J) = 0
930 IF G <> 4 THEN 950
940 A(5,J) = 0
950 IF G > 3 THEN 1030
960 A(3,J) = 0
970 A(7,J) = 0
980 IF G <> 2 THEN 1000
990 A(5,J) = 0
1000 IF G <> 1 THEN 1030
1010 A(1,J) = 0
1020 A(9,J) = 0
1030 RETURN
1040 :
1050 REM *** RESULTAAT ***
1060 REM ** AANWIJZING = E$ EN F$ **
1070 IF ( B(1) <> B(2) ) AND ( S <> 7 ) THEN 1190
1080 E$ = " - SUL -"

```

```

1090 P = 1
1100 IF S <> 7 THEN 1130
1110 E$ = " - XANTIPPE -"
1120 P = 2
1130 C(Q,P) = 1 :REM OPSLAAN RESULTAAT
1140 IF ( C(Q,1) <> 1 ) OR ( C(Q,2) <> 1 ) THEN 1190
1150 F$ = "JIJ HEBT GEWONNEN, "
1160 IF Q <> 2 THEN 1180
1170 F$ = "IK HEB GEWONNEN, "
1180 H = 2
1190 RETURN
1200 :
1210 REM *** UITVOER ***
1220 CLS :REM SCHERM WISSEN
1230 PRINT "*** SUL EN XANTIPPE ***"
1240 PRINT "AANTAL WORPEN SPELER:";D(1)
1250 PRINT "AANTAL WORPEN COMPUTER:";D(2)
1260 PRINT
1270 PRINT A$
1280 FOR I = 1 TO 3
1290 PRINT A$(I)
1300 NEXT I
1310 PRINT A$
1320 PRINT
1330 PRINT "TOTAAL GEGOOID ";S;"OGEN WORP: ";D$
1340 PRINT
1350 PRINT F$;
1360 IF H <> 2 THEN 1380
1370 PRINT " MET"; D(Q);"WORPEN."
1380 PRINT
1390 RETURN
1400 :
1410 REM *** BEURT ***
1420 REM ** AANWIJZING = D$ **
1430 H = H * (-1)
1440 Q = 1
1450 D$ = "SPELER"
1460 IF H < 0 THEN 1490
1470 D$ = "COMPUTER"
1480 Q = 2
1490 RETURN
1500 :
1510 REM ***** STUURMODULE *****
1520 REM *** SPELREGELS ***
1530 GOSUB 190
1540 REM *** INITIALIZEREN ***
1550 GOSUB 70
1560 REM *** SPEL ***
1570 REM *** BEURT ***
1580 GOSUB 1430
1590 REM *** STRINGS WISSEN ***
1600 GOSUB 370
1610 K = INT(RND(0)*50+1) :REM WILLEKEURIG GETAL
1620 IF Q <> 1 THEN 1660
1630 REM *** INVOER SPELER ***

```

```

1640 GOSUB 480
1650 REM *** ROLLEN DOBBELSTENEN ***
1660 FOR J = 1 TO 2
1670 GOSUB 530
1680 REM *** DOBBELSTEEN OPBOUWEN ***
1690 GOSUB 770
1700 NEXT J
1710 REM *** RESULTAAT ***
1720 GOSUB 1070
1730 REM *** UITVOER OPBOUWEN ***
1740 GOSUB 600
1750 REM *** UITVOER ***
1760 GOSUB 1220
1770 IF Q <> 1 THEN 1810
1780 REM *** WACHTLUS ***
1790 FOR I = 1 TO 100
1800 NEXT I
1810 IF H <> 2 THEN 1580
1820 END

```

Voorbeeld

```

HOEVEEL KEER MOETEN DE DOBBELSTENEN ROLLEN 7
*** SUL EN XANTIPPE ***
AANTAL WORPEN SPELER: 1
AANTAL WORPEN COMPUTER: 0

```

```

|-----| |-----|
| 0      | |         |
|   0    | |         |
|    0   | |         |
|-----| |-----|

```

```
TOTAAL GEGOOID 4 OGEN WORP: SPELER
```

2.5 Eenentwintigen

Dit spel is een variant op het bekende gokspel Blackjack of in gewoon Nederlands: Eenentwintigen. Het is de bedoeling om 21 punten te scoren of er zo dicht mogelijk bij te komen. Krijgt men meer dan 21 punten dan is men stuk.


De computer speelt de bank. In het spel worden de aas, de boer, de vrouw, de heer, 7, 8, 9 en 10 van alle vier de kleuren gebruikt. De aas telt naar keuze voor 1 of 11 punten. De boer voor 2, de vrouw voor 3 en de heer voor 4 punten. De overige kaarten hebben hun normale waardering. De kleur van de kaart maakt niets uit.

De bank (de computer) trekt twee kaarten voor de speler en twee kaarten voor zichzelf. De kaarten van de speler worden direct gedraaid. Afhankelijk van de kaarten beslist de speler of hij nog meer kaarten wil trekken of dat hij past. Pas nadat de speler heeft gepast, gaat de bank na of hij nog kaarten wil trekken of niet. Past de bank, dan wordt nagegaan wie er heeft gewonnen.

Suggestie 1: Bij eenentwintigen is het gebruikelijk om bedragen in te zetten. Daar is in dit programma niet in voorzien. Het is vrij eenvoudig deze module alsnog in te bouwen. Als de speler wint, wordt de inzet verdubbeld, verliest hij dan is hij zijn inzet kwijt.

Suggestie 2: Sommige computers hebben in de karakterset de symbolen van de kaarten. Het is natuurlijk veel leuker om de kaarten in plaats van de namen op het beeldscherm te zien.

Suggestie 3: Het spelsysteem van de computer ligt vast. Wanneer de computer een bepaald aantal punten overschrijdt, past hij. Wanneer men dit mechanisme beïnvloedt door middel van een random-waarde wordt het spel veel leuker.



Programma

```
10 REM *** EENENTWINTIGEN ***
20 DIM T$(8), A(10), B(2,7)
30 RANDOM
40 GOTO 2100
50 :
60 REM *** SPELREGELS ***
70 CLS
80 PRINT "*** EENENTWINTIGEN ***"           -SPELREGELS -"
90 PRINT
100 PRINT "DE BEDOELING IS MET EEN AANTAL KAARTEN 21"
110 PRINT "PUNTEN TE BEHALEN OF ER DICHT BIJ TE KOMEN."
120 PRINT "GAAT MEN OVER DE 21 PUNTEN HEEN DAN HEEFT MEN"
130 PRINT "VERLOREN. PAST MEN, DAN HEEFT DIEGENE GEWONNEN,"
140 PRINT "DIE HET DICHTSTE BIJ 21 PUNTEN GEKOMEN IS."
150 PRINT "IN HET GEVAL DAT EEN GELIJK AANTAL PUNTEN"
160 PRINT "BEREIKT WORDT; WINT DE BANK (DE COMPUTER). "
170 PRINT
180 PRINT "VOOR HET SPEL ZIJN DE VOLGENDE KAARTEN IN"
190 PRINT "GEBRUIK AAS = 1 OF 11, BOER = 2, VROUW = 3,"
200 PRINT "HEER = 4, 7, 8, 9 EN 10 VAN DE VIER KLEUREN."
210 PRINT
220 PRINT "HEEFT MEN DIT GELEZEN; DRUK DAN <RETURN> IN";
230 INPUT A$
240 RETURN
250 :
260 REM *** INITIALIZEREN ***
270 T$(1) = "SCHOPPEN "
280 T$(2) = "HARTEN   "
290 T$(3) = "KLAVEREN "
300 T$(4) = "RUITEN   "
310 T$(5) = "AAS 11 OF"
320 T$(6) = "BOER     "
330 T$(7) = "VROUW    "
340 T$(8) = "HEER     "
350 T = 0           :REM AANTAL KRT.
360 A = 0           :REM TOT PUNTEN COMPUTER
370 B = 0           :REM AANTAL AZEN COMPUTER
380 C = 2           :REM AANTAL KRT COMP
390 D = 0           :REM TOT PUNTEN SPELER
400 E = 0           :REM AANTAL AZEN SPELER
410 F = 2           :REM AANTAL KRT SPELER
420 RETURN
430 :
440 REM *** EEN KAART VAN DE STAPEL PAKKEN ***
450 M = 0           :REM AAS GETROKKEN
460 K = INT( RND(0) * 4 + 1 ) :REM KLEUR
470 W = INT( RND(0) * 7 + 7 ) :REM WAARDE KRT
480 IF (T < 4) AND (W < 9) THEN 470
490 REM ** IS KAART AL GETROKKEN ? **
500 H = 0           :REM STUURVARIABLE
510 X = K * 100 + W :REM GECODEERDE INFO
520 FOR Y = 1 TO T
530 IF A(Y) <> X THEN 560
```



```

540 H = 1
550 Y = T
560 NEXT Y
570 IF H <> 0 THEN 450
580 IF W <> 11 THEN 600
590 M = 1                               :REM AAS GETROKKEN
600 IF W < 12 THEN 620
610 W = W - 10
620 T = T + 1                             :REM AANTAL KRT + 1
630 A(T) = X                               :REM KRT GETROKKEN
640 RETURN
650 :
660 REM *** EERSTE KEER KAARTEN DELEN ***
670 FOR I = 1 TO 2
680 Z = 0                                   :REM LOK. WAARDE TELLER
690 S = 0                                   :REM LOK. AZEN TELLER
700 FOR J = 1 TO 2
710 REM ** KAART UIT STAPEL **
720 GOSUB 450
730 B(I,J) = X                             :REM KRT IN TABEL
740 S = S + M                               :REM AAS +1
750 Z = Z + W                               :REM WAARDE
760 NEXT J
770 A = Z
780 B = S
790 IF I = 2 THEN 820
800 D = Z
810 E = S
820 NEXT I
830 RETURN
840 :
850 REM *** UITVOER ***
860 CLS
870 PRINT "*** EENENTWINTIGEN ***"
880 PRINT
890 PRINT "JE HEBT ";F;"KAARTEN. ";
900 PRINT TAB(33)"DE BANK HEEFT ";C;"KAARTEN."
910 PRINT
920 P = C                                   :REM MEESTE KRT ?
930 IF C > F THEN 950
940 P = F
950 FOR J = 1 TO P
960 FOR I = 1 TO 2
970 IF (H < 3) AND (I = 2) THEN 1080
980 IF (I = 1) AND (J > F) THEN 1080
990 IF (I = 2) AND (J > C) THEN 1080
1000 X = INT( B(I,J) / 100 )
1010 Y = B(I,J) - X * 100                 :REM WAARDE BEPALEN
1020 A$ = " "
1030 IF Y < 11 THEN 1060
1040 Y = Y - 10                           :REM CORRECTIE VOOR PLAATJE
1050 A$ = T$(Y + 4)
1060 Z = (I - 1) * 33                     :REM TAB-PLAATS IN SCHERM
1070 PRINT TAB(Z) T$(X);" ";A$;Y;
1080 NEXT I

```

```

1090 PRINT
1100 NEXT J
1110 PRINT
1120 IF H < 3 THEN 1190
1130 REM ** ALLEEN UITVOER BIJ EINDE **
1140 PRINT "TOTAAL PUNTEN SPELER ";D;
1150 PRINT TAB(33) "TOTAAL PUNTEN BANK ";A
1160 PRINT
1170 PRINT C$; :REM OPMERKINGEN
1180 PRINT TAB(33) D$
1190 RETURN
1200 :
1210 REM *** SPEL SPELER ***
1220 PRINT "WILT U PASSEN OF EEN KAART TREKKEN <TREK, PAS>:";
1230 INPUT B$
1240 B$ = LEFT$(B$,1) :REM MEEST LINKSE KAR.
1250 H = 0 :REM STUURVARIABLE
1260 IF B$ <> "P" THEN 1280
1270 H = 2 :REM SPELER PAST
1280 IF B$ <> "T" THEN 1300
1290 H = 1 :REM SPELER TREKT KRT
1300 IF H = 0 THEN 1220 :REM FOUTE INVOER
1310 REM ** SPEL SPELER **
1320 IF H = 2 THEN 1470
1330 REM ** KAART TREKKEN **
1340 GOSUB 450
1350 F = F + 1 :REM KRT + 1
1360 B(1,F) = X :REM OPNEMEN TABEL
1370 E = E + M :REM AAS BIJWERKEN
1380 D = D + W :REM WAARDE BIJWERKEN
1390 D1 = D - E * 10 :REM AZEN VOOR 1 GEREKEND
1400 REM ** SPELER STUK ? **
1410 IF D1 < 22 THEN 1470
1420 C$ = "JIJ BENT STUK !!"
1430 D$ = "IK HEB GEWONNEN."
1440 H = 3 :REM SPELER GEWONNEN
1450 D = D1 :REM VAST HOUDEN COMPWAARDE
1460 E = 0 :REM AZEN OP 0 STELLEN
1470 RETURN
1480 :
1490 REM *** SPEL COMPUTER ***
1500 IF H > 2 THEN 1780
1510 H = 4
1520 FOR Y = 0 TO B
1530 A1 = A - Y * 10
1540 IF A1 > 21 THEN 1630
1550 H = 0
1560 IF A1 < 21 THEN 1590
1570 D$ = "IK HEB GEWONNEN MET 21 PUNTEN."
1580 H = 1
1590 IF A1 < 18 THEN 1630
1600 REM ** COMPUTER PAST **
1610 H = H + 2
1620 Y = B
1630 NEXT Y

```

```

1640 IF H > 1 THEN 1720
1650 REM ** KAART TREKKEN **
1660 GOSUB 450
1670 C = C + 1 :REM AANTAL KAARTEN
1680 B(2,C) = X :REM BIJHOUDEN KART
1690 B = B + M :REM AANTAL AZEN
1700 A = A + W :REM TOT.WAARDE
1710 REM ** COMPUTER STUK **
1720 IF A1 < 22 THEN 1780
1730 A = A1 :REM WAARDE STUK
1740 B = 0 :REM AZEN GELIJK 0
1750 D$ = "IK BEN STUK !!"
1760 C$ = "JIJ HEBT GEWONNEN."
1770 H = 4
1780 RETURN
1790 :
1800 REM *** RESULTAAT ***
1810 REM ** HOOGSTE WAARDE SPELER **
1820 FOR X = 0 TO E
1830 D1 = D - X * 10
1840 IF D1 > 21 THEN 1920
1850 D = D1
1860 X = E :REM UIT LUS
1870 IF D <> 21 THEN 1920
1880 IF H > 2 THEN 1920
1890 H = 4
1900 C$ = "JIJ HEBT GEWONNEN MET 21 PUNTEN MAZZELAAR !"
1910 D$ = ""
1920 NEXT X
1930 FOR X = 0 TO B
1940 A1 = A - X * 10
1950 IF A1 > 21 THEN 1980
1960 A = A1 :REM HOOGST WAARDE < 21
1970 X = B :REM UIT LUS
1980 NEXT X
1990 IF H > 2 THEN 2060
2000 REM ** BEPALING WINNAAR GEEN 21 PUNTEN **
2010 D$ = "IK HEB GEWONNEN "
2020 H = 3
2030 IF A >= D THEN 2060
2040 C$ = "JIJ HEBT GEWONNEN."
2050 H = 4
2060 RETURN
2070 :
2080 REM ***** STUURMODULE *****
2090 REM *** SPELREGELS ***
2100 GOSUB 70
2110 REM *** INITIALIZEREN ***
2120 GOSUB 270
2130 REM *** EERSTE KEER KAART TREKKEN ***
2140 GOSUB 670
2150 REM *** UITVOER ***
2160 GOSUB 860
2170 REM *** INVOER SPELER ***
2180 GOSUB 1220

```

```

2190 IF H < 2 THEN 2160
2200 REM *** SPEL COMPUTER ***
2210 GOSUB 1500
2220 IF H < 2 THEN 2210
2230 REM *** RESULTAAT ***
2240 GOSUB 1820
2250 REM *** UITVOER ***
2260 GOSUB 860
2270 END

```

Voorbeeld

*** EENENTWINTIGEN ***

JE HEBT 2 KAARTEN.

DE BANK HEEFT 2 KAARTEN.

RUITEN	BOER	2
SCHOPPEN		9

WILT U PASSEN OF EEN KAART TREKKEN <TREK, PAS>:TREK

*** EENENTWINTIGEN ***

JE HEBT 3 KAARTEN.

DE BANK HEEFT 2 KAARTEN.

RUITEN	BOER	2
SCHOPPEN		9
KLAVEREN	VROUW	3

WILT U PASSEN OF EEN KAART TREKKEN <TREK, PAS>:TREK

*** EENENTWINTIGEN ***

JE HEBT 4 KAARTEN.

DE BANK HEEFT 2 KAARTEN.

RUITEN	BOER	2
SCHOPPEN		9
KLAVEREN	VROUW	3
RUITEN		8

RUITEN	10
KLAVEREN	9

TOTAAL PUNTEN SPELER 22

TOTAAL PUNTEN BANK 19

JIJ BENT STUK !!

IK HEB GEWONNEN.

2.6 Hoog-laag

Bij dit spel speelt de computer tegen de speler en de speler tegen de computer. De speler wordt gevraagd een getal te nemen tussen 0 en 100. De computer doet hetzelfde. De computer laat een getal zien en vraagt of dit het getal is dat de speler heeft gekozen. Klopt het, dan geeft men 'juist'. Is het getal dat de speler heeft hoger respectievelijk lager, dan geeft men 'hoger' respectievelijk 'lager'. Maakt de speler een fout dan heeft de computer dat uiteindelijk door en maakt de speler er op attent. De computer doet exact hetzelfde, hij geeft de speler aanwijzingen of hij hoger of lager moet kiezen.

Het programma maakt gebruik van de bekende binaire zoekmethode.

Suggestie 1: Er wordt uitsluitend gespeeld met getallen van 0 t/m 100. Door het omzetten van een variabele kan men ook met getallen van 0 t/m 1000 of van 0 t/m 10 spelen.

Suggestie 2: Dit spel is uitsluitend opgezet om met getallen te spelen. Het zou ook heel aantrekkelijk zijn om het spel met letters te spelen (1 t/m 26). Een soort alfabetiseren. Men moet dan gebruik maken van de numerieke waarde van de letters. De wijzigingen die men moet doorvoeren, zijn heel beperkt. Men vraagt een karakter op via het toetsenbord, converteert het karakter naar een numerieke waarde en vergelijkt deze met de numerieke waarde van het karakter dat de computer heeft. Al naar gelang de uitslag geeft men 'hoger', 'lager' of juist. De speler doet hetzelfde bij de suggesties van de computer.

Programma

```
10 REM *** HOOG - LAAG ***
20 RANDOM
30 GOTO 1040
40 :
50 REM *** SPELREGELS ***
60 CLS :REM SCHERM WISSEN
70 PRINT "*** HOOG - LAAG ***"
80 PRINT
90 PRINT "DE COMPUTER EN DE SPELER NEMEN BEIDE EEN"
100 PRINT "GEHEEL GETAL TUSSEN 1 EN 100 IN GEDACHTEN"
110 PRINT "DE SPELER BEGINT MET RADEN. AL NAAR GELANG"
120 PRINT "HET RESULTAAT GEEFT DE COMPUTER 'HOGER', "
130 PRINT "'LAGER' OF 'JUIST'. DE SPELER DOET HETZELFDE"
140 PRINT "BIJ DE SUGGESTIE VAN DE COMPUTER."
150 PRINT "DEGENE, DIE ALS EERSTE HET GETAL RAADT, HEEFT"
160 PRINT "GEWONNEN."
170 PRINT
180 PRINT "> NEEM EEN GETAL IN GEDACHTEN TUSSEN 1 EN 100"
190 PRINT
200 PRINT "GEDAAN !! DRUK DAN DE <RETURN-TOETS> IN";
210 INPUT A$
220 CLS :REM SCHERM WISSEN
230 RETURN
240 :
250 REM *** INITIALIZEREN ***
260 N = 100 :REM AANTAL GETALLEN
270 X1 = 1 :REM ONDERGREN
```

```

280 X2 = N + 1 :REM BOVENGRENS
290 X3 = X1 + INT((X2-X1) / 2) :REM EERSTE GOK
300 G = INT(RND(0) * 100 + 1) :REM GETAL COMPUTER
310 S = 0 :REM BEURTENTELLER SPELER
320 C = 0 :REM BEURTENTELLER COMPUTER
330 E = 0 :REM VORIGE GETAL COMPUTER
340 GOSUB 800
350 RETURN
360 :
370 REM *** INVOER SPELER GETAL ***
380 PRINT "WELK GETAL DENK JE DAT IK HEB ";
390 INPUT K
400 K = INT(K)
410 REM ** CONTROLE SPELER **
420 REM ** AANWIJZING = B$ **
430 H = 0
440 B$ = "JIJ MOET HOGER !"
450 IF K < G THEN 470
460 B$ = "JIJ MOET LAGER !"
470 IF K <> G THEN 500
480 B$ = "INDERDAAD DAT WAS HET GETAL !"
490 H = 2
500 RETURN
510 :
520 REM *** INVOER SPELER ANTWOORD ***
530 REM ** AANWIJZING = C$ **
540 H = 0
550 PRINT "MOET IK 'HOGER', 'LAGER' OF IS HET JUIST";
560 INPUT Q$
570 Q$ = LEFT$(Q$,1) :REM ALLEEN EERST KAR.
580 IF Q$ <> "H" THEN 610
590 X1 = X3
600 H = 1
610 IF Q$ <> "L" THEN 640
620 X2 = X3
630 H = 1
640 IF Q$ <> "J" THEN 670
650 H = 2
660 C$ = "GOED HE !!"
670 IF H <> 0 THEN 690
680 C$ = "IK BEGRIJP JE NIET <HOGER, LAGER OF JUIST>"
690 IF H = 2 THEN 770
700 REM ** BEREKEN GETAL **
710 REM ** AANWIJZING = D$ **
720 X3 = X1 + INT ((X2-X1)/2)
730 IF E <> X3 THEN 760
740 D$ = "VOLGENS MIJ WEET JIJ NIET WAT HOGER EN LAGER IS !!"
750 H = 2 :REM STOPPEN
760 E = X3
770 RETURN
780 :
790 REM *** STRINGS WISSEN ***
800 B$ = ""
810 C$ = ""
820 D$ = ""

```

```

830 E$ = ""
840 S = S + 1 :REM OPHOGEN BEURTTELLER
850 RETURN
860 :
870 REM *** UITVOER COMPUTER ***
880 CLS :REM SCHERM WISSEN
890 PRINT "*** HOOG - LAAG ***";
900 PRINT TAB(35) "BEURT:";S
910 PRINT
920 PRINT "JIJ KOOS ";K ;B$
930 PRINT
940 IF H = 2 THEN 970
950 PRINT "IS JOUW GETAL SOMS ":";X3
960 PRINT
970 RETURN
980 :
990 REM *** UITVOER SPELER ***
1000 PRINT C$
1010 PRINT D$
1020 RETURN
1030 :
1040 REM **** STUURMODULE ****
1050 REM *** SPELREGELS ***
1060 GOSUB 60
1070 REM *** INITIALIZEREN ***
1080 GOSUB 260
1090 REM *** INVOER SPELER GETAL ***
1100 GOSUB 380
1110 REM *** UITVOER ***
1120 GOSUB 880
1130 IF H = 2 THEN 1240
1140 REM *** INVOER SPELER ANTWOORD ***
1150 GOSUB 540
1160 REM *** UITVOER SPELER ***
1170 GOSUB 1000
1180 REM *** STRINGS WISSEN ***
1190 GOSUB 800
1200 IF H < 2 THEN 1100
1210 REM *** EINDE ***
1220 PRINT "MIJN GETAL WAS ";G
1230 PRINT
1240 END

```


Voorbeeld

WELK GETAL DENK JE DAT IK HEB 10
*** HOOG - LAAG *** BEURT: 2

JIJ KOOS 10 JIJ MOET HOGER !

IS JOUW GETAL SOMS : 51

MOET IK 'HOGER', 'LAGER' OF IS HET JUISTHOGER
WELK GETAL DENK JE DAT IK HEB 51
*** HOOG - LAAG *** BEURT: 3

JIJ KOOS 51 JIJ MOET HOGER !

IS JOUW GETAL SOMS : 76

MOET IK 'HOGER', 'LAGER' OF IS HET JUISTJUIST
GOED HE !!

2.7 Gokken

In dit spel geeft het programma een getal tussen 0 en 100. De speler wordt gevraagd of het volgende getal hoger of lager is dan het vorige. Na beantwoording kan men inzetten op zijn gok.

Is de gok juist dan wordt de inzet verdubbeld, was het fout, dan is men zijn inzet kwijt. Het is alleen jammer dan de machine de winst niet uitkeert.

Suggestie : Dit programma is zó gemaakt, dat het slechts door één persoon kan worden gespeeld. Vaak is het veel leuker wanneer meer mensen mee kunnen spelen. Men zou het programma zodanig kunnen wijzigen dat dit kan. Men moet dan wel een vastgesteld aantal ronden spelen. Aan het eind van het spel kan de winnaar worden bepaald.



Programma

```
10 REM *** GOKKEN ***
20 RANDOM
30 GOTO 990
40 :
50 REM *** SPELREGELS ***
60 CLS                               :REM SCHERM WISSEN
70 PRINT "*** GOKKEN ***"           --SPELREGELS --"
80 PRINT
90 PRINT "DE BEDOELING VAN DIT SPEL IS TE RADEN OF "
100 PRINT "HET VOLGENDE TE TREKKEN GETAL HOGER OF LAGER"
110 PRINT "IS DAN HET GETROKKEN GETAL. HEEFT MEN GOED"
120 PRINT "GEGOKT DAN WORDT DE INZET VERDUBBELD. "
130 PRINT "IN HET ANDERE GEVAL IS MEN ZIJN INZET KWIJT."
140 PRINT "BLIJKT HET VOLGENDE GETAL ECHTER GELIJK TE ZIJN"
150 PRINT "DAN IS MEN EVENEENS ZIJN INZET KWIJT."
```

```

160 PRINT
170 PRINT "DE GETALLEN, DIE GETROKKEN WORDEN LIGGEN"
180 PRINT "TUSSEN 1 EN 100."
190 PRINT "INZETTEN KAN MEN DOEN TOT MAXIMAAL 100 GLD
200 PRINT
210 PRINT "HEEFT MEN DIT GELEZEN DRUK DAN <RETURN> IN";
220 INPUT A$
230 RETURN
240 :
250 REM *** INITIALIZEREN ***
260 S = 0 :REM TOTAAL WINST
270 G = INT(RND(0) * 100 + 1) :REM GETROKKEN GETAL
280 P = 0
290 T = 0 :REM TOTAAL INZET
300 RETURN
310 :
320 REM *** UITVOER ***
330 CLS :REM SCHERM WISSEN
340 PRINT "*** GOKKEN ***"
350 PRINT
360 PRINT "TOTALE WINST :";S;
370 PRINT TAB(30) "TOTAAL INGEZET :";T
380 PRINT
390 IF T = 0 THEN 470
400 PRINT "HET EERSTE GETAL WAS :";X
410 PRINT "HET TWEDE GETAL WAS :";Q
420 PRINT
430 PRINT "JIJ GOKTE ";A$;" EN HET IS ";C$
440 PRINT D$;Z
450 PRINT
460 PRINT"-----"
470 PRINT "HET EERSTE GETAL IS :";G
480 PRINT
490 RETURN
500 :
510 REM *** INVOER ***
520 PRINT "IS HET VOLGENDE GETAL HOGER OF LAGER ";
530 PRINT "<HOGER,LAGER,STOP>";
540 INPUT A$
550 P = 0
560 A$ =LEFT$(A$,1) :REM LINKER KAR
570 IF A$ <> "S" THEN 590
580 P = 3
590 IF A$ <> "H" THEN 620
600 A$ = "HOGER"
610 P = 1
620 IF A$ <> "L" THEN 650
630 A$ = "LAGER"
640 P = 2
650 IF P = 0 THEN 520
660 IF P = 3 THEN 720
670 REM ** INZET **
680 PRINT "HOEVEEL ZET U IN ";
690 INPUT Z
700 IF ABS(Z) > 100 THEN 680

```

```

710 Z = ABS(Z)                                :REM ALLEEN POS WAARDEN
720 RETURN
730 :
740 REM *** BEREKENING ***
750 H = 0                                       :REM STUURVARIABLE
760 Q = INT(RND(0)* 100 + 1)
770 X = G
780 C$ = "GELIJK"
790 IF G < Q THEN 840
800 REM ** TWEEDE WAARDE IS KLEINER **
810 C$ = "LAGER"
820 IF P <> 2 THEN 890
830 H = 1
840 IF G > Q THEN 890
850 REM ** TWEEDE WAARDE IS GROTER **
860 C$ = "HOGER"
870 IF P <> 1 THEN 890
880 H = 1
890 D$ = "JE BENT JE INZET KWIJT VAN : "
900 IF H <> 1 THEN 930
910 D$ = "JE VERDUBBELDE JE INZET VAN : "
920 S = S + 2 * Z
930 T = T + Z
940 G = INT(RND(0) * 100 + 1)
950 RETURN
960 :
970 REM **** STUURMODULE ****
980 REM *** SPELREGELS ***
990 GOSUB 60
1000 REM *** INITIALIZEREN ***
1010 GOSUB 260
1020 REM *** BEREKENING ***
1030 GOSUB 750
1040 REM *** UITVOER ***
1050 GOSUB 330
1060 REM *** INVOER ***
1070 GOSUB 520
1080 IF P <> 3 THEN 1030
1090 PRINT "BEDANKT VOOR HET SPEL, TOT DE VOLGENDE KEER."
1100 END

```

Voorbeeld

*** GOKKEN ***

TOTALE WINST : 0

TOTAAL INGEZET : 0

HET EERSTE GETAL IS : 27

IS HET VOLGENDE GETAL HOGER OF LAGER <HOGER,LAGER,STOP>HOGER
HOEVEEL ZET U IN 50

*** GOKKEN ***

TOTALE WINST : 100

TOTAAL INGEZET : 50

HET EERSTE GETAL WAS : 27

HET TWEEDE GETAL WAS : 45

JIJ GOKTE HOGER EN HET IS HOGER
JE VERDUBBELDE JE INZET VAN : 50

HET EERSTE GETAL IS : 58

IS HET VOLGENDE GETAL HOGER OF LAGER <HOGER,LAGER,STOP>HOGER
HOEVEEL ZET U IN 25

*** GOKKEN ***

TOTALE WINST : 100

TOTAAL INGEZET : 75

HET EERSTE GETAL WAS : 58

HET TWEEDE GETAL WAS : 2

JIJ GOKTE HOGER EN HET IS LAGER
JE BENT JE INZET KWIJT VAN : 25

HET EERSTE GETAL IS : 16

IS HET VOLGENDE GETAL HOGER OF LAGER <HOGER,LAGER,STOP>STOP
BEDANKT VOOR HET SPEL, TOT DE VOLGENDE KEER.

2.8 Letters

Letters is een programma dat men onder de leerprogramma's zou kunnen rangschikken. De werking van het programma is vrij eenvoudig. Met behulp van een module worden één of meer willekeurige letters op het beeldscherm geplaatst. Afhankelijk van de tijd blijven deze letters op het beeldscherm staan. Na het verstrijken van de tijd vraagt het programma welke letters op het beeldscherm stonden. Men moet dan de letters die men heeft gezien in de juiste volgorde weer intikken. Het programma geeft aan of men het goed of fout heeft gedaan en wat het had moeten zijn. Heeft men het goed gedaan, dan zal het aantal letters met één worden uitgebreid, maar de tijd wordt met één derde bekort. Was het antwoord fout dan wordt het aantal letters niet met één vermeerderd en de tijd wordt met een derde gedeelte verlengd.

Het record is op dit moment twaalf letters.

Het is opmerkelijk dat kinderen bij dit spelletje beter scoren dan volwassenen.

Suggestie 1: Dit spel wordt uitsluitend met letters gespeeld. Het is een bekend gegeven dat getallen gemakkelijker worden onthouden. Met dit gegeven in het achterhoofd kan men spelniveaus in het programma aanbrengen. Het laagste spelniveau werkt alleen met getallen. Het tweede spelniveau uitsluitend met karakters. Het hoogste spelniveau is een combinatie van letters en karakters.

Suggestie 2: In suggestie 1 kan de speler zelf het spelniveau kiezen. Het is best leuk dit spelniveau door het programma te laten bepalen. Het programma begint bij het laagste niveau, wanneer blijkt dat de speler weinig fouten maakt wordt overgeschakeld naar spelniveau 2. Gaat dit ook weer goed dan kan worden overgeschakeld naar spelniveau 3.

Suggestie 3: Bij de ontwikkeling is ervan uitgegaan dat men alléén speelt. Men zou ook een zeker competitie-element kunnen toevoegen door te vragen met hoeveel mensen men speelt en de scores bij te houden. Er moet dan door het programma wel een maximaal aantal beurten worden vastgesteld. Bijvoorbeeld tien beurten per speler. Is de competitie voorbij, dan bepaalt het programma de winnaar.

Programma

```
10 REM *** LETTERS ***
20 RANDOM
30 GOTO 980
40 :
50 REM *** SPELREGELS ***
60 CLS                                :REM SCHERM WISSEN
70 PRINT "*** LETTERS ***           -SPELREGELS-"
80 PRINT
90 PRINT "DE COMPUTER GEEFT EEN OF MEERDERE LETTERS. DEZE"
100 PRINT "LETTERS BLIJVEN GEDURENDE EEN KORTE TIJD STAAN,"
110 PRINT "DAARNA VERDWIJNEN ZE PLOTSELING."
120 PRINT "JIJ MOET DAN PROBEREN DE LETTERS IN DE GOEDE"
130 PRINT "VOLGORDE WEER IN TE TIKKEN."
140 PRINT
150 PRINT "VERGEET NA DE LETTERS NIET <RETURN>"
160 PRINT "IN TE DRUKKEN."
```

```

170 PRINT "HEB JE DIT GELEZEN, DRUK DAN <RETURN> IN.";
180 INPUT Q$
190 RETURN
200 :
210 REM *** INITIALIZEREN ***
220 T = 400                                :REM TIJDSVERTRAGING
230 Q = 1                                  :REM AANTAL LETTERS
240 V = 0                                  :REM BEURT
250 G = 0                                  :REM AANTAL GOEDEN
260 A$ = "ABCDEFGHIJKLMNOPQRSTUVWXYZ"
270 RETURN
280 :
290 REM *** LETTERS TREKKEN ***
300 V$ = ""                                :REM VRAAGSTRING SCHOON
310 V = V + 1                              :REM BEURT + 1
320 FOR I = 1 TO Q
330 L = INT(RND(0) * 26 + 1)
340 B$ = LEFT$(A$,L)
350 B$ = RIGHT$(B$,1)
360 V$ = V$ + B$                          :REM CONCATENATIE
370 NEXT I
380 RETURN
390 :
400 REM *** RESULTAAT ***
410 H = 1
420 T1 = INT(T / 4)                        :REM TIJD + OF -
430 R$ = "JAMMER ! HET IS FOUT."
440 IF V$ <> Q$ THEN 490
450 H = -1
460 Q = Q + 1                              :REM EEN LETTER ERBIJ
470 G = G + 1                              :REM AANTAL GOEDEN +1
480 R$ = "JE ANTWOORD WAS GOED !!"
490 T = T + T1 * H
500 RETURN
510 :
520 REM *** UITVOER-KOP ***
530 CLS                                    :REM SCHERM WISSEN
540 PRINT "*** LETTERS ***";
550 PRINT TAB(35) "AANTAL BEURTEN :";V
560 PRINT TAB(35) "AANTAL GOEDEN :";G
570 PRINT TAB(35) "AANTAL LETTERS :";Q
580 PRINT
590 PRINT
600 RETURN
610 :
620 REM *** UITVOER-RESULTAAT ***
630 GOSUB 530
640 PRINT "DE VRAAG WAS :";V$
650 PRINT "HET ANTWOORD WAS :";Q$
660 PRINT
670 PRINT R$
680 PRINT
690 PRINT "WIL JE DOORGAAN JA/NEE :";
700 INPUT Q$
710 RETURN

```

```

720 :
730 REM *** UITVOER-VRAAG ***
740 GOSUB 530
750 PRINT "DE LETTERS ZIJN   : "V$
760 FOR I = 1 TO T
770 NEXT I
780 CLS                               : REM SCHERM WISSEN
790 RETURN
800 :
810 REM *** INVOER ***
820 GOSUB 530
830 PRINT "WELKE LETTERS WAREN HET : ";
840 INPUT Q$
850 RETURN
860 :
870 REM *** STOP ***
880 GOSUB 530
890 X = INT( G / V * 1000) / 10
900 PRINT "JE BENT TOT ";Q;" LETTERS GEKOMEN";
910 PRINT " IN ";V;" BEURTEN,"
920 PRINT "WAARVAN ";X ;"% GOED."
930 PRINT "GOED HOOR !!"
940 RETURN
950 :
960 REM ***** STUURMODULE *****
970 REM *** INITIALIZEREN ***
980 GOSUB 220
990 REM *** SPELREGELS ***
1000 GOSUB 60
1010 REM *** LETTERS TREKKEN ***
1020 GOSUB 300
1030 REM *** UITVOER-VRAAG ***
1040 GOSUB 740
1050 REM *** INVOER ***
1060 GOSUB 820
1070 REM *** RESULTAAT ***
1080 GOSUB 410
1090 REM *** UITVOER RESULTAAT ***
1100 GOSUB 630
1110 IF LEFT$(Q$,1) <> "N" THEN 1020
1120 REM *** STOP ***
1130 GOSUB 880
1140 END

```

Voorbeeld

```

*** LETTERS ***
AANTAL BEURTEN : 3
AANTAL GOEDEN  : 2
AANTAL LETTERS : 3

```

```
DE LETTERS ZIJN   :EZX
```


*** LETTERS ***

AANTAL BEURTEN : 3
AANTAL GOEDEN : 2
AANTAL LETTERS : 3

WELKE LETTERS WAREN HET :EZS

*** LETTERS ***

AANTAL BEURTEN : 3
AANTAL GOEDEN : 3
AANTAL LETTERS : 4

DE VRAAG WAS :EZS
HET ANTWOORD WAS :EZS

JE ANTWOORD WAS GOED !!

WIL JE DOORGAAN JA/NEE :J

*** LETTERS ***

AANTAL BEURTEN : 4
AANTAL GOEDEN : 3
AANTAL LETTERS : 4

DE LETTERS ZIJN :VHBJ

*** LETTERS ***

AANTAL BEURTEN : 4
AANTAL GOEDEN : 3
AANTAL LETTERS : 4

WELKE LETTERS WAREN HET :VBHJ

*** LETTERS ***

AANTAL BEURTEN : 4
AANTAL GOEDEN : 3
AANTAL LETTERS : 4

DE VRAAG WAS :VHBJ
HET ANTWOORD WAS :VBHJ

JAMMER ! HET IS FOUT.

WIL JE DOORGAAN JA/NEE :NEE

*** LETTERS ***

AANTAL BEURTEN : 4
AANTAL GOEDEN : 3
AANTAL LETTERS : 4

JE BENT TOT 4 LETTERS GEKOMEN IN 4 BEURTEN,
WAARVAN 75 % GOED.
GOED HOOR !!

2.9 Galgje

Tijdens de Romeinse tijd werd in Gallië ook al galgje gespeeld. Men speelde het toen wel iets anders dan tegenwoordig. Uit verschillende geschriften blijkt dat het lot iemand aanwees. De ongelukkige werd een strop om de hals gedaan en aan een staak gehangen. Hij kreeg een zwaard in zijn handen, daarmee moest hij de strop zien door te snijden. Of dit vaak lukte, verhaalt de overlevering niet.

Dit spel is niet zo bloeddorstig, alhoewel !

De computer (de beul) kiest een woord. Men moet proberen dit woord te raden. Wanneer men van een letter denkt dat deze voorkomt, geeft men dit op. Is de letter juist, dan zal hij op de goede plaats op het scherm worden gezet. Is de letter fout dan begint de beul met zijn werk. Heeft men meer dan tien keer fout geraden, dan hangt men.

Suggestie 1: Heeft men een paar keer gespeeld, dan kent men alle woorden die in de interne datafile zitten uit het hoofd. Het eerste wat men zou kunnen doen, is de datafile drastisch uitbreiden met andere woorden.

Suggestie 2: U zult hebben gemerkt dat de woorden die u gebruikt niet al te groot zijn. Het is best leuk om moeilijke woorden te nemen. Men moet dan bijvoorbeeld wel de eerste en laatste letter geven. Dit is slechts een kleine aanpassing in de module WOORD OP-HALEN .

Suggestie 3: Daar in dit boek is uitgegaan van standaard-BASIC zijn de plaatjes natuurlijk niet zo spectaculair. Het plaatje van het galgje is zeker voor verbetering vatbaar. Zeker wanneer men beschikt over 'high resolution'.

Programma

```
10 REM *** GALGJE ***
20 DIM A$(10), B$(20), D$(20)
30 DATA COMPUTER, CLOWN, VUUR, MOLEN, AARDBEI, DEUR,
40 DATA BOOMGAARD, SPEELGOED, ZOMER, HERFST
50 DATA POP, AUTO, MOTOR, VLIEGTUIG, BOOT
60 RANDOM
70 GOTO 1640
80 :
90 REM *** SPELREGELS ***
100 CLS                                :REM SCHERM WISSEN
110 PRINT "*** GALGJE ***"             -SPELREGELS-"
120 PRINT
130 PRINT "DE COMPUTER TREKT EEN WOORD. U MOET DAT WOORD"
140 PRINT "ZIEN TE RADEN. EEN LETTER, WAARVAN U DENKT"
150 PRINT "DAT HIJ IN HET WOORD VOORKOMT, GEEFT U OP."
160 PRINT "KOMT DE LETTER INDERDAAD VOOR, DAN VERSCHIJNT"
170 PRINT "DE LETTER BOVEN IN HET BEELDSCHERM OP DE JUISTE"
180 PRINT "PLAATS. IS HIJ FOUT, DAN BEGINT DE BEUL MET DE"
190 PRINT "TERECHTSTELLING U KUNT MAXIMAAL 10 FOUTEN MAKEN,"
200 PRINT "ANDERS HANGT U! KOMT EEN LETTER MEERDERE MALEN"
210 PRINT "VOOR, DAN MOET U HEM OOK MEERDERE KEREN VRAGEN."
220 PRINT "DENKT U HET WOORD TE WETEN, TOETS HET DAN IN."
```

```

230 PRINT
240 PRINT "DRUK <RETURN> IN, WANNEER U DIT GELEZEN HEEFT. ";
250 INPUT A$
260 RETURN
270 :
280 REM *** INITIALIZEREN ***
290 A$(1) = "HET PLATFORM"
300 A$(2) = "DE STAAK"
310 A$(3) = "HET DWARSHOUT"
320 A$(4) = "DE STROP"
330 A$(5) = "HET HOOFD"
340 A$(6) = "DE ROMP"
350 A$(7) = "DE LINKERARM"
360 A$(8) = "RECHTERARM"
370 A$(9) = "HET LINKERBEEN"
380 A$(10) = "HET RECHTERBEEN"
390 G$ = "" : REM GEBRUIKTE KARAKTERS
400 C$ = ""
410 B = 0 : REM BEURT TELLER
420 H = 0
430 F = 0 : REM FOUTTELLER
440 RETURN
450 :
460 REM *** WOORD OPHALEN UIT INTERNE FILE ***
470 X = INT(RND(0) * 15 + 1) : REM WOORDNUMMER
480 RESTORE
490 FOR I = 1 TO X
500 READ A$
510 NEXT I
520 L = LEN(A$) : REM LENGTE WOORD
530 FOR I = 1 TO L
540 B$(I) = "."
550 Q$ = LEFT$(A$, I)
560 D$(I) = RIGHT$(Q$, 1)
570 NEXT I
580 RETURN
590 :
600 REM *** KOMT LETTER VOOR ? ***
610 H = 0
620 E$ = "" : REM STRING WISSEN
630 FOR I = 1 TO L
640 IF D$(I) <> Q$ THEN 700
650 H = 1
660 IF B$(I) = Q$ THEN 700
670 H = 2
680 B$(I) = Q$
690 I = L
700 NEXT I
710 IF H <> 0 THEN 760
720 C$ = Q$ + " : DEZE LETTER KOMT NIET VOOR !"
730 G$ = G$ + Q$
740 F = F + 1
750 E$ = A$(F)
760 IF H <> 1 THEN 780
770 C$ = Q$ + " : DEZE LETTER IS AL GEBRUIKT !"

```

```

780 IF H <> 2 THEN 880
790 C$ = Q$ + " :DE LETTER IS GOED"
800 RETURN
810 :
820 REM *** INVOER ***
830 PRINT "GEEF EEN LETTER OF GEEF HET HELE WOORD ";
840 INPUT Q$
850 X = LEN(Q$)           :REM AANTAL INGETOETSTE KAR.
860 IF X = 1 THEN 880
870 X = 2
880 RETURN
890 :
900 REM *** HELE WOORD ***
910 H = 0
920 F = F + 1
930 C$ = "NEE, HOOR HET WOORD IS ANDERS !"
940 IF Q$ <> A$ THEN 1020
950 REM ** WOORD GERADEN **
960 C$ = "JA, HOOR DAT WAS HET WOORD "
970 H = 3
980 F = F - 1
990 FOR I = 1 TO L
1000 B$(I) = D$(I)
1010 NEXT I
1020 RETURN
1030 :
1040 REM *** UITVOER ***
1050 CLS           :REM SCHERM WISSEN
1060 PRINT "*** GALGJE ***";
1070 PRINT TAB(45) "BEURT ";B
1080 PRINT
1090 PRINT TAB(5) "! ";
1100 FOR I = 1 TO L
1110 PRINT B$(I);
1120 NEXT I
1130 PRINT " !"
1140 IF F <= 1 THEN 1180
1150 PRINT "I";
1160 IF F = 2 THEN 1180
1170 PRINT "XXXXXXXXXXXXXXXX";
1180 PRINT TAB(23) "HET WOORD BESTAAT UIT ";L;"LETTERS."
1190 IF F <= 1 THEN 1230
1200 PRINT "I";
1210 IF F < 4 THEN 1230
1220 PRINT TAB(14) "X";
1230 PRINT TAB(23) "DE FOUT GEBRUIKTE LETTERS WAREN :";
1240 IF F <= 1 THEN 1280
1250 PRINT "I";
1260 IF F < 5 THEN 1280
1270 PRINT TAB(14) "O";
1280 PRINT TAB(23) G$
1290 IF F <= 1 THEN 1540
1300 PRINT "I";
1310 IF F < 6 THEN 1370
1320 IF F < 7 THEN 1340

```

```

1330 PRINT TAB(10) "XXXX";
1340 PRINT TAB(14) "X";
1350 IF F < 8 THEN 1370
1360 PRINT "XXXX";
1370 PRINT
1380 PRINT "I";
1390 FOR I = 1 TO 2
1400 IF F < 6 THEN 1420
1410 PRINT TAB(14) "X";
1420 PRINT
1430 PRINT "I";
1440 NEXT I
1450 FOR I = 1 TO 2
1460 IF F < 9 THEN 1500
1470 PRINT TAB(14 - I) "X";
1480 IF F < 10 THEN 1500
1490 PRINT TAB(14 + I) "X";
1500 PRINT
1510 PRINT "I";
1520 NEXT I
1530 IF F > 1 THEN 1570
1540 FOR I = 1 TO 5
1550 PRINT
1560 NEXT I
1570 PRINT TAB(23) C$
1580 IF F = 0 THEN 1600
1590 PRINT "XXXXXXXXXXXXXXXXXXXX"; TAB(23) E$
1600 RETURN
1610 :
1620 REM **** STUURMODULE ****
1630 REM *** SPELREGELS ***
1640 GOSUB 100
1650 REM *** INITIALIZEREN ***
1660 GOSUB 290
1670 REM *** WOORD OPHALEN ***
1680 GOSUB 470
1690 REM *** UITVOER ***
1700 GOSUB 1050
1710 REM *** INVOER ***
1720 GOSUB 830
1730 B = B + 1
1740 REM *** BEWERKING ***
1750 ON X GOSUB 610 , 910
1760 IF (H <> 3) AND (F < 10) THEN 1700
1770 REM *** UITVOER ***
1780 GOSUB 1050
1790 C$ = "GOED HOOR, HET WOORD WAS INDERDAAD : "
1800 IF H = 3 THEN 1820
1810 C$ = "JE HANGT, HET WOORD WAS : "
1820 PRINT C$;A$
1830 END

```



Voorbeeld

*** GALGJE ***

BEURT 0

! !

HET WOORD BESTAAT UIT 5 LETTERS.
DE FOUT GEBRUIKTE LETTERS WAREN :

GEEF EEN LETTER OF GEEF HET HELE WOORD W
*** GALGJE ***

BEURT 1

W :DE LETTER IS GOED
GEEF EEN LETTER OF GEEF HET HELE WOORD Y
*** GALGJE ***

BEURT 2

! ...W. !

HET WOORD BESTAAT UIT 5 LETTERS.
DE FOUT GEBRUIKTE LETTERS WAREN :
Y

Y :DEZE LETTER KOMT NIET VOOR !
XXXXXXXXXXXXXXXXXXXXX HET PLATFORM
GEEF EEN LETTER OF GEEF HET HELE WOORD A

De beurten 3 t/m 11 zijn niet afgedrukt.

*** GALGJE ***

BEURT 12

! C.OWN !

XXXXXXXXXXXXXXXXXXXXX
I X
I O
I XXXXXXXXXX
I X
I X
I
I
I

HET WOORD BESTAAT UIT 5 LETTERS.
DE FOUT GEBRUIKTE LETTERS WAREN :
YAHDIIKU

O :DE LETTER IS GOED
XXXXXXXXXXXXXXXXXXXXX
GEEF EEN LETTER OF GEEF HET HELE WOORD CLOWN

*** GALGJE ***

BEURT 13

! CLOWN !

XXXXXXXXXXXXXXXXXXXXX
I X
I O
I XXXXXXXXXX
I X
I X
I
I
I

HET WOORD BESTAAT UIT 5 LETTERS.
DE FOUT GEBRUIKTE LETTERS WAREN :
YAHDIIKU

JA, HOOR DAT WAS HET WOORD
XXXXXXXXXXXXXXXXXXXXX
GOED HOOR, HET WOORD WAS INDERDAAD : CLOWN

2.10 Mastermind

Dit spel, dat overigens is afgeleid van het in Engeland bekende spel 'cows en bulls', is vrij algemeen bekend. Vandaar dat hier slechts een vrij summiere uitleg wordt gegeven. Het programma ontwikkelt een geheime cijfercode van vier verschillende cijfers. Het is de bedoeling dat de speler deze code breekt. Om dit te bereiken, geeft het programma aanwijzingen.


Is een cijfer juist, maar de plaats foutief dan geeft het programma dit aan met een '*'. Is zowel een cijfer als de plaats juist dan wordt dit aangegeven met een '#'. Om het wat gemakkelijker te maken, worden de laatste vijf pogingen met de resultaten op het beeldscherm gezet.

Suggestie 1: In deze uitvoering wordt uitgegaan van vier verschillende cijfers. Het is natuurlijk vrij eenvoudig dit te wijzigen in vier verschillende letters. Een andere mogelijkheid zou kunnen zijn dat cijfers twee keer voor mogen komen.

Suggestie 2: Men speelt dit spel tegen de computer. Het is natuurlijk interessant wanneer de computer ook tegen de speler speelt. Op het eerste gezicht lijkt dit een vrij moeilijke programma-ontwikkeling. Een aanzet kan zijn om bij zichzelf eens na te gaan hoe men bepaalt uit welke cijfers de code bestaat. Aan de hand van deze gedragslijn moet het mogelijk zijn een programma te ontwerpen. Het is wel aan te bevelen uit te gaan van een code van vier verschillende cijfers.

Programma

```
10 REM *** MASTERMIND ***
20 RANDOM
30 DIM A(5,8), B$(2), C(4), G(4)
40 GOTO 1360
50 :
60 REM *** SPELREGELS ***
70 CLS                                :REM SCHERM WISSEN
80 PRINT "*** MASTERMIND ***"        -SPELREGELS-"
90 PRINT
100 PRINT "DE COMPUTER MAAKT EEN GEHEIME CODE VAN VIER"
110 PRINT "VERSCHILLENDE CIJFERS. U MOET PROBEREN, DIE"
120 PRINT "RADEN. DE CIJFERS LIGGEN TUSSEN 0 EN 9. HEEFT"
130 PRINT "MEN EEN CIJFER GOED EN DE PLAATS VAN HET"
140 PRINT "CIJFER DAN GEEFT DE COMPUTER DIT AAN MET #."
150 PRINT "HEEFT MEN WEL EEN CIJFER GOED, MAAR NIET ZIJN"
160 PRINT "JUIST PLAATS DAN WORDT DIT AAN GEGEVEN MET *."
170 PRINT "VOOR HET GEMAK VAN DE SPELER WORDEN DE LAATSTE"
180 PRINT "VIJF GETALLEN GETOOND, DIE AL GEPROBEERD ZIJN."
190 PRINT
200 PRINT "NA ELK CIJFER DIENT EEN KOMMA TE STAAN !"
210 PRINT
220 PRINT "HEEFT MEN DIT GELEZEN, DRUK <RETURN> IN";
230 INPUT A$
240 RETURN
250 :
260 REM *** INITIALIZEREN ***
```

```

270 REM ** SCHOONVEGEN MAT A **
280 FOR I = 1 TO 5
290 FOR J = 1 TO 8
300 A(I,J) = 0
310 NEXT J
320 NEXT I
330 REM ** GEHEIME CODE AANMAKEN **
340 FOR I = 1 TO 4
350 X = INT(RND(0) * 10)
360 IF I = 1 THEN 460
370 REM ** CIJFER AL GEBRUIKT **
380 H = 0
390 FOR J = 1 TO I-1
400 IF X <> C(J) THEN 440
410 REM ** CIJFER KOMT VOOR **
420 H = 1
430 J = I
440 NEXT J
450 IF H <> 0 THEN 350
460 C(I) = X
470 NEXT I
480 REM ** STRINGMATRIX VULLEN **
490 B$(1) = " * "
500 B$(2) = " # "
510 REM ** STARTVARIABLEN **
520 B = 0
530 T = 0

```

:REM BEURT
:REM CONTROLE-TELLER

```

540 A$ = ""
550 RETURN
560 :
570 H = 0
580 REM *** UITVOER ***
590 CLS :REM SCHERM WISSEN
600 PRINT "*** MASTERMIND ***";
610 PRINT TAB(27) "BEURT :";B
620 PRINT TAB(27) "PLAATS EN CIJFER GOED: - # -"
630 PRINT TAB(27) "ALLEEN CIJFER GOED: - * -"
640 IF T = 0 THEN 740
650 FOR I = 1 TO T
660 PRINT TAB(3) A(I,1);A(I,2);A(I,3);A(I,4);TAB(15);"---";
670 FOR J = 5 TO 8
680 IF A(I,J) = 0 THEN 710
690 X = A(I,J)
700 PRINT B$(X);
710 NEXT J
720 PRINT
730 NEXT I
740 PRINT A$
750 RETURN
760 :
770 REM *** INVOER ***
780 PRINT "WAT IS DE CODE <VERGEET DE KOMMA'S NIET>";
790 INPUT G(1),G(2),G(3),G(4)
800 REM ** CONTROLE OP CIJFERS **
810 H = 0
820 FOR I = 1 TO 4
830 IF (G(I) >= 0) AND (G(I) <= 9) THEN 860
840 A$ = "CIJFERS INVOEREN ! GEEN GETALLEN."
850 H = 1
860 NEXT I
870 IF H = 1 THEN 970
880 FOR I = 1 TO 4
890 FOR J = 1 TO 4
900 IF (G(I) <> G(J)) OR (I = J) THEN 950
910 H = 1
920 I = 4
930 J = 4
940 A$ = "DE CIJFERS MOETEN VERSCHILLEND ZIJN !"
950 NEXT J
960 NEXT I
970 RETURN
980 :
990 REM *** BEREKENING ***
1000 T = T + 1
1010 IF T < 6 THEN 1100
1020 REM ** GETALLEN OPSCHUIVEN **
1030 FOR I = 1 TO 4
1040 FOR J = 1 TO 8
1050 A(I,J) = A(I+1,J)
1060 NEXT J
1070 NEXT I
1080 T = T - 1

```

```

1090 REM ** GOK OPNEMEN IN RIJ **
1100 FOR J = 1 TO 8
1110 A(T,J) = 0
1120 IF J > 4 THEN 1140
1130 A(T,J) = G(J)           :REM OVERNEMEN GOK
1140 NEXT J
1150 REM ** CONTROLE OP GETALLEN **
1160 L = 4                   :REM HULPVARIABLELE
1170 S = 0
1180 FOR I = 1 TO 4
1190 FOR J = 1 TO 4
1200 IF G(I) <> C(J) THEN 1270
1210 P = 1
1220 IF I <> J THEN 1250
1230 P = 2
1240 S = S + 2
1250 L = L + 1
1260 A(T,L) = P             :REM PLAATS/CIJFER
1270 NEXT J
1280 NEXT I
1290 IF S <> 8 THEN 1320
1300 A$ = "U HEEFT DE CODE GERADEN."
1310 H = 3
1320 RETURN
1330 :
1340 REM **** STUURMODULE ****
1350 REM *** SPELREGELS ***
1360 GOSUB 70
1370 REM *** INITIALIZEREN ***
1380 GOSUB 280
1390 B = B + 1               :REM BEURTTELLER + 1
1400 REM *** UITVOER ***
1410 GOSUB 590
1420 A$ = ""
1430 REM *** INVOER ***
1440 GOSUB 780
1450 IF H = 1 THEN 1390
1460 REM *** BEREKENING ***
1470 GOSUB 1000
1480 IF H < 3 THEN 1390
1490 REM *** UITVOER ***
1500 GOSUB 590
1510 REM *** RESULTAAT ***
1520 PRINT "U HAD IN TOTAAL";B;"BEURTEN NODIG."
1530 END

```

Voorbeeld

```

*** MASTERMIND ***           BEURT : 1
                               PLAATS EN CIJFER GOED: - # -
                               ALLEEN CIJFER GOED:   - * -
WAT IS DE CODE <VERGEET DE KOMMA'S NIET> 0 , 1 , 2 , 3

```

*** MASTERMIND *** BEURT : 2
PLAATS EN CIJFER GOED: - # -
ALLEEN CIJFER GOED: - * -

0 1 2 3 -- *
WAT IS DE CODE <VERGEET DE KOMMA'S NIET> 2 , 3 , 4 , 5

*** MASTERMIND *** BEURT : 3
PLAATS EN CIJFER GOED: - # -
ALLEEN CIJFER GOED: - * -

0 1 2 3 -- *
2 3 4 5 -- * # *
WAT IS DE CODE <VERGEET DE KOMMA'S NIET> 1 , 3 , 4 , 5

*** MASTERMIND *** BEURT : 4
PLAATS EN CIJFER GOED: - # -
ALLEEN CIJFER GOED: - * -

0 1 2 3 -- *
2 3 4 5 -- * # *
1 3 4 5 -- # *
WAT IS DE CODE <VERGEET DE KOMMA'S NIET> 6 , 5 , 5 , 2

*** MASTERMIND *** BEURT : 5
PLAATS EN CIJFER GOED: - # -
ALLEEN CIJFER GOED: - * -

0 1 2 3 -- *
2 3 4 5 -- * # *
1 3 4 5 -- # *
DE CIJFERS MOETEN VERSCHILLENDE ZIJN !
WAT IS DE CODE <VERGEET DE KOMMA'S NIET> 7 , 5 , 4 , 2

*** MASTERMIND *** BEURT : 6
PLAATS EN CIJFER GOED: - # -
ALLEEN CIJFER GOED: - * -

0 1 2 3 -- *
2 3 4 5 -- * # *
1 3 4 5 -- # *
7 5 4 2 -- # # #
WAT IS DE CODE <VERGEET DE KOMMA'S NIET> 8 , 5 , 4 , 2

*** MASTERMIND *** BEURT : 6
PLAATS EN CIJFER GOED: - # -
ALLEEN CIJFER GOED: - * -

0 1 2 3 -- *
2 3 4 5 -- * # *
1 3 4 5 -- # *
7 5 4 2 -- # # #
8 5 4 2 -- # # # #
U HEEFT DE CODE GERADEN.
U HAD IN TOTAAL 6 BEURTEN NODIG.

2.11 Vier op een rij

De spelregels van 'vier op een rij' zijn vrij eenvoudig. Het is de bedoeling vier schijven op een rij te krijgen. Dit geldt zowel horizontaal, verticaal als diagonaal.

Het spel wordt tegen de computer gespeeld. Het zou te ver voeren om dit programma tot de categorie kunstmatige intelligentie te rekenen. Het programma neemt wel beslissingen op basis van bepaalde spelsituaties. Zo wordt de positie van drie eigen symbolen op een rij anders gewaardeerd dan drie symbolen van de tegenstander. De werking van het programma kan men het beste volgen aan de hand van de listing. Alleen de waardering verdient een nadere uitleg.

Het programma bekijkt steeds de situatie van vier opeenvolgende vakken. Dit gebeurt horizontaal, verticaal, diagonaal van rechts naar links en diagonaal van links naar rechts. Dit is overigens meteen de verklaring waarom het zo lang duurt, voordat het programma met een zet komt. Het werkt eerst het hele speelbord af.

Kan het programma een zet doen waarbij hij vier eigen symbolen op een rij kan krijgen, dan wordt dit met de hoogste waardering gehonoreerd. Kan door een zet van het programma worden voorkomen dat de tegenstander vier op een rij maakt dan krijgt die zet de op één na hoogste zetwaardering. Dit gaat zo door voor drie en twee op een rij. Begint de computer dan wordt met een random-zet begonnen.

Veel strategische bordspellen die als programma verkrijgbaar zijn, werken ongeveer allemaal volgens dit principe. Een aardig spel dat in het verlengde ligt van 'vier op een rij' is Gomoko.

Gomoko wordt gespeeld op een grid van 19 bij 19. Men moet proberen vijf op een rij te krijgen. In elk boek over bordspellen staat een beschrijving van Gomoko.

Suggestie 1: Wanneer men 'vier op een rij' speelt, zal het opvallen dat het programma vrij lang moet nadenken. Dit is te optimaliseren. Hiertoe moet de scanmodule worden verbeterd.

Suggestie 2: Men kan experimenteren met de waardering van de zetten. Op deze manier krijgt men een aardig inzicht in wat er gebeurt. Het programma is in het begin van het spel niet zo sterk, dit komt omdat alleen de volgende zet wordt bepaald. Dit laatste is te verhelpen door niet één zet diep door te rekenen, maar bijvoorbeeld twee zetten. Er zal dan wel wat moeten worden gedaan aan de lange rekentijd.

Programma

```
10 REM *** VIER OP EEN RIJ ***
20 RANDOM
30 DIM A(6,7), B(7), C(7), D(2), W(6,2)
40 DATA 1, 0, 1, 7, 1, 3           :REM SCAN RECHTOMHOOG
50 DATA 0, 1, 1, 4, 1, 6           :REM SCAN HORIZONTAAL
60 DATA 1, 1, 1, 4, 1, 3           :REM SCAN DIAGONAAL L-R
70 DATA 1, -1, 4, 7, 1, 3          :REM SCAN DIAGONAAL R-L
80 DATA 9999, 3, -6666, 3, 100, 2
90 DATA -50, 2, 3, 1, -1, 1
100 GOTO 2030
110 :
120 REM *** SPELREGELS ***
130 CLS                               :REM SCHERM WISSEN
```

```

140 PRINT "*** VIER OP EEN RIJ ***"           -SPELREGELS -"
150 PRINT
160 PRINT "HET SPEELBORD BESTAAT UIT ZEVEN KOLOMMEN."
170 PRINT "DE SPELER EN DE COMPUTER GOOIEN OM BEURTEN"
180 PRINT "HUN SYMBOOL IN EEN VAN DE KOLOMMEN."
190 PRINT "HET IS DE BEDOELING VIER EIGEN SYMBOLEN OP EEN"
200 PRINT "RIJ TE KRIJGEN. DIT GELDT ZOWEL HORIZONTAAL,"
210 PRINT "VERTICAAL ALS DIAGONAAL. DEGENE, DIE DIT HET"
220 PRINT "EERST LUKT HEEFT GEWONNEN."
230 PRINT
240 PRINT "HET SYMBOOL VAN DE SPELER IS 'X'."
250 PRINT "HET SYMBOOL VAN DE COMPUTER IS 'O'."
260 PRINT
270 RETURN
280 :
290 REM *** WIE BEGINT ? ***
300 PRINT "WIL JIJ BEGINNEN JA/NEE :";
310 INPUT Q$
320 Q$ = LEFT$(Q$,1)
330 M = 2                               :REM SPELER BEGINT
340 IF Q$ <> "J" THEN 360
350 M = 1                               :REM COMPUTER BEGINT
360 RETURN
370 :
380 REM *** INITIALIZEREN ***
390 REM ** DUMMY LEZEN **
400 RESTORE                               :REM DATAPONTER VOORAAN
410 FOR I = 1 TO 24
420 READ X
430 NEXT I
440 REM ** VARIABELEN STARTWAARDE GEVEN **
450 FOR I = 1 TO 6
460 B(I) = 0
470 C(I) = 0                               :REM MATRIX MET KOLOMVULLING
480 REM ** WAARDERINGSMATRIX W VULLEN **
490 READ W(I,1), W(I,2)                   :REM WAARDERING INLEZEN
500 REM ** REGEL SPEELVELD WISSEN **
510 FOR J = 1 TO 7
520 A(I,J) = 0
530 NEXT J
540 NEXT I
550 Z = 1                               :REM ZETTENTELLER
560 A$ = "+-----+-----+-----+-----+-----+"
570 RETURN
580 :
590 REM *** ZET SPELER ***
600 PRINT "WELKE KOLOM :";
610 INPUT K
620 IF ( K < 0 ) OR ( K > 7 ) THEN 600
630 K = INT(K)                             :REM ZEKER EEN GEHEEL GETAL
640 IF C(K) > 6 THEN 600                   :REM KAN ER NOG SYMBOOL BIJ ?
650 C(K) = C(K) + 1                       :REM HOOGTE BIJWERKEN
660 P = C(K)
670 A(P,K) = 0                             :REM ZET PLAATSEN SPELER
680 Z = Z + 1                             :REM ZETTENTELLER VERHOGEN

```

```

690 RETURN
700 :
710 REM *** UITVOER SPEELVELD ***
720 CLS :REM SCHERM WISSEN
730 PRINT "*** VIER OP EEN RIJ ***";
740 PRINT TAB(35) "ZETNR.:";Z
750 PRINT
760 IF M < 2 THEN 780
770 PRINT "DE COMPUTER ZET IN KOLOM:";Y
780 PRINT TAB(10) A$ :REM AFDRUK SCHEIDINGSREGEL
790 FOR I = 6 TO 1 STEP -1
800 FOR J = 1 TO 7
810 K = (J - 1) * 4 + 10
820 PRINT TAB(K) "!";
830 B$ = " "
840 IF A(I,J) = 0 THEN 880
850 B$ = " 0 "
860 IF A(I,J) <> 9 THEN 880
870 B$ = " X "
880 PRINT B$;
890 NEXT J
900 PRINT TAB(38) "!"
910 NEXT I
920 PRINT TAB(10) A$ :REM AFDRUK SCHEIDINGSREGEL
930 PRINT "KOLOM 1 2 3 4 5 6 7"
940 PRINT
950 PRINT C$
960 RETURN
970 :
980 REM *** ZETBESLISSING ***
990 IF H > 2 THEN 1100 :REM SPELER GEWONNEN ?
1000 Y = 1
1010 FOR J = 2 TO 7
1020 IF B(Y) => B(J) THEN 1040
1030 Y = J
1040 NEXT J
1050 REM ** ZET **
1060 C(Y) = C(Y) + 1
1070 P = C(Y)
1080 A(P,Y) = 1 :REM COMPUTER ZET
1090 Z = Z + 1 :REM ZETTENTELLER VERHOGEN
1100 RETURN
1110 :
1120 REM *** AKTIE ***
1130 REM ** AANWIJZING = C$ **
1140 I = R :REM RELATIEVE START-REGEL
1150 J = K :REM RELATIEVE START-KOLOM
1160 FOR N = 1 TO 4
1170 IF A(I,J) <> 0 THEN 1280
1180 REM ** KAN PLAATS GEBRUIKT WORDEN **
1190 IF C(J) + 1 <> I THEN 1280
1200 REM ** PLAATS KAN GEBRUIKT WORDEN **
1210 B(J) = B(J) + ABS( W(G,1) ) :REM WAARDERING
1220 T = 1
1230 H = 1

```

```

1240 IF W(G,1) <> 9999 THEN 1280
1250 C$ = "IK HEB GEWONNEN, GOED HE !"
1260 N = 4
1270 H = 2
1280 I = I + V1
1290 J = J + V2
1300 NEXT N
1310 RETURN
1320 :
1330 REM *** SCAN ***
1340 REM ** AANWIJZING = C$ **
1350 FOR X = 1 TO 4
1360 IF A(I,J) = 0 THEN 1420
1370 Q = 1 :REM COMPUTER-AANDUIDING
1380 IF A(I,J) <> 9 THEN 1400
1390 Q = 2 :REM SPELER-AANDUIDING
1400 S = S + 1 :REM AANTAL BEZETTE PLAATSEN
1410 D(Q) = D(Q) + 1 :REM AANTAL SPELER OF COMP.
1420 I = I + V1 :REM REGELAANPASSING
1430 J = J + V2 :REM KOLOMAANPASSING
1440 NEXT X
1450 IF S = 0 THEN 1610
1460 REM ** VOORBEWERKING **
1470 H = 3
1480 C$ = "JE HEBT GEWONNEN. GOED HOOR !"
1490 IF D(2) = 4 THEN 1610
1500 C$ = ""
1510 H = 0
1520 FOR G = 1 TO 6
1530 P = 1 :REM AANDUIDING SPELER
1540 IF W(G,1) > 0 THEN 1560
1550 P = 2 :REM AANDUIDING COMPUTER
1560 IF D(P) <> W(G,2) THEN 1600
1570 REM ** AKTIE **
1580 GOSUB 1140
1590 G = 6
1600 NEXT G
1610 RETURN
1620 :
1630 REM *** SCAN-STURING ***
1640 T = 0
1650 RESTORE :REM DATAPONTER VOORAAN
1660 H = 0
1670 FOR I = 1 TO 7
1680 B(I) = 0
1690 NEXT I
1700 FOR E = 1 TO 4
1710 READ V1, V2, Y1, Y2, L1, L2 :REM STUURWAARDEN
1720 FOR K = Y1 TO Y2
1730 FOR R = L1 TO L2
1740 I = R :REM REGEL
1750 J = K :REM KOLOM
1760 D(1) = 0 :REM MAT D SCHOONVEGEN
1770 D(2) = 0
1780 S = 0 :REM PLAATSENTELLER = 0

```



```

1790 REM ** SCAN **
1800 GOSUB 1350
1810 IF H < 2 THEN 1860
1820 REM ** LUSSEN BEEINDIGEN **
1830 R = L2
1840 K = Y2
1850 E = 4
1860 NEXT R
1870 NEXT K
1880 NEXT E
1890 IF T <> 0 THEN 1920
1900 C# = "GELIJK SPEL !"
1910 H = 3
1920 RETURN
1930 :
1940 REM *** RANDOM-ZET PROGRAMMA ***
1950 Y = INT( RND(0) * 3 + 3 ) :REM RANDOM KOLOM
1960 C(Y) = C(Y) + 1 :REM VERWERKEN IN VULLING
1970 P = C(Y)
1980 A(P,Y) = 1 :REM ZET PLAATSEN
1990 RETURN
2000 :
2010 REM **** STUURMODULE ****
2020 REM *** SPELREGELS ***
2030 GOSUB 130
2040 REM *** INITIALIZEREN ***
2050 GOSUB 400
2060 REM *** WIE BEGINT ***
2070 GOSUB 300
2080 IF M <> 2 THEN 2120
2090 REM *** RANDOM ZET ***
2100 GOSUB 1950
2110 REM *** UITVOER SPEELVELD ***
2120 GOSUB 720
2130 REM *** ZET SPELER ***
2140 GOSUB 600
2150 M = 1 :REM SPELER UITVOER
2160 REM *** UITVOER SPEELVELD ***
2170 GOSUB 720
2180 M = 2 :REM PROGRAMMA UITVOER
2190 REM *** SCAN-STURING ***
2200 GOSUB 1650
2210 REM *** ZET-BESLISSING ***
2220 GOSUB 990
2230 IF H <2 THEN 2120
2240 REM *** UITVOER SPEELVELD ***
2250 GOSUB 720
2260 END

```

Voorbeeld

*** VIER OP EEN RIJ ***

ZETNR.: 1

KOLOM 1 2 3 4 5 6 7

WELKE KOLOM : 4

*** VIER OP EEN RIJ ***

ZETNR.: 2

				X			

KOLOM 1 2 3 4 5 6 7

*** VIER OP EEN RIJ ***

ZETNR.: 3

DE COMPUTER ZET IN KOLOM: 3

			O	X			

KOLOM 1 2 3 4 5 6 7

WELKE KOLOM : 5

De zetten 4 t/m 14 zijn niet afgedrukt.

*** VIER OP EEN RIJ ***

ZETNR.: 15

DE COMPUTER ZET IN KOLOM: 2

			O	O			
		O	X	X			
		X	X	X	O		
		O	O	X	X	O	

KOLOM 1 2 3 4 5 6 7

WELKE KOLOM :

3. Science Fiction-spellen

3.1 Maanlander

Dit programma is de voorloper van het programma 'Ruimtevaartuig'. De meeste suggesties die bij deze maanlander zouden kunnen worden gegeven, zijn in Ruimtevaartuig verwezenlijkt. Als u beide genoemde programma's naast elkaar legt, zal de overeenkomst blijken. Overigens kan men wel goed zien hoe men van het kleine naar het grote programma overgaat, zonder de structuur van het programma aan te tasten.

Het programma Maanlander stelt de gebruiker in staat een Lunar Module op de maan te zetten. Op dit moment is dat niet zo actueel meer, maar enkele jaren geleden was het iets spectaculairs.

Wanneer men zelf een dergelijk programma zou willen maken, kan men uit dit programma alle formules halen. Er wordt overigens bij deze formules van uitgegaan, dat er geen luchtweerstand is.

Dit programma geeft geen plaatje, maar uitsluitend de numerieke gegevens over de daling naar het maanoppervlak.

Ook hier moet men een bepaald tijdsinterval kiezen (0.1 t/m 100 s) alsmede de stuwdruk en de richting hiervan. Deze gegevens moeten in één keer worden ingevoerd, men moet de komma's tussen de gegevens dan ook niet vergeten.

Suggestie: Een automatisch landingsmechanisme is niet ingebouwd. Het is een aardige klus om dat er nog in te bouwen.

Programma

```
10 REM *** MAANLANDER ***
20 RANDOM
30 GOTO 1420
40 :
50 REM *** INSTRUCTIE ***
60 CLS                                :REM SCHERM WISSEN
70 PRINT "*** MAANLANDER ***"
80 PRINT
90 PRINT "DE BEDOELING IS DIT RUIMTEVAARTUIG BEHOUDEN OP"
100 PRINT "EEN VASTGESTELDE PLAATS TE LATEN LANDEN."
110 PRINT "HET RUIMTEVAARTUIG BEVINDT ZICH AL VOLLEDIG"
120 PRINT "BINNEN DE AANTREKKINGSKRACHT VAN DE MAAN."
130 PRINT
140 PRINT "HEEFT MEN DIT GELEZEN, DRUK DAN <RETURN> IN";
150 INPUT H$
160 CLS
170 PRINT "*** RUIMTEVAARTUIG ***   -BESTURING-"
180 PRINT
190 PRINT "1 DE STUWDruk IS TE REGELEN VAN 1 T/M 100 % VAN "
200 PRINT "  HET MOTORVERMOGEN."
210 PRINT "2 DE RICHTING WAARIN DE STUWDruk GEGEVEN KAN"
220 PRINT "  WORDEN, IS AAN TE GEVEN VAN -180 T/M 180"
```

```

230 PRINT " GRADEN. RECHT OMHOOG IS 0 GRADEN, 90 GRADEN"
240 PRINT " IS NAAR RECHTS -90 GRADEN IS NAAR LINKS EN"
250 PRINT " 180 EN -180 GRADEN IS RECHT OMLAAG. ALLE"
260 PRINT " TUSSENLIJGENDE WAARDEN ZIJN OOK MOGELIJK."
270 PRINT " 3 HET TIJDSINTERVAL MOET BIJ IEDERE BESTURINGS-"
280 PRINT " OPDRACHT GEGEVEN WORDEN. DE TIJDSEENHEID IS "
290 PRINT " SECONDEN."
300 PRINT
310 PRINT "HEEFT MEN DIT GELEZEN DRUK DAN <RETURN> IN";
320 INPUT H$
330 RETURN
340 :
350 REM *** INITIALIZEREN ***
360 REM ** VARIABELEN EEN STARTWAARDE GEVEN **
370 C = 3.14159 / 180 :REM GRADEN NAAR RADIAAL
380 G = 1.62 :REM GRAVITATIE MAAN
390 S = 0 :REM VERSTREKEN TIJD
400 T = 1 :REM TIJDSINTERVAL = 1SEC
410 K = 0 :REM HULPTELLER
420 Z = INT(RND(0)*50/G + 5) :REM WILLEKEURIGE TIJD
430 M = INT(RND(0)*500+1000) :REM TOTALE MASSA
440 M2 = INT(0.6 * M) :REM HOEVEELHEID BRANDSTOF
450 M1 = M - M2 :REM LEEGGEWICHT LANDER
460 F = INT( M*G*(RND(0)*3+5) ) :REM TOTALE STUWDRUK
470 A = (F/M - G) * (RND(0)+1) * (-1) :REM INIT. ACCELERATIE
480 Y = (INT(0.5*ABS(A)*Z^2/10))*10 :REM STARTHOOGTE
490 K = K + 1 :REM HULPTELLER
500 IF K < 10 THEN 520
510 Y = (RND(0) + 1) * 300
520 IF ( Y > 3000 ) OR ( Y < 300 ) THEN 420
530 X = (INT((RND(0)*Y/2 + Y/2)/10))*10 :REM LANDINGSPL.
540 B1 = ( M2/90 )*( RND(0) + 1 ) :REM BRANDSTOFFACTOR
550 R = INT(RND(0)*180 + 90 ) :REM RICHTING RUIMTEVAARTUIG
560 IF R => 180 THEN 580
570 X = X * (-1)
580 U = RND(0)*100 :REM PERCENTAGE STUWDRUK
590 V1 = 0 :REM STARTSNELHEID HOR.
600 V2 = -50 :REM STARTSNELHEID VERT.
610 RETURN
620 :
630 REM *** BEREKENING ***
640 REM ** STUWDRUK **
650 U = U / 100 :REM PERC. WORDT BREUK
660 F1 = U * F :REM ACTIEVE STUWDRUK
670 R1 = SIN ( R * C ) * F1 :REM HORIZONTALE DEEL
680 R2 = COS ( R * C ) * F1 :REM VERTICALE DEEL
690 V1 = V1 + T * (R1/M)/2 :REM HORIZ.GEM.SNELH.
700 V2 = V2 + T * (R2/M -G )/2 :REM VERT.GEM.SNELH.
710 V = (SQR(V1 ^ 2 + V2 ^ 2)) * 3.6 :REM SNELHEID KM/H
720 X = X + V1 * T :REM AFSTAND IN METERS
730 Y = Y + V2 * T :REM HOOGTE IN METERS
740 IF Y > 5 THEN 760 :REM GELAND HOE ?
750 Y = 0
760 S = S + T :REM AANTAL VLUCHTSEC.
770 B = (U^2*0.75+U*0.25)*T*B1 :REM BRANDSTOFVERBR.

```

```

780 M2 = INT(M2 - B)           :REM RESTANT BRANDSTOF
790 IF M2 > 0 THEN 840
800 REM ** BRANDSTOF UITGEPUT **
810 M2 = 0
820 F = 0
830 T = 1                       :REM TIJDSINT.= 1 SEC
840 M = M1 + M2                 :REM MASSA RUIMTEV.
850 RETURN
860 :
870 REM *** STRINGS WISSEN ***
880 H = 0
890 B$ = ""
900 C$ = ""
910 D$ = ""
920 RETURN
930 :
940 REM *** UITVOER ***
950 CLS
960 PRINT "*** MAANLANDER ***";
970 PRINT TAB(35) "VERSTREKEN TIJD :";S
980 PRINT
990 PRINT " HOOGTE           :";INT(Y);"M"
1000 PRINT " AFSTAND          :";INT(X);"M"
1010 PRINT " SNELHEID KM/H     :";INT(V)
1020 PRINT " HOR.SNELHEID M/S  :";INT(V1)
1030 PRINT " VERT.SNELHEID M/S :";INT(V2)
1040 PRINT " BRANDSTOF       :";INT(M2)
1050 PRINT B$
1060 PRINT C$
1070 PRINT D$
1080 PRINT
1090 RETURN
1100 :
1110 REM *** INVOER ***
1120 PRINT "TIJD, STUWR., RICHT.";
1130 PRINT">>STOP = -9, -9, -9<<AKTIE:";
1140 INPUT A1, A2, A3
1150 REM ** CONTROLE **
1160 IF A1 < 0 THEN 1180
1170 T = A1
1180 IF ( A2 < 0 ) OR ( A2 > 100 ) THEN 1200
1190 U = INT(A2)
1200 IF ( A3 < -180 ) OR ( A3 >180 ) THEN 1220
1210 R = A3
1220 RETURN
1230 :
1240 REM *** AANWIJZINGEN ***
1250 IF M2 > 0 THEN 1270
1260 B$ = " BRANDSTOF UITGEPUT"
1270 IF Y < 10000 THEN 1300
1280 D$ = " COMPUTER UITGEVALLEN !"
1290 H = 1
1300 IF Y > 0 THEN 1380
1310 C$ = " *** CRASH ***"
1320 IF V > 20 THEN 1340

```

```

1330 C$ = " VEILIG GELAND "
1340 D$ = " IN HET LANDINGSGBIED"
1350 IF ABS(X) < 10 THEN 1370
1360 D$ = " BUITEN HET LANDINGSGBIED"
1370 H = 1
1380 RETURN
1390 :
1400 REM ***** STUURMODULE *****
1410 REM *** INSTRUCTIE ***
1420 GOSUB 60
1430 REM *** INITIALIZEREN ***
1440 GOSUB 370
1450 REM *** STRINGS WISSEN ***
1460 GOSUB 880
1470 REM *** BEREKENING ***
1480 GOSUB 650
1490 REM *** AANWIJZINGEN ***
1500 GOSUB 1250
1510 REM *** UITVOER ***
1520 GOSUB 950
1530 IF H > 0 THEN 1580
1540 IF M2 = 0 THEN 1460           :REM VRIJE VAL
1550 REM *** INVOER ***
1560 GOSUB 1120
1570 IF (A1 <> -9) OR (A2 <> -9) THEN 1460
1580 END

```

Voorbeeld

*** MAANLANDER ***

VERSTREKEN TIJD : 25

HOOGTE : 300 M
AFSTAND : 60 M
SNELHEID KM/H : 183
HOR.SNELHEID M/S :-1
VERT.SNELHEID M/S :-51
BRANDSTOF : 40

TIJD,STUWDR.,RICHT.>>STOP = -9,-9,-9<<AKTIE: 1,45,100

*** MAANLANDER ***

VERSTREKEN TIJD : 26

HOOGTE : 247 M
AFSTAND : 61 M
SNELHEID KM/H : 187
HOR.SNELHEID M/S : 1
VERT.SNELHEID M/S :-53
BRANDSTOF : 37

TIJD,STUWDR.,RICHT.>>STOP = -9,-9,-9<<AKTIE: 1,-90,100

*** MAANLANDER ***

VERSTREKEN TIJD : 35

HOOGTE : 31 M
AFSTAND : 20 M
SNELHEID KM/H : 199
HOR.SNELHEID M/S : 1
VERT.SNELHEID M/S :-56
BRANDSTOF : 0
BRANDSTOF UITGEPUT


*** MAANLANDER ***

VERSTREKEN TIJD : 36

HOOGTE : 0 M
AFSTAND : 21 M
SNELHEID KM/H : 202
HOR.SNELHEID M/S : 1
VERT.SNELHEID M/S :-57
BRANDSTOF : 0
BRANDSTOF UITGEPUT
*** CRASH ***
BUITEN HET LANDINGSGBIED

3.2 Ruimtevaartuig

Het moederschip van een grote ruimtevaartexpeditie is uitgerust met een aantal ruimtesloepen, van het type HEX 32. Deze ruimtevaartuigen zijn uitermate geschikt voor landingen op de diverse planeten. De expeditie heeft ons eigen zonnestelsel als doelgebied. Afhankelijk van welke planeet wordt gekozen, kan het moederschip zich al binnen een tiental seconden in een lage parkeerbaan om de gekozen planeet bevinden. De astronauten maken zich gereed om aan boord te gaan van de HEX 32. Ondertussen worden de boordcomputers van de sloep met nieuwe programma's geladen en worden alle systemen voor de laatste keer getest. Nadat een HEX 32 is gelanceerd, start volautomatisch de raketmotor die er voor zorgt dat het ruimtevaartuig binnen de aantrekkingskracht van de planeet komt. De hele procedure staat onder controle van de eigen boordcomputers. Alhoewel het laatste gedeelte van de afdaling naar de planeet ook wel automatisch kan geschieden, wordt toch vaak op handbesturing overgegaan. De voornaamste reden is wel dat men vaak op lokaties landt waar nog niet alle gegevens van bekend zijn. Het zal niet de eerste keer zijn dat een HEX 32 in een pulserend magnetisch veld komt en daardoor onbestuurbaar wordt.



Het ruimtevaartuig wordt voor een bepaald tijdsinterval vooruit geprogrammeerd. Dit gebeurt altijd in seconden waarbij het tijdsinterval kan liggen tussen 0,1 en 100 seconden. Tegelijkertijd dient de astronaut de richting van de stuwdruk en de hoeveelheid stuwdruk op te geven. De stuwdruk wordt opgegeven als percentage van de maximale stuwdruk en de richting daarvan wordt opgegeven in graden. De boordcomputers houden bij wat de effecten zijn op het ruimtevaartuig. Hiertoe wordt een beeldscherm opgebouwd met daarin alle noodzakelijke gegevens.

Links op het beeldscherm wordt het vluchtbeeld weergegeven met de afstand tot de lan-

dingsplaats en de hoogte boven het planeetoppervlak. Rechts staan een aantal vluchtgegevens met daarin o.a. een gradenverdeling als geheugensteun voor de bepaling van de richting van de stuwdruk. Tevens worden de snelheid van het ruimtevaartuig, de verticale en de horizontale snelheid in m/s, de brandstofvoorraad, de hoogte en de afstand gegeven.

Het zal de eerste keer niet meevallen het ruimtevaartuig veilig op de grond te zetten. Mocht de brandstof opraken, dan slaat men zeker te pletter op het planeetoppervlak.

Suggestie 1: De planeten waaruit kan worden gekozen, zijn allemaal planeten uit ons eigen zonnestelsel. Het is best wel eens aardig om een planeet toe te voegen met een extreem hoge zwaartekracht. Bijvoorbeeld de planeet Silico(zie programma 3.5).

Suggestie 2: De opbouw van de uitvoer verloopt via twee tamelijke grote modules. Door met POKE-opdrachten te werken, kan men de opbouwmodule verwijderen en de feitelijke uitvoermodule aanmerkelijk verkleinen. Het een en ander zal tot gevolg hebben dat het programma aanmerkelijk sneller zal draaien.

Suggestie 3: Wanneer men beschikt over statements waarmee men karakters direct van het toetsenbord kan opnemen (GET of INKEY\$), kan men een real-time game maken.

Programma

```
10 REM *** RUIMTEVAARTUIG ***
20 RANDOM
30 DATA PLUTO, 4.89, SATURNUS, 8.77, MARS, 3.84, AARDE, 9.81
40 DATA MAAN, 1.62, IO, 2.26, EUROPA, 1.98, CALLISTO, 3.20
50 DATA PALAS, 0.54, JUNO, 0.21
60 GOTO 2650
70 :
80 REM *** INSTRUCTIE ***
90 CLS :REM SCHERM WISSEN
100 PRINT "*** RUIMTEVAARTUIG ***"
110 PRINT
120 PRINT "DE BEDOELING IS DIT RUIMTEVAARTUIG VEILIG OP EEN"
130 PRINT "VASTGESTELDE PLAATS TE LATEN LANDEN (0,0). HET"
140 PRINT "RUIMTEVAARTUIG BEVINDT ZICH AL VOLLEDIG BINNEN"
150 PRINT "DE AANTREKKINGSKRACHT VAN DE GEKOZEN PLANEET."
160 PRINT
170 PRINT "HEEFT MEN DIT GELEZEN, DRUK DAN <RETURN> IN";
180 INPUT H$
190 CLS
200 PRINT "*** RUIMTEVAARTUIG *** --BESTURING--"
210 PRINT
220 PRINT "1 DE STUWDRUK IS TE REGELEN VAN 1 T/M"
230 PRINT " 100% VAN HET MOTORVERMOGEN."
240 PRINT "2 DE RICHTING WAARIN DE STUWDRUK GEGEVEN KAN"
250 PRINT " WORDEN, IS AAN TE GEVEN VAN -180 T/M 180"
260 PRINT " GRADEN. RECHT OMHOOG IS 0 GRADEN, 90 GRADEN"
270 PRINT " IS NAAR RECHTS, -90 GRADEN IS NAAR LINKS"
280 PRINT " EN 180 EN -180 GRADEN IS RECHT OMLAAG. "
290 PRINT " ALLE TUSSENLIJGENDE WAARDEN ZIJN OOK MOGELIJK."
300 PRINT "3 HET TIJDSINTERVAL MOET BIJ IEDERE BESTURINGS--"
```

```

310 PRINT "  OPDRACHT GEGEVEN WORDEN. DE TIJDSEENHEID IS "
320 PRINT "  SECONDEN."
330 PRINT
340 PRINT "HEEFT MEN DIT GELEZEN DRUK DAN <RETURN> IN";
350 INPUT H$
360 RETURN
370 :
380 REM *** INVOER 1 ***
390 CLS                               :REM SCHERM WISSEN
400 PRINT "*** RUIMTEVAARTUIG ***"
410 PRINT "OP DE VOLGENDE HEMELLIJCHAMEN";
420 PRINT "ZIJN LANDINGSPLAATSEN:"
430 PRINT
440 REM ** INLEZEN NAMEN HEMELLIJCHAMEN **
450 RESTORE
460 FOR I=1 TO 9 STEP 2
470 READ H$, G, A$, G
480 PRINT I;" = "; TAB(8); H$;
490 PRINT TAB(25) I+1; TAB(30) "= ";TAB(35) A$
500 NEXT I
510 REM ** KEUZE **
520 PRINT
530 PRINT "WELK HEMELLIJCHAAM  -GEEF KEUZENR.:";
540 INPUT K
550 IF ( K < 1 ) OR ( K > 10 ) THEN 390
560 RETURN
570 :
580 REM *** INITIALIZEREN ***
590 REM ** OPNIEUW LEZEN INTERNE DATAFILE **
600 RESTORE                           :REM DATAPONTER VOORAAN
610 FOR I = 1 TO K
620 READ H$, G                         :REM INLEZEN NAAM + GRAVITATIE
630 NEXT I
640 REM ** VARIABELEN EEN STARTWAARDE GEVEN **
650 C = 3.14159 / 180                  :REM GRADEN NAAR RADIAAL
660 S = 0                              :REM VERSTREKEN TIJD
670 T = 1                              :REM TIJDSINTERVAL = 1SEC
680 K = 0                              :REM HULPTELLER
690 Z = INT(RND(0)*50/G + 5)          :REM WILLEKEURIGE TIJD
700 M = INT(RND(0)*500 + 1000) :REM TOTALE MASSA
710 M2 = INT(0.6 * M)                 :REM HOEVEELHEID BRANDSTOF
720 M1 = M - M2                       :REM LEEGGEWICHT RUIMTEVAART.
730 F = INT( M*G*(RND(0)*3+5) ) :REM TOTALE STUWDRIJK
740 A = (F/M - G)*(RND(0)+1)*(-1) :REM INIT. ACCELERATIE
750 Y = (INT(0.5*ABS(A)*Z↑2/10))*10 :REM *STARTHOOGTE
760 K = K + 1                          :REM HULPTELLER
770 IF K < 10 THEN 790
780 Y = (RND(0) + 1) * 300
790 IF ( Y > 3000 ) OR ( Y < 300 ) THEN 690
800 X = (INT((RND(0)*Y/2 +Y/2)/10))*10 :REM  LANDINGSPL.
810 B1 = ( M2/90 )*( RND(0) + 1 )      :REM BRANDSTOFFACTOR
820 R = INT(RND(0)*180 + 90 )         :REM RICHTING RUIMTEVAART.
830 IF R => 180 THEN 850
840 X = X * (-1)
850 U = RND(0)*100                     :REM PERCENTAGE STUWDRIJK

```

```

860 V1 = 0 :REM STARTSNELHEID HOR
870 V2 = -50 :REM STARTSNELHEID VERTICAAL
880 RETURN
890 :
900 REM *** BEREKENING ***
910 REM ** STUWDruk **
920 U = U / 100 :REM PERC. WORDT BREUK
930 F1 = U * F :REM ACTIEVE STUWDruk
940 R1 = SIN (R * C) * F1 :REM HORIZONTALLE DEEL
950 R2 = COS (R * C) * F1 :REM VERTICALE DEEL
960 V1 = V1 + T * (R1/M)/2 :REM HORIZ.GEM.SNELH.
970 V2 = V2 + T * (R2/M -G )/2 :REM VERT.GEM.SNELH.
980 V = (SQR(V1^2 + V2^2))*3.6 :REM SNELHEID KM/H
990 X = X + V1 * T :REM AFSTAND IN METERS
1000 Y = Y + V2 * T :REM HOOGTE IN METERS
1010 IF Y > 5 THEN 1030 :REM GELAND HOE ?
1020 Y = 0
1030 S = S + T :REM AANTAL VLUCHTSEC.
1040 B = (U^2*0.75+U*0.25)*T*B1 :REM BRANDSTOFVERBR.
1050 M2 = INT(M2 - B) :REM RESTANT BRANDSTOF
1060 IF M2 > 0 THEN 1100
1070 M2 = 0
1080 F = 0
1090 T = 1 :REM TIJDSINT.= 1 SEC
1100 M = M1 + M2 :REM MASSA RUIMTEV.
1110 RETURN
1120 :
1130 REM *** STRINGS WISSEN ***
1140 H = 0
1150 B$ = ""
1160 C$ = ""
1170 D$ = ""
1180 E$ = ""
1190 F$ = ""
1200 RETURN
1210 :
1220 REM *** OPBOUWEN PLAATJE ***
1230 REM ** Y-WAARDEN **
1240 FOR I = 1 TO 5
1250 N = 10 / I :REM BOVENGRENS GRAFIEK
1260 Y1 = INT ( Y / N )
1270 IF Y1 > 0 THEN 1290
1280 I = 5
1290 NEXT I
1300 Y2 = INT(Y * 10 / N + 0.5) :REM Y-COORDINAAT
1310 Y3 = N / 10 :REM STAPGROOTTE
1320 REM ** X-WAARDEN **
1330 REM ** RUIMTEVAARTUIG DICHT BIJ LANDINGSPLAATS **
1340 K = ABS(X)
1350 X1 = -50 :REM ONDERGRENS X
1360 X3 = 4 :REM STAPGROOTTE
1370 X4 = INT (( X + 50 ) / 4 + 0.5 )
1380 IF K < 51 THEN 1580
1390 REM ** LANDINGSPLAATS VERDERWEG **
1400 FOR I = 2 TO 5

```

```

1410 K = ABS(X)
1420 X2 = 10*I                                :REM BOVENGRENS X
1430 K = INT ( K / X2 )
1440 IF K > 0 THEN 1560
1450 X1 = 0
1460 IF X > 0 THEN 1510
1470 REM * LANDINGSPLAATS LIGT LINKS *
1480 REM * GRENZEN WISSELEN *
1490 X1 = X2 * (-1)
1500 X2 = 0
1510 I = 5
1520 X3 = ABS( X2 - X1 ) / 25                 :REM STAPGROOTTE
1530 X4 = INT ( X / X3 + 0.5 )               :REM X-COORDINAAT
1540 IF X > 0 THEN 1560
1550 X4 = 25 + X4                             :REM VERSVHUIVEN
1560 NEXT I
1570 REM ** PLAATS IN PLAATJE **
1580 P = ( 10 - Y2 ) * 26                     :REM Y-COORDINAAT
1590 P = P + X4                                :REM X-COORDINAAT
1600 REM ** TEKSTEN IN PLAATJE **
1610 E$ = "( * 1 M)"
1620 IF Y < 1000 THEN 1670
1630 E$ = "( * 1.000 M)"
1640 N = N / 1000
1650 Y3 = Y3 / 1000
1660 REM ** X-AS **
1670 F$ = " (* 1 M)"
1680 IF ABS(X) < 1000 THEN 1750
1690 F$ = " (* 1.000 M)"
1700 X1 = X1 / 1000
1710 IF ABS(X1) > 0.9 THEN 1730
1720 X1 = 0
1730 X2 = X2 / 1000
1740 X3 = X3 / 1000
1750 RETURN
1760 :
1770 :
1780 REM *** AFDRUK RUIMTEVAARTUIG ***
1790 Q = Q + 1
1800 IF Q <> P THEN 1820
1810 A$ = "*"
1820 PRINT A$;
1830 RETURN
1840 :
1850 REM *** UITVOER ***
1860 CLS
1870 PRINT "*** RUIMTEVAARTUIG ***";
1880 PRINT TAB(35) "VERSTREKEN TIJD :";S
1890 REM ** PLAATJE **
1900 Q = -1                                    :REM LOKATIE-TELLER
1910 PRINT E$; TAB(35) H$
1920 FOR I = 0 TO 10
1930 K = N - Y3 * I                            :REM HOOGTE AANDUIDING
1940 PRINT INT(K) ; TAB(7) "--";
1950 FOR J = 1 TO 5

```

```

1960 A$ = "+"
1970 GOSUB 1780
1980 FOR L = 1 TO 4
1990 A$ = "-"
2000 GOSUB 1780
2010 NEXT L
2020 NEXT J
2030 A$ = "+ "
2040 GOSUB 1780
2050 IF I <> 1 THEN 2070
2060 PRINT " HOOGTE      :";INT(Y);"M";
2070 IF I <> 2 THEN 2090
2080 PRINT " AFSTAND    :";INT(X);"M";
2090 IF I <> 3 THEN 2110
2100 PRINT " SNELHEID KM/H :";INT(V);
2110 IF I <> 4 THEN 2130
2120 PRINT " HOR.SNELHEID M/S :";INT(V1);
2130 IF I <> 5 THEN 2150
2140 PRINT " VERT.SNELHEID M/S :";INT(V2);
2150 IF I <> 6 THEN 2170
2160 PRINT " BRANDSTOF      :";INT(M2);
2170 IF I <> 7 THEN 2190
2180 PRINT B$;
2190 IF I <> 8 THEN 2210
2200 PRINT C$;
2210 IF I <> 9 THEN 2230
2220 PRINT D$;
2230 PRINT
2240 NEXT I
2250 REM ** AFDRUK ONDERSTE REGEL **
2260 FOR I = 0 TO 5
2270 Q = X1 + I * X3 * 5
2280 K = 7 + I * 5                      :REM AFSTAND--AANDUIDING
2290 PRINT TAB(K) Q;
2300 NEXT I
2310 PRINT F$
2320 RETURN
2330 :
2340 REM *** INVOER ***
2350 PRINT "TIJD, STUWDR., RICHT.";
2360 PRINT">>STOP = -9, -9, -9<<AKTIE:";
2370 INPUT A1, A2, A3
2380 REM ** CONTROLE **
2390 IF A1 < 0 THEN 2410
2400 T = A1
2410 IF ( A2 < 0 ) OR ( A2 > 100 ) THEN 2430
2420 U = INT(A2)
2430 IF ( A3 < -180 ) OR ( A3 >180 ) THEN 2450
2440 R = A3
2450 RETURN
2460 :
2470 REM *** AANWIJZINGEN ***
2480 IF M2 > 0 THEN 2500
2490 B$ = " BRANDSTOF UITGEPUT"
2500 IF Y < 10000 THEN 2530

```

```

2510 D$ = " COMPUTER UITGEVALLEN !"
2520 H = 1
2530 IF Y > 0 THEN 2610
2540 C$ = " *** CRASH ***"
2550 IF V > 20 THEN 2570
2560 C$ = " VEILIG GELAND "
2570 D$ = " IN HET LANDINGSGBIED"
2580 IF ABS(X) < 10 THEN 2600
2590 D$ = " BUITEN HET LANDINGSGBIED"
2600 H = 1
2610 RETURN
2620 :
2630 REM **** STUURMODULE ****
2640 REM *** INSTRUCTIE ***
2650 GOSUB 90
2660 REM *** INVOER 1 ***
2670 GOSUB 390
2680 REM *** INITIALIZEREN ***
2690 GOSUB 600
2700 REM *** STRINGS WISSEN ***
2710 GOSUB 1140
2720 REM *** BEREKENING ***
2730 GOSUB 920
2740 REM *** OPBOUWEN PLAATJE ***
2750 GOSUB 1240
2760 REM *** AANWIJZINGEN ***
2770 GOSUB 2480
2780 REM *** UITVOER ***
2790 GOSUB 1860
2800 IF H > 0 THEN 2850
2810 IF M2 = 0 THEN 2710           :REM VRIJE VAL
2820 REM *** INVOER ***
2830 GOSUB 2350
2840 IF (A1 <> -9) OR (A2 <> -9) THEN 2710
2850 END

```

Voorbeeld

*** RUIMTEVAARTUIG ***

OP DE VOLGENDE HEMELLIJCHAMEN ZIJN LANDINGSPLAATSEN:

1 =	PLUTO	2 =	SATURNUS
3 =	MARS	4 =	AARDE
5 =	MAAN	6 =	IO
7 =	EUROPA	8 =	CALLISTO
9 =	PALAS	10 =	JUNO

WELK HEMELLIJCHAAM -GEEF KEUZENR.: 2

*** RUIMTEVAARTUIG ***

(* 1.000 M)

VERSTREKEN TIJD : 1

SATURNUS

10	-+-----+-----+-----+-----+-----+	
9	-+-----+-----+-----+-----+-----+	HOOGTE : 1690 M
8	-+-----+-----+-----+-----+-----+	AFSTAND :-1138 M
7	-+-----+-----+-----+-----+-----+	SNELHEID KM/H : 250
6	-+-----+-----+-----+-----+-----+	HOR.SNELHEID M/S : 2
5	-+-----+-----+-----+-----+-----+	VERT.SNELHEID M/S :-70
4	-+-----+-----+-----+-----+-----+	BRANDSTOF : 682
3	-+-----+-----+-----+-----+-----+	
2	-+-----+-----+-----+-----+-----+	
1	-+-----+-----+-----+-----+-----+	
0	-+-----+-----+-----+-----+-----+	
	-10 -8 -6 -4 -2 0	(* 1.000 M)

TIJD, STUWDR., RICHT.>>STOP = -9, -9, -9<<AKTIE: 2, 90, 100

*** RUIMTEVAARTUIG ***

(* 1.000 M)

VERSTREKEN TIJD : 3

SATURNUS

10	-+-----+-----+-----+-----+-----+	
9	-+-----+-----+-----+-----+-----+	HOOGTE : 1511 M
8	-+-----+-----+-----+-----+-----+	AFSTAND :-1010 M
7	-+-----+-----+-----+-----+-----+	SNELHEID KM/H : 395
6	-+-----+-----+-----+-----+-----+	HOR.SNELHEID M/S : 64
5	-+-----+-----+-----+-----+-----+	VERT.SNELHEID M/S :-90
4	-+-----+-----+-----+-----+-----+	BRANDSTOF : 666
3	-+-----+-----+-----+-----+-----+	
2	-+-----+-----+-----+-----+-----+	
1	-+-----+-----+-----+-----+-----+	
0	-+-----+-----+-----+-----+-----+	
	-10 -8 -6 -4 -2 0	(* 1.000 M)

TIJD, STUWDR., RICHT.>>STOP = -9, -9, -9<<AKTIE:-9, -9, -9

3.3 De blijde


De blijde werd niet alleen op de planeet Aarde als wapentuig gehanteerd, maar zoals uit het Galactisch jaarboek blijkt was het op de planeet Dzijn van het stelsel 3C236 ook al een geliefd wapen om stamvetes mee uit te vechten.

Een blijde is een groot soort katapult waarmee stenen van enkele tonnen naar de tegenstander kunnen worden geslingerd. Op Dzijn heeft men de blijde vergaand geperfectioneerd. Men is nu in staat om de werphoek en ook de werpkracht exact in te stellen. Het gevolg is dat de trefkans zeer hoog wordt, wanneer men zo'n blijde laat instellen door een 'Dzijnball'.

Direct na Dzijnwende (zoiets als zonnwende) worden de blijdewedstrijden gehouden, vete of geen vete. Deze wedstrijden worden gehouden in het Kooiland. Een vlak gebied met in het midden een inslagkratertje. Met het kratertje als middelpunt wordt een roos aangebracht, net zoals op Aarde bij het dartspel. Kan men zijn steen in het kratertje werpen dan krijgt men 100 punten. Komt men in een van de buitenste ringen dan krijgt men minder punten.

Degene met het hoogste aantal punten heeft gewonnen en is daarmee Dzijnball (een soort schutterskoning).

Om een eerlijk spelverloop te krijgen, staat de blijde bij elke speler op een andere afstand.



Suggestie 1: In het programma wordt op het einde een scorebord getoond. Het zou leuk zijn, wanneer de scores van de vorige spelers in het beeldscherm worden getoond als de volgende speler aan de beurt is.

Het scorebord kan dan na afloop van het spel worden uitgebreid met een stukje programma om de winnaar aan te wijzen.

Suggestie 2: Wanneer men over PEEK- en POKE-instructies beschikt, kan men de worp in het plaatje POKEn met de rest van de gegevens.

Hierdoor wordt het beeld aanmerkelijk rustiger. Een bijkomend voordeel is dat het programma kleiner wordt.

Het programma geeft alleen een plaatje van 'bovenaf'. Met een kleine uitbreiding en enkele formules kan men de baan van de steen op het beeldscherm zetten. Op zo'n manier krijgt men een plaatje van opzij.

Suggestie 3: Het programma vraagt om de hoek en de kracht waarmee de steen de blijde moet verlaten. Moeilijker wordt het als ook de richting van de worp in het programma wordt verwerkt. Als informatie voor de speler moet het programma de afstand van de inslag tot de roos geven en de richting van de inslag ten opzichte van de roos.

Deze extra mogelijkheid geeft meer dimensie aan het spel.

Programma

```
10 REM *** DE BLIJDE ***
20 RANDOM
30 DIM A(10)
40 GOTO 1400
50 :
60 REM *** INSTRUCTIES ***
70 CLS                                :REM SCHERM WISSEN
80 PRINT "*** DE BLIJDE *** -INSTRUCTIES-"
90 PRINT
100 PRINT "JAARLIJKS NA DZJINWENDE WORDEN DE BLIJDE--"
110 PRINT "WEDSTRIJDEN GEHOUDEN TUSSEN VERSCHILLENDE STAM--"
120 PRINT "MEN. U BENT AANVOERDER VAN EEN VAN DE PLOEGEN."
130 PRINT "DE BLIJDE IS EEN SOORT KATAPULT, WAARMEE"
140 PRINT "STENEN NAAR DE VIJAND WORDEN GESCHOTEN."
150 PRINT "DE WEDSTRIJDEN ZIJN WAT VRIENDELIJKER."
160 PRINT "ER WORDT UITSLUITEND OM DE EER GESPEELD."
170 PRINT "VOOR DEZE WEDSTRIJD HEEFT MEN OP EEN VLAK"
180 PRINT "TERREIN EEN GROTE ROOS AANGEBRACHT."
190 PRINT
200 PRINT "DRUK <RETURN> IN ALS U DIT GELEZEN HEEFT. ";
210 INPUT Q$
220 CLS                                :REM SCHERM WISSEN
230 PRINT "*** DE LEPELBLIJDE *** -INSTRUCTIES-"
240 PRINT
250 PRINT "DE BEDOELING IS EEN STEEN VAN VAST GEWICHT IN"
260 PRINT "IN EEN KLEIN KRATERJE TE SCHIETEN."
270 PRINT "HET KRATERJE WORDT AANGEGEVEN MET -X-."
280 PRINT "OM DE BLIJDE TE KUNNEN RICHTEN MOET U TWEE"
290 PRINT "OPDRACHTEN GEVEN:"
300 PRINT
310 PRINT " 1. DE HOEK, WAARONDER DE STEEN DE BLIJDE VERLAAT."
320 PRINT " 2. EEN PERCENTAGE VAN DE MAXIMALE WERPKRACHT."
330 PRINT
```

```

340 PRINT "DE AANVOERDER MET DE MEESTE PUNTEN IS WINNAAR."
350 PRINT "ELKE BEURT BESTAAT UIT VIJF WORPEN."
360 PRINT "DIT SPEL IS MET MAXIMAAL 10 PERSONEN TE SPELEN ."
370 PRINT "MET HOEVEEL SPELERS SPEELT U ";
380 INPUT Q
390 IF (Q <= 0) OR (Q > 10) THEN 220
400 RETURN
410 :
420 REM *** INITIALISEREN ***
430 REM ** SCHOONVEGEN MAT A **
440 FOR I = 1 TO 10
450 A(I) = 0
460 NEXT I
470 REM ** VARIABELEN STARTWAARDE GEVEN **
480 C = 3.14159277 / 180 :REM GRADEN - RADIALEN
490 D$ = "! ! ! X ! ! !"
500 RETURN
510 :
520 REM *** UITVOER ***
530 CLS
540 PRINT "*** DE BLIJDE ***";
550 PRINT TAB(40) "SPELER :";I
560 PRINT TAB(40) "BEURT :";B
570 PRINT "AANTAL GESCOORDE PUNTEN :";A(I)
580 PRINT F$;TAB(25)";";ABS(L);
590 PRINT TAB(35) "+05-----+";
600 PRINT "HOEK, WAARONDER AFGEVUURD:";H;
610 PRINT TAB(35) "! +10-----+ !";
620 PRINT "PERCENTAGE WERPKRACHT :";P;
630 PRINT TAB(35) "! ! +20-----+ ! !";
640 PRINT TAB(35) "! ! ! ! ! !";
650 PRINT "<BLIJDE>----->"; TAB(35) Z$
660 PRINT TAB(35) "! ! ! ! ! !";
670 PRINT TAB(35) "! ! +-----20+ !";
680 PRINT TAB(35) "! ! +-----10+ !";
690 PRINT TAB(35) "+-----05+";
700 PRINT
710 RETURN
720 :
730 REM *** INVOER ***
740 PRINT "WAT IS DE HOEK VAN DE WORP <1-89> :";
750 INPUT H
760 IF (H < 1) OR (H > 89) THEN 740
770 PRINT "WAT IS HET PERCENTAGE WERPKRACHT :";
780 INPUT P
790 IF (P < 0) OR (P > 100) THEN 770
800 RETURN
810 :
820 REM *** VOLGENDE SPELER ***
830 CLS
840 PRINT "*** DE BLIJDE ***"
850 FOR X = 1 TO 5
860 PRINT
870 NEXT X
880 PRINT TAB(20) "SPELER";I;"GEREED.";

```

```

890 INPUT Q$
900 PRINT
910 REM ** STARTWAARDEN **
920 H = INT(RND(0) * 40 + 30)      :REM HOEK
930 V = INT(RND(0) * 10 + 10) * 10 :REM SNELHEID
940 P = INT(RND(0) * 40 + 40)
950 S = V * P * 0.01
960 A = S↑2 * SIN(2 * H * C)      :REM AFSTAND
970 RETURN
980 :
990 REM *** BEREKENING ***
1000 S = V * P/100
1010 R = S↑2 * SIN(2 * H * C)     :REM AFSTAND
1020 L = INT(A - R + 0.5)         :REM AFSTAND TOT KRATER
1030 F$ = "ACHTER HET DOEL"
1040 IF L < 0 THEN 1060
1050 F$ = "VOOR HET DOEL"
1060 IF ABS(L) > 75 THEN 1090
1070 F$ = "MIDDEN IN DE ROOS"
1080 L = 0
1090 X = ABS(L)
1100 Y = 100
1110 IF X = 0 THEN 1190
1120 Y = 20
1130 IF X <= 375 THEN 1190
1140 Y = 10
1150 IF X <= 675 THEN 1190
1160 Y = 5
1170 IF X <= 975 THEN 1190
1180 Y = 0
1190 Z$ = D$
1200 A(I) = A(I) + Y              :REM PUNTEN
1210 REM ** PLAATS IN DE ROOS **
1220 F = INT(L/75) * (-1) + 12
1230 IF (F < 1) OR (F > 25) THEN 1250
1240 Z$ = LEFT$(D$,F) + "*" + RIGHT$(D$,24-F)
1250 RETURN
1260 :
1270 REM *** RESULTAAT ***
1280 CLS
1290 PRINT "*** DE BLIJDE ***"
1300 PRINT "UITSLAG VAN DE WEDSTRIJD"
1310 PRINT "AANTAL DEELNEMERS :";Q
1320 PRINT
1330 FOR I = 1 TO Q
1340 PRINT "AANVOERDER :";I;TAB(18) A(I);TAB(28)"PUNTEN
1350 NEXT I
1360 RETURN
1370 :
1380 REM **** STUURMODULE ****
1390 REM *** INSTRUCTIE ***
1400 GOSUB 70
1410 REM *** INITIALIZEREN ***
1420 GOSUB 440
1430 FOR I = 1 TO Q

```

```
1440 REM *** VOLGENDE SPELER ***
1450 GOSUB 830
1460 FOR B = 1 TO 5
1470 REM *** INVOER ***
1480 GOSUB 740
1490 REM *** BEREKENING ***
1500 GOSUB 1000
1510 REM *** UITVOER ***
1520 GOSUB 530
1530 NEXT B
1540 REM ** WACHTLUS **
1550 FOR Y = 1 TO 700
1560 NEXT Y
1570 NEXT I
1580 REM *** RESULTAAT ***
1590 GOSUB 1280
1600 END
```


Voorbeeld

*** DE BLIJDE ***

AANTAL GESCOORDE PUNTEN : 0
ACHTER HET DOEL : 3713
HOEK, WAARONDER AFGEVUURD : 45
PERCENTAGE WERPKRACHT : 75

<BLIJDE>----->

SPELER : 1
BEURT : 1




WAT IS DE HOEK VAN DE WORP <1-89> : 45
WAT IS HET PERCENTAGE WERPKRACHT : 55

*** DE BLIJDE ***

AANTAL GESCOORDE PUNTEN : 10
VOOR HET DOEL : 681
HOEK, WAARONDER AFGEVUURD : 45
PERCENTAGE WERPKRACHT : 55

<BLIJDE>----->

SPELER : 1
BEURT : 3




WAT IS DE HOEK VAN DE WORP <1-89> : 45
WAT IS HET PERCENTAGE WERPKRACHT : 59

*** DE BLIJDE ***

AANTAL GESCOORDE PUNTEN : 30
ACHTER HET DOEL : 89
HOEK, WAARONDER AFGEVUURD : 45
PERCENTAGE WERPKRACHT : 59

<BLIJDE>----->

SPELER : 1
BEURT : 4



3.4 De Compusaurus

In het Dinosgebergte ten zuidwesten van de Voorburgen huist de Compusaurus. Ooit heeft iemand eens geprobeerd zo'n wezen te vangen, maar dat is fataal afgelopen.

De Compusaurus is een bijzonder vraatzuchtig monster dat de passen in het Dinosgebergte onveilig maakt voor reizigers die er doorheen moeten trekken. De passen zijn het jachtgebied van de Compusaurus maar het aantal passen in het Dinosgebergte is zó groot en ze lijken zó veel op elkaar dat de Compusaurus nooit weet bij welke pas hij is.

Is een reiziger een pas eenmaal gepasseerd dan is hij veilig. Niemand weet hoe het komt, maar als een reiziger aan het begin van een pas staat, duikt altijd aan het andere einde de Compusaurus op.

Een pas lijkt wel wat op een speelbord met stapschijven. Zowel de Compusaurus als de reiziger kunnen per stap maar maximaal 8 schijven oversteken. Door een speling van de natuur zijn sommige stapschijven tevens de deksels van valkuilen. Trapt men op zo'n schijf, dan vliegt men een tunnel in en men komt op een andere plaats weer boven. Omdat na elke jacht de Compusaurus alle valkuilen weer vergeet, kan ook hij op zo'n valkuildeksel terecht komen !

De Compusaurus staat de reiziger op schijf 100 op te wachten. Kan de reiziger eerder op schijf 100 komen, dan de Compusaurus op 1, dan ligt de weg naar de vrijheid open. De Compusaurus is ondanks alles 'sportief', hij geeft iedereen een eerlijke kans, alleen is hij wel verduiveld slim. Wanneer de reiziger of hijzelf een stap doet, let hij donders goed op wat er gebeurt en onthoudt dit ook.

In de loop van het spel zal hij zich vaak afvragen of hij de reiziger te slim af kan zijn door hem te 'slaan'. De Compusaurus is vaak overmoedig. Juist op die momenten kan hij door de reiziger buiten gevecht worden gesteld.

Suggestie 1: Het is tamelijk vervelend iedere keer het gehele beeldscherm opnieuw op te bouwen voor slechts een kleine wijziging. Het verdient hier zeker aanbeveling om met behulp van POKE-instructies de wijzigingen in het spelverloop aan te brengen. Als men de wijzigingen in het programma heeft aangebracht, zal het programma korter worden en sneller lopen.

Suggestie 2: De beslissingen van de Compusaurus zijn afhankelijk van het doorlopen van de beslissingsmodules. Wanneer men in de volgorde, waarin de beslissingsmodules worden doorlopen veranderingen aanbrengt, zal ook het gedrag veranderen.

Suggestie 3: Het kan misschien erg leuk zijn nog een Compusaurus in het spel te betrekken. Het speelveld zal dan wel drastisch vergroot moeten worden.

Programma

```
10 REM *** DE JAGENDE COMPUSAURUS ***
20 RANDOM
30 DIM A(100), B(100)
40 REM ** INTERNE DATA-FILE MET JACHTGEBIED **
50 DATA 20, 29, 16, 29, 4, 29
60 DATA 4, 29, 16, 29, 20, 29
70 GOTO 2560
80 REM *** SPELREGELS ***
```

```

90 CLS                                :REM SCHERM WISSEN
100 PRINT "*** DE COMPUSAURUS ***"
110 PRINT
120 PRINT "MEN IS TERECHT GEKOMEN IN HET JACHTGEBIED VAN"
130 PRINT "DE COMPUSAURUS. EEN VRAATZUCHTIGE BEEST MET"
140 PRINT "UITSTEKENDE OGEN EN EEN BIJZONDER "
150 PRINT "GOED GEHEUGEN. JUIST DIT GEHEUGEN MAAKT"
160 PRINT "HEM LEVENSGEVAARLIJK."
170 PRINT "DE SPELER WORDT VOORGESTELD DOOR - * -,"
180 PRINT "DE COMPUSAURUS DOOR - # -."
190 PRINT
200 PRINT"Druk <RETURN> IN, WANNEER MEN KLAAR IS";
210 INPUT T$
220 CLS                                :REM SCHERM WISSEN
230 PRINT "*** DE COMPUSAURUS ***"
240 PRINT
250 PRINT "MEN ZAL MOETEN PROBEREN VAN 1 NAAR 100"
260 PRINT "TE KOMEN. DE COMPUSAURUS PROBEERT HET"
270 PRINT "TEGENOVERGESTELDE TE DOEN, DUS VAN 100 NAAR 1."
280 PRINT "PER ZET KAN MEN MAXIMAAL 8 STAPPEN VOOR- OF"
290 PRINT "ACHTERUIT. ACHTERUIT GEEFT MEN MET EEN MIN-"
300 PRINT "TEKEN. AAN. DE COMPUSAURUS KAN VERSLAGEN"
310 PRINT "WORDEN, DOOR MET HET AANTAL TE NEMEN STAPPEN"
320 PRINT "PRECIES OP ZIJN PLAATS UIT TE KOMEN."
330 PRINT "DE COMPUSAURUS DOET DAT BIJ JOU OOK. LET "
340 PRINT "VOORAL OP DE TUNNELS ! HET IS MAAR EEN HINT."
350 PRINT
360 PRINT"Druk <RETURN> IN, WANNEER MEN KLAAR IS";
370 INPUT T$
380 RETURN
390 REM *** INITIALIZEREN ***
400 FOR I=1 TO 100
410 A(I) = I                                :REM TUNNELS WISSEN
420 B(I) = 999                              :REM GEHEUGEN C-SAURUS WISSEN
430 NEXT I
440 REM ** TUNNELS PLAATSEN ***
450 X = INT(RND(0)*60 +25) :REM RANDOM AANTAL TUNNELS
460 FOR Y = 1 TO X
470 I = INT(RND(0)*90+5) :REM PLAATS TUNNELS
480 Z = INT(RND(0)*30+1) :REM RELATIEVE VERPLAATSING
490 W = INT(RND(0)*1+1.5) :REM WAARDE VOOR POS OF NEG
500 IF W > 1 THEN 520
510 Z = Z * (-1) :REM TERUG !!
520 A(I) = I + Z
530 REM ** CONTROLE OP REL.VERPLAATSING **
540 IF ( A(I) < 2 ) OR ( A(I) > 99 ) THEN 480
550 NEXT Y
560 REM ** VARIABELEN STARTWAARDE GEVEN **
570 A = -1 :REM ZETTENTELLER SPELER
580 B = 0 :REM ZETTENTELLER COMPUSAURUS
590 D = 1 :REM STARTPLAATS SPELER
600 C = 100 :REM STARTPLAATS COMPUSAURUS
610 E = 0 :REM VORIGE ZET C-SAURIS
620 REM ** TELLER **
630 GOSUB 670

```

```

640 RETURN
650 :
660 REM *** SPELER TELLER + STRING WISSEN ***
670 A = A + 1 :REM ZETTENTELLER PLUS 1
680 F = 0 :REM STUURVAR. VOOR UITVOER
690 T$ = ""
700 RETURN
710 :
720 REM *** C-SAURUS TELLER + STRINGS WISSEN ***
730 B = B + 1 :REM OPHOGEN TELLER
740 F = 1 :REM STUURVAR. UITVOER
750 T$ = "" :REM STRING WISSEN
760 RETURN
770 :
780 REM *** UITVOER ***
790 CLS
800 T = 0
810 PRINT "*** HET JACHTGEBIED VAN DE COMPUSAURUS ***"
820 PRINT TAB(4) "AANTAL ZETTEN COMPUSAURUS:";B
830 PRINT TAB(4) "AANTAL ZETTEN SPELER :";A
840 PRINT
850 RESTORE :REM INTERNE DATAPONTER
860 REM ** UITPRINTEN JACHTGEBIED **
870 FOR J = 1 TO 6
880 READ W,Z
890 PRINT TAB(W)" ";
900 FOR I= W TO Z
910 T = T + 1 :REM CONTROLE-TELLER
920 IF ( D = T ) OR ( C = T ) THEN 940
930 PRINT "0";
940 IF D <> T THEN 960
950 PRINT "*";
960 IF C <> T THEN 980
970 PRINT "#";
980 IF I = Z THEN 1000
990 PRINT "-";
1000 NEXT I
1010 PRINT
1020 NEXT J
1030 PRINT T$ :REM AANWIJZING
1040 IF F < 1 THEN 1060
1050 PRINT "IK HEB ";G;"STAPPEN GEDAAN."
1060 PRINT
1070 RETURN
1080 :
1090 REM *** INVOER-SPELER ***
1100 PRINT "HOEVEEL STAPPEN WIL JE DOEN <MAX.8 STAPPEN>";
1110 INPUT K
1120 K = INT(K)
1130 RETURN
1140 :
1150 REM *** BEREKENING SPELER ***
1160 REM ** AANWIJZING = T$ **
1170 H = 0 :REM STUURVARIABELE
1180 IF ABS(K) < 9 THEN 1220

```



```

1190 T$ = "JE MOET DE BOEL NIET PROBEREN TE BEDONDEREN !!"
1200 A = A + 10 :REM STRAFPUNTEN
1210 K = D * (-1) +1 :REM SPELER TERUG OP START
1220 D = D + K :REM GEWENSTE LOKATIE
1230 IF D > 1 THEN 1250
1240 D = 1
1250 IF D < 101 THEN 1290
1260 T$ = "JE ZIT AAN HET BEGIN, JE MOET WEL UITKIJKEN."
1270 D = D - 100
1280 REM ** COMPUSAURUS KIJKT NAAR SPELER **
1290 B(D) = A(D) :REM C-SAURUS KIJKT OOK
1300 IF A(D) = D THEN 1340
1310 T$= "ZO DAT SCHIET OP !!"
1320 IF A(D) > D THEN 1340
1330 T$ = "HAASTIGE SPOED IS ZELDEN GOED."
1340 D = A(D) :REM UITEINDELIJKE LOKATIE
1350 IF D < 100 THEN 1380
1360 T$ = "JE HEBT HET EINDPUNT BEREIKT, GEFELICITEERD !!"
1370 H = 2 :REM STUURVARIABLE
1380 IF D <> C THEN 1420
1390 T$ = "** JE HEBT DE COMPUSAURUS TE PAKKEN **"
1400 H = 1 :REM STUURVARIABLE
1410 C = 0 :REM C-SAURUS UIT HET SPEL
1420 RETURN
1430 :
1440 REM *** GEWONNEN OP '1' ***
1450 IF C > 1 THEN 1490
1460 T$ = "IK ZIT OP '1', IK HEB GEWONNEN JAMMER HE !!"
1470 H = 2 :REM STUURVARIABLE
1480 C = 1
1490 RETURN
1500 :
1510 REM *** AANVAL COMPUSAURUS ***
1520 REM ** AANWIJZING = T$ **
1530 REM ** DIRECTE - + INDIRECTE AANVAL **
1540 H = 0 :REM STUURVARIABLE
1550 Y = C
1560 FOR I = Y-8 TO Y+8 :REM MOGELIJKE STAPPEN C-SAURUS
1570 IF ( I < 1 ) OR ( I > 100 ) THEN 1760
1580 IF ( D <> I ) OR ( B(I) <> 999 ) THEN 1660
1590 REM ** COMPUSAURUS GOKT **
1600 T$ = "IK DACHT JE TE GRIJPEN,MAAR ER WAS EEN TUNNEL!"
1610 H = 1 :REM STUURVARIABLE
1620 G = I - C :REM AANTAL STAPPEN C-SAURUS
1630 C = A(I) :REM PLAATS WAAR SAURUS UITKOMT
1640 B(I) = A(I) :REM GEHEUGEN VULLEN
1650 IF A(I) <> I THEN 1740
1660 IF B(I) <> D THEN 1760
1670 G = I - C :REM AANTAL STAPPEN C-SAURUS
1680 C = A(I) :REM PLAATS WAAR SAURUS UITKOMT
1690 D = 0 :REM SPELER UIT SPEL
1700 H = 2
1710 T$ = "IK HEB JE TE PAKKEN !!"
1720 IF B(I) = I THEN 1740
1730 T$ = "IK HEB JE TE PAKKEN VIA EEN TUNNEL,GEMEEN HE !"

```

```

1740 GOSUB 1450
1750 I = Y + 8
1760 NEXT I
1770 RETURN
1780 :
1790 REM *** COMPUSAURUS HINDERLAAG ***
1800 REM ** C-SAURUS PROBEERT HINDERLAAG **
1810 H = 0
1820 Y = C
1830 FOR I= Y-8 TO Y+8
1840 IF ( I < 1 ) OR ( I > 100 ) THEN 1970
1850 IF ( B(I) = 999 ) OR ( B(I) = I ) THEN 1970
1860 REM ** LOKATIE IN GEHEUGEN BEKEND **
1870 X = B(I)
1880 Z = ABS( X - D )           :REM AFSTAND SPELER - C-SAURUS
1890 IF Z > 8 THEN 1970
1900 REM ** COMPUSAURUS IN BEREIK SPELER **
1910 IF ( B(X) = 999 ) OR ( B(X) = X ) THEN 1970
1920 REM ** HINDERLAAG LEGGEN **
1930 G = I - C                 :REM AANTAL STAPPEN
1940 C = A(I)                 :REM LOKATIE C-SAURUS
1950 I = Y + 8
1960 H = 1
1970 NEXT I
1980 RETURN
1990 :
2000 REM *** COMPUSAURUS PROBEERT NAAR 1 TE KOMEN ***
2010 V = 999                   :REM EXTREME VERGELIJKINGSWAARDE
2020 Y = C
2030 FOR I = Y-8 TO Y+8

```



```

2040 IF ( I < 1 ) OR ( I > 100 ) THEN 2140
2050 IF ( B(I) = 999 ) THEN 2140
2060 IF ( B(I) = I ) OR ( B(I) >= C ) THEN 2140
2070 IF ABS( B(I) - D ) < 9 THEN 2140
2080 IF E = I THEN 2140
2090 REM ** WISSELEN **
2100 IF B(I) > V THEN 2140
2110 H = 1
2120 V = B(I)
2130 X = I
2140 NEXT I
2150 IF H = 0 THEN 2200
2160 G = X - C :REM AANTAL STAPPEN
2170 C = V
2180 GOSUB 1450
2190 E = X :REM ZET ONTHOUDEN
2200 RETURN
2210 :
2220 REM *** COMPUSAURUS GOKT NAAR 1 ***
2230 H = 0
2240 Y = C :REM VERGELIJKINGSWAARDE
2250 FOR I = Y-8 TO Y+8
2260 IF ( I < 1 ) OR ( I > 100 ) THEN 2390
2270 IF ( B(I) <> 999 ) AND ( B(I) <> I ) THEN 2390
2280 IF I > 1 THEN 2330
2290 G = I - C :REM AANTAL STAPPEN
2300 C = 1
2310 GOSUB 1450
2320 I = 8 + Y :REM UIT LUS
2330 IF ( ABS( I - D ) < 9 ) OR ( H = 2 ) THEN 2390
2340 IF E = I THEN 2390
2350 H = 1
2360 IF I > Y THEN 2390
2370 Y = I
2380 Q = I - C
2390 NEXT I
2400 IF H = 2 THEN 2530
2410 IF H = 1 THEN 2470
2420 REM ** C-SAURUS WEET NIET WAT HIJ DOEN MOET ! **
2430 Q = INT( RND(0) * (-8) - 1 )
2440 IF ABS( C + Q - L ) > 8 THEN 2470
2450 Q = Q * (-1)
2460 IF ( C+Q < 1 ) OR ( C+Q >100 ) THEN 2430
2470 G = Q :REM AANTAL STAPPEN
2480 C = C + Q
2490 E = C
2500 B(C) = A(C)
2510 C = B(C)
2520 GOSUB 1450
2530 RETURN
2540 :
2550 REM **** STUUR-MODULE COMPUSAURUS ****
2560 REM *** SPELREGELS ***
2570 GOSUB 90
2580 REM *** INITIALIZEREN ***

```

```
2590 GOSUB 400
2600 REM *** UITVOER ***
2610 GOSUB 790
2620 REM *** TELLER SPELER ***
2630 GOSUB 670
2640 REM *** INVOER SPELER ***
2650 GOSUB 1100
2660 REM *** BEREKENING SPELER ***
2670 GOSUB 1170
2680 REM *** UITVOER ***
2690 GOSUB 790
2700 IF H = 2 THEN 2820
2710 IF C = 0 THEN 2610
2720 REM *** TELLER C-SAURUS ***
2730 GOSUB 730
2740 REM *** AGRESSIEF AANVALS-PATROON ***
2750 GOSUB 1540
2760 IF H <> 0 THEN 2820
2770 GOSUB 1810
2780 IF H <> 0 THEN 2820
2790 GOSUB 2010
2800 IF H <> 0 THEN 2820
2810 GOSUB 2230
2820 IF H < 2 THEN 2610
2830 GOSUB 790
2840 END
```

Voorbeeld

*** HET JACHTGEBIED VAN DE COMPUSAURUS ***
 AANTAL ZETTEN COMPUSAURUS: 0
 AANTAL ZETTEN SPELER : 0

```

        *-0-0-0-0-0-0-0-0-0-0
      0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
    0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
    0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
      0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
        0-0-0-0-0-0-0-0-0-0-0-0-#
    
```

HOEVEEL STAPPEN WIL JE DOEN < MAX. 8 STAPPEN > 8

*** HET JACHTGEBIED VAN DE COMPUSAURUS ***
 AANTAL ZETTEN COMPUSAURUS: 0
 AANTAL ZETTEN SPELER : 1

```

      0-0-0-0-0-0-0-0-0-0-0-0*-0
    0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
  0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
  0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
    0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
      0-0-0-0-0-0-0-0-0-0-0-0-#
    
```

*** HET JACHTGEBIED VAN DE COMPUSAURUS ***
 AANTAL ZETTEN COMPUSAURUS: 1
 AANTAL ZETTEN SPELER : 1

```

      0-0-0-0-0-0-0-0-0-0-0-0*-0
    0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
  0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
  0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
    0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0-0
      0-#-0-0-0-0-0-0-0-0-0-0-0
    
```

IK HEB -8 STAPPEN GEDAAN.

HOEVEEL STAPPEN WIL JE DOEN < MAX. 8 STAPPEN > 8

3.5 De zandplaneet Silico

De zandplaneet Silico is de vijftigste planeet in de Schild-sector. Silico maakt een baan om een dubbelsterstelsel en uit waarnemingen is gebleken dat de massa van de planeet enorm is. Deze twee aspecten maken Silico uniek. Waarschijnlijk als gevolg van de baan heeft de planeet een zeer hoge rotatiesnelheid. De hoge rotatiesnelheid gecombineerd met de enorme zwaartekracht zorgen ervoor dat Silico meer weg heeft van een schijf dan van een bol. Hierdoor kan bij de navigatie op de planeet niet van boldriehoeksmeting worden uitgegaan. De planeet is ontzettend droog en ze bestaat voor het grootste gedeelte uit silicium, daarom wordt zij ook wel 'De Zandplaneet' genoemd. Op de planeet stormt het verschrikkelijk en er ontstaan plotseling enorme zandstromen die soms dagen aanhouden. De zandstromen zijn ware rivieren die zich met verschillende snelheden in verschillende richtingen bewegen.

U zult begrijpen dat zo'n wereld vanuit wetenschappelijk oogpunt geweldig interessant is. Het Galactisch Genootschap (the Galactic Council) heeft daarom op de planeet een expeditiebasis ingericht. Deze basis staat op een zodanig punt dat hij niet door de zandstromen kan worden meegevoerd.

Zoals uit de instructie blijkt, is het uw taak een survey-robot, de XAM, naar de expeditiebasis te dirigeren. Vanwege de hoge zwaartekracht is het voor het ruimteschip onmogelijk om in een stationaire omloopbaan te komen. De enig mogelijke omloopbaan, is die van precies 60 minuten. Het duurt dus altijd een uur voordat men kan zien hoe de XAM op de instructies heeft gereageerd.

Het beeldscherm laat behalve een kaart van het gebied ook nog een aantal belangrijke gegevens zien. Rechtsboven staan de omloopteller en de kompasroos. In het midden, onder de kompasroos, staan de coördinaten van de expeditiebasis. Links de X-coördinaat, rechts de Y-coördinaat. Rechtsmidden en rechtsonder staan twee kolommen getallen. Links de getallen op basis van de ingevoerde gegevens zonder zandstromen en rechts de waarden zoals doorgeseind door de XAM. Afwijkingen tussen beide ontstaan door de zandstromen.

Suggestie 1: Indien men op de gebruikte computer beschikt over PEEK- en POKE-instructies, hoeft men niet steeds het gehele scherm te wissen. Men hoeft alleen maar de oude plaats van de XAM te wissen en de nieuwe plaats in het plaatje te POKEn. Het zelfde geldt voor de andere gegevens. Wanneer men deze wijziging kan doorvoeren, zal het programma ook aanmerkelijk sneller lopen. Tevens kan het programma aanmerkelijk worden bekort. Onder andere kan de sorteergang ten behoeve van de uitvoer verdwijnen en de uitvoer worden vereenvoudigd.

Suggestie 2: Heeft men de beschikking over zgn. 'high resolution' dan zou men de feitelijke tocht van de XAM met een lijn kunnen aangeven. Men kan dan ook de plaats aangeven waar de survey-robot uit zou komen zonder invloed van de zandstromen.

Suggestie 3: Men zou het programma Ruimtevaartuig kunnen koppelen aan de zandplaneet Silico. Men moet dan wel over veel geheugenruimte beschikken of men moet in staat zijn programma's te koppelen via 'chaining'. Een aantal waarden moet dan via het common-block worden doorgegeven.

Wanneer men deze suggestie zou verwezenlijken, krijgt men eerst de landing van XAM op Silico en daarna de tocht naar de expeditiebasis.

Niet bepaald een eenvoudige opgave.

Programma

```
10 REM *** DE ZANDPLANEET SILICO ***
20 DATA 4,1.8, 4,2.6, 3,1.4, 3,3.4, 2,4.8, 1,0.4, 1,5.8
30 RANDOM
40 DIM P(9,3), Z(10,2), A$(4)
50 GOTO 2340
60 :
70 REM *** INSTRUCTIES ***
80 CLS
90 PRINT "*** DE ZANDPLANEET SILICO ***"      -VERKORTE INSTRUCT
100 PRINT
110 PRINT "U BEVINDT ZICH MET UW RUIMTESCHIP IN EEN"
120 PRINT "OMLOOPBAAN VAN 60 MINUTEN OM DE PLANEET"
130 PRINT "SILICO, DE VIJFTIGSTE PLANEET IN DE SCHILD--"
140 PRINT "SECTOR. XAM, EEN SURVEY-ROBOT, IS OP DIT"
150 PRINT "MOMENT GELAND. UIT DE EERSTE BERICHTEN"
160 PRINT "BLIJKT DAT DIT VLAKBIJ DE EXPEDITIE-BASIS IS."
170 PRINT "ALLEEN DE BERUCHE ZANDZEE MET HAAR ZAND--"
180 PRINT "STROMEN EN EEN BERGRUG SCHEIDEN DE XAM NOG."
190 PRINT "VANWEGE DE OMLOOPTIJD VAN 60 MIN. KAN XAM"
200 PRINT "ELK UUR VAN NIEUWE INSTRUCTIES WORDEN"
210 PRINT "VOORZIEN. DEZE INSTRUCTIES BESTAAN UIT EEN"
220 PRINT "EEN KOERS IN GRADEN EN EEN SNELHEID ,DIE"
230 PRINT "LIGT TUSSEN DE 0 EN 30 KNOTS."
240 PRINT "-- HEEFT U DIT GELEZEN DRUK DAN <RETURN> IN ---";
250 INPUT A$
260 CLS
270 PRINT "*** DE ZANDPLANEET SILICO ** KORTE INSTRUCTIE"
280 PRINT
290 PRINT "DE COMPUTER ZAL AAN DE HAND VAN DE GEGEVENS DE"
300 PRINT "COORDINATEN VAN DE BEREKENDE LOKATIE EN DE "
310 PRINT "WERKELIJKE LOKATIE GEVEN. EVENEENS WORDT IN"
320 PRINT "DE KAART DE WERKELIJKE PLAATS VAN XAM AANGE--"
330 PRINT "GEGEVEN (*). AFWIJKINGEN TUSSEN BEIDE LOKATIES"
340 PRINT "ONTSTAAN DOOR DE BERUCHE ZANDSTROMEN."
350 PRINT "-- DE EXPEDITIE-BASIS WORDT AANGEGEVEN MET --#--."
360 PRINT "-- ZANDBERGEN WORDEN AANGEGEVEN MET -O--."
370 PRINT "DE BEDOELING ZAL DUIDELIJK ZIJN:"
380 PRINT "DE XAM MOET NAAR DE EXPEDITIE-BASIS GEDIRIGEERD"
390 PRINT "WORDEN. EEN MOEILIJKE ZWARE TOCHT."
400 PRINT "-- HEEFT U DIT GELEZEN DRUK DAN <RETURN> IN ---";
410 INPUT A$
420 RETURN
430 :
440 REM *** INITIALIZEREN ***
450 REM ** ZANDSTROMEN PER BREEDTE **
460 FOR I = 1 TO 10
470 Z(I,1) = INT(RND(0) * 360) :REM RICHTING ZANDSTR.
480 Z(I,2) = INT(RND(0)*15+1)/100 :REM SNELHEID ZANDSTR.
490 NEXT I
500 REM ** PLAATSEN VAN BERGRUG **
510 RESTORE
520 FOR I = 1 TO 7
530 READ P(I,1), P(I,2) :REM INLEZEN LOKATIES
```

```

540 P(I,3) = 1 :REM AANDUIDING BERG
550 NEXT I
560 REM ** PLAATSEN VAN BASIS **
570 A = INT(RND(0) * 2 + 6) :REM Y-COORDINAAT
580 B = INT(RND(0) * 20 + 45)/10 :REM X-COORDINAAT
590 P(8,3) = 2 :REM BASIS PLAATSEN
600 P(8,1) = A
610 P(8,2) = B
620 REM ** PLAATSEN SURVEY-ROBOT XAM **
630 Y = 0 :REM Y-COORD XAM
640 X = INT(RND(0) * 15 + 15) / 10 :REM X-COORD XAM
650 Y1 = Y :REM BEREK Y-COORD
660 X1 = X :REM BEREK X-COORD
670 P(9,1) = Y
680 P(9,2) = X
690 P(9,3) = 3
700 REM ** VARIABELEN EEN STARTWAARDE GEVEN **
710 REM * STRING$ *
720 A$ = "0----1----2----3----4----5----6----7"
730 D$ = ""
740 C$(1) = "0" :REM SYMBOOL BERG
750 C$(2) = "#" :REM SYMBOOL BASIS
760 C$(3) = "*" :REM SYMBOOL XAM
770 C$(4) = " " :REM SYMBOOL XAM
780 REM * NUMERIEKE VARIABELEN *
790 W = 1 :REM OMLOOP
800 C = 3.1415926536 / 180 :REM GRADEN - RADIALEN
810 D = 180 / 3.1415926536 :REM RADIALEN - GRADEN
820 F = 0 :REM SNELHEID = 0
830 H = 0
840 RETURN
850 :
860 REM *** UITVOER ***
870 CLS
880 REM ** SORTEREN MAT P **
890 GOSUB 1290
900 PRINT "*** DE ZANDPLANEET SILICO ***";
910 PRINT TAB(35)"< OMLOOP >";W;
920 PRINT TAB(54)"360"
930 PRINT TAB(3)A$;TAB(55) "0"
940 M = 9 :REM STUURVARIABLE
950 FOR I = 10 TO 1 STEP (-1)
960 PRINT I-1; TAB(3)"!";
970 IF M <= 0 THEN 1060
980 IF P(M,1) <> I - 1 THEN 1060
990 P = P(M,2) * 35 / 7 + 3 :REM TAB-LOKATIE
1000 L = P(M,3) :REM SYMBOOL
1010 IF L <> 3 THEN 1030
1020 Z = M
1030 PRINT TAB(P)C$(L);
1040 M = M - 1
1050 IF M <> 0 THEN 980
1060 PRINT TAB(38)"!";
1070 IF I <> 10 THEN 1090
1080 PRINT TAB(49)"270 + 90";

```



```

1090 IF I <> 9 THEN 1110
1100 PRINT TAB(54)"180";
1110 IF I <> 8 THEN 1130
1120 PRINT TAB(41)B; TAB(47)"< X-B-Y >";A;
1130 IF I <> 7 THEN 1150
1140 PRINT TAB(40)"<BEREK.>-----<WERK.>";
1150 IF I <> 6 THEN 1170
1160 PRINT TAB(41)K1; TAB(47)"< KOERS >";K;
1170 IF I <> 5 THEN 1190
1180 PRINT TAB(41)F1; TAB(47)"< SNE LH >";F;
1190 IF I <> 3 THEN 1210
1200 PRINT TAB(41)X1; TAB(47)"< X-COR >";X;
1210 IF I <> 2 THEN 1230
1220 PRINT TAB(41)Y1; TAB(47)"< Y-COR >";Y;
1230 IF I <> 1 THEN 1250
1240 PRINT TAB(41) D$;
1250 PRINT
1260 NEXT I
1270 PRINT TAB(3)A$
1280 RETURN
1290 :
1300 REM *** SORTEREN MAT P ***
1310 FOR Q = 9 TO 2 STEP (-1)
1320 FOR R = 1 TO Q
1330 P = P(Q,1)*1000 - P(Q,2)
1340 L = P(R,1)*1000 - P(R,2)
1350 IF P > L THEN 1410
1360 FOR J = 1 TO 3
1370 M = P(Q,J)
1380 P(Q,J) = P(R,J)
1390 P(R,J) = M
1400 NEXT J
1410 NEXT R
1420 NEXT Q
1430 RETURN
1440 :
1450 REM *** GEGIST BESTEK ***
1460 P = K1 * C :REM KOERS IN RADIALEN
1470 Y1 = INT((S * COS(P) + Y)*10 + 0.5)/10 :REM NWE BR
1480 X1 = INT((S * SIN(P) + X)*10 + 0.5)/10 :REM NWE L
1490 W = W + 1 :REM OMLOOP + 1
1500 RETURN
1510 :
1520 REM *** FEITELIJKE KOERS ***
1530 S = S / 10
1540 FOR I = 1 TO 10
1550 L = INT(Y) + 1
1560 K = Z(L,1)
1570 S1 = Z(L,2)
1580 IF S = 0 THEN 1740
1590 REM ** KOERSCORRECTIE I.V.M. STROOM **
1600 R = Z(L,1) - K1 :REM STROOMRICHTING
1610 S1 = S↑2 + Z(L,2)↑2 - 2*S*Z(L,2)*COS((180-R) * C)
1620 IF ABS(S1) > 0.000001 THEN 1650
1630 S1 = 0

```

```

1640 K = -1
1650 S1 = SQR(S1) :REM RELATIEVE SNELHEID
1660 IF S1 = 0 THEN 1800
1670 P = SIN((180 - R) * C) / S1 * Z(L,2)
1680 P = ATN(P / SQR(-P * P + 1)) :REM C-KOERS IN RAD
1690 K = INT(P * D + K1+0.5) :REM FEITELIJKE KOERS
1700 IF K <= 360 THEN 1720
1710 K = K - 360
1720 IF K >= 0 THEN 1740
1730 K = K + 360
1740 P = K * C
1750 F = INT(S1 * 30 * 10 + 0.5)
1760 REM ** GEGIST BESTEK OP BASIS FEITELIJKE KOERS **
1770 Y = S1 * COS(P) + Y :REM NIEUW Y-COORD
1780 X = S1 * SIN(P) + X
1790 REM ** ERGENS IN DE BUURT **
1800 FOR J = 1 TO 9
1810 Y2 = P(J,1) - Y :REM AFSTAND Y-RICHT.
1820 X2 = P(J,2) - X :REM AFSTAND X-RICHT.
1830 IF ABS(Y2) > 0.15 THEN 1990
1840 IF ABS(X2) > 0.15 THEN 1990
1850 REM * WAAR DICHTBIJ ? *
1860 IF P(J,3) <> 1 THEN 1910
1870 REM * TEGEN BERG *
1880 H = 2 :REM STUURVARIABLE
1890 L = J :REM COORD OVERNEMEN
1900 D$ = "> XAM UITGESCHAKELD <"
1910 IF P(J,3) <> 2 THEN 1960
1920 REM * AANGEKOMEN BIJ BASIS *
1930 H = 3
1940 L = J
1950 D$ = "> * BASIS BEREIKT * <"
1960 IF H < 2 THEN 1990
1970 J = 9 :REM UIT J-LUS
1980 I = 10 :REM UIT I-LUS
1990 NEXT J
2000 NEXT I
2010 Y = INT(Y * 10 + 0.5) / 10 :REM AFGEROND COORD
2020 X = INT(X * 10 + 0.5) / 10 :REM AFGEROND COORD
2030 RETURN
2040 :
2050 REM *** NIEUWE LOKATIE XAM ***
2060 P(Z,1) = INT(Y)
2070 P(Z,2) = X
2080 IF H < 2 THEN 2130
2090 P = 0.1
2100 IF X2 >= 0 THEN 2120
2110 P = -0.1
2120 P(Z,2) = P(L,2) + P
2130 IF (P(Z,1) < 0) OR (P(Z,1) > 9) THEN 2150
2140 IF (P(Z,2) >= 0) AND (P(Z,2) <= 7) THEN 2200
2150 P(Z,1) = 0
2160 P(Z,2) = 0
2170 P(Z,3) = 4
2180 D$ = "> XAM BUITEN BEREIK <"

```

```

2190 H = 2
2200 RETURN
2210 :
2220 REM *** INVOER ***
2230 PRINT "KOERS IN GRADEN 0 - 360 :";
2240 INPUT K1
2250 IF (K1 < 0) OR (K1 > 360) THEN 2230
2260 PRINT "SNELHEID IN KNOTS 0-30 :";
2270 INPUT F1
2280 IF (F1 < 0) OR (F1 > 30) THEN 2260
2290 S = F1 / 30 :REM RELATIEVE SNELH
2300 RETURN
2310 :
2320 REM ***** STUURMODULE *****
2330 REM *** INSTRUCTIES ***
2340 GOSUB 80
2350 REM *** INITIALIZEREN ***
2360 GOSUB 460
2370 REM *** UITVOER ***
2380 GOSUB 870
2390 REM *** INVOER ***
2400 GOSUB 2230
2410 REM *** GEGIST BESTEK ***
2420 GOSUB 1460
2430 REM *** FEITELIJKE KOERS ***
2440 GOSUB 1530
2450 REM *** NIEUWE LOKATIE ***
2460 GOSUB 2060
2470 IF H < 2 THEN 2380
2480 REM *** UITVOER ***
2490 GOSUB 870
2500 END

```

Voorbeeld

```

*** DE ZANDPLANEET SILICO ***          < OMLOOP > 1          360
 0-----1-----2-----3-----4-----5-----6-----7
 9 !                                     !                       0
 8 !                                     !                       270 + 90
 7 !                                     !                       180
 6 !                                     ! #                       ! 5.1 < X-B-Y > 7
 5 !                                     ! <BEREK.>-----<WERK.>
 4 !           0       0                 ! 0 < KOERS > 0
 3 !             0       0               ! 0 < SNELH > 0
 2 !               0       0             ! 2.3 < X-COR > 2.3
 1 ! 0                 0                 ! 0 < Y-COR > 0
 0 !               *                       !
 0-----1-----2-----3-----4-----5-----6-----7
KOERS IN GRADEN 0 - 360 : 0
SNELHEID IN KNOTS 0-30 : 30

```

```

*** DE ZANDPLANEET SILICO ***          < OMLOOP > 2          360
 0-----1-----2-----3-----4-----5-----6-----7
 9 !                                     !                       0
 8 !                                     !                       270 + 90
 7 !                                     !                       180
 6 !                                     ! #                       ! 5.1 < X-B-Y > 7
 5 !                                     ! <BEREK.>-----<WERK.>
 4 !           0       0                 ! 0 < KOERS > 27
 3 !             0       0               ! 30 < SNELH > 33
 2 !               0       0             ! 2.3 < X-COR > 2.8
 1 ! 0                 *                 ! 1 < Y-COR > 1
 0 !               *                       !
 0-----1-----2-----3-----4-----5-----6-----7
KOERS IN GRADEN 0 - 360 : 355
SNELHEID IN KNOTS 0-30 : 20

```

```


*** DE ZANDPLANEET SILICO ***          < OMLOOP > 3          360
 0-----1-----2-----3-----4-----5-----6-----7
 9 !                                     !                       0
 8 !                                     !                       270 + 90
 7 !                                     !                       180
 6 !                                     ! #                       ! 5.1 < X-B-Y > 7
 5 !                                     ! <BEREK.>-----<WERK.>
 4 !           0       0                 ! 355 < KOERS > 7
 3 !             0       0               ! 20 < SNELH > 19
 2 !               0       0             ! 2.7 < X-COR > 2.9
 1 ! 0                 *                 ! 1.7 < Y-COR > 1.6
 0 !               *                       !
 0-----1-----2-----3-----4-----5-----6-----7
KOERS IN GRADEN 0 - 360 :

```

3.6 Rigel Zuiderkwartier

In dit spellenboek komt 'Startrek', het meest bekende computerspel uit de zeventiger jaren, niet voor. Dat Startrek niet is opgenomen, heeft twee oorzaken; ten eerste is het originele programma bijzonder groot en er zou veel uitleg bij moeten omdat een groot aantal mensen zich de televisieserie waarschijnlijk niet zullen herinneren.

De auteur vond daarom dat de andere klassieker 'Wumpus' niet mocht ontbreken. Alleen 'Wumpus' is in een nieuw jasje gestoken. De lokatie, waar het drama zich nu afspeelt, is het bekende Rigel Zuiderkwartier, een gebied dat aan de Middengolf ligt met in het achterland majestueuze bergen en diepe dalen. Een geliefd oord voor de verwerde interplanetaire vakantieganger.



Tijdens een excursie is een toerist (de speler), in een grottenstelsel gevallen. Het is een grottenstelsel waarin een 'Xlussel' huist. De enige manier om er levend uit te komen, is de Xlussel te vinden en buiten gevecht te stellen.


In de reisfolders over dit gebied wordt niet verteld dat de beruchte Xlussel en de Trocks nog in de krochten en onderaardse gewelven huizen van het Rigel Zuiderkwartier.

De Xlussel is een vreesaanjagend monster, dat alles van vlees en bloed met huid en haar verslindt. Hij stinkt dan ook twee grotten ver naar geronnen bloed. De meeste tijd brengt hij slapend door, zodat hij behoedzaam beslopen kan worden. Komt men echter in de grot waarin de Xlussel slaapt dan wordt hij razend snel wakker en er is geen ontkomen meer aan. De Trocks hebben niets van de Xlussel te vrezen. Door hun gezang houden zij de Xlussel meestal op afstand en mochten zij hem of iets anders tegenkomen dan grijpen ze dat en smijten het in een willekeurige grot neer.

In sommige grotten bevinden zich schachten die tot aan de wortels van de bergen reiken. Zo'n schacht veroorzaakt een enorme tocht, wat in een naastliggende grot te voelen is.

Suggestie 1: Op de tekening kan men zien hoe het grottenstelsel in elkaar zit. Aan de hand van de listing kan men nagaan, hoe het komt alsof het lijkt dat men zich iedere keer in een ander grottenstelsel waant. Men kan nu het grottenstelsel uitbreiden waardoor het spel zeker leuker wordt. Men moet er wel op bedacht zijn dat elke grot drie gangen moet hebben naar andere grotten. Houdt men zich hier niet aan dan loopt het hele programma niet meer. Men kan wel een grot naar zichzelf laten wijzen. Op deze manier kan men ook een doorgang naar een volgende grot creëren door een grot twee keer naar zichzelf te laten wijzen. Op deze manier kan men suggereren dat men in een ander grottenstelsel is terechtgekomen.

Suggestie 2: Heeft men de Xlussel verstoord door met een mes te gooien dan kan men het programma nog zó inrichten dat er daarna twee Xlussels verschijnen. Dit maakt het spel weer een stuk spannender.



Programma

```
10 REM *** RIGEL ZUIDERKWARTIER ***
20 DIM P(22,3), G(22), V(7)
30 RANDOM
40 GOTO 2070
50 DATA 11,2,12, 1,3,13, 2,4,14, 3,5,15, 4,6,16, 5,7,17
60 DATA 6,8,18, 7,9,19, 8,10,20, 9,11,21, 10,1,22, 22,13,1
70 DATA 12,14,2, 13,15,3, 14,16,4, 15,17,5, 16,18,6
80 DATA 17,19,7, 18,20,8, 19,21,9, 20,22,10, 21,12,11
90 RANDOM
100 GOTO 2070
110 :
120 REM *** INSTRUCTIES ***
130 CLS :REM SCHERM WISSEN
140 PRINT "** RIGEL-ZUIDERKWARTIER ** -KORTE INSTRUCTIE-"
150 PRINT
160 PRINT "U HEBT HET CONTACT MET HET REISGEZELSCHAP"
170 PRINT "VERLOREN. OM VOOR HET NOODWEER TE SCHUILEN"
180 PRINT "BENT U EEN GROT INGEKROPEN, VANWEGE HET"
190 PRINT "SLECHTE LICHT VIEL U IN EEN SPELONK."
200 PRINT "IN HET VAGE SCHIJNSEL DAT VAN BOVEN INVALT,"
210 PRINT "ZIET U HET NAARGEESTIG WOORD -XLUSSEL- IN DE"
220 PRINT "WAND GEKRAS. HET ENIGE DAT MEN VAN DE XLUSSEL"
230 PRINT "WEET, IS DAT HIJ NAAR GERONNEN BLOED STINKT"
240 PRINT "EN DAT MEN DAT MINSTENS TWEE GROTTEN VER KAN"
250 PRINT "RUIKEN. VERSTOORT MEN DE XLUSSEL DAN VERSLINDT"
260 PRINT "HIJ ELK SCHEPSEL MET HUID EN HAAR."
270 PRINT "DRUK <RETURN> IN ALS U DIT GELEZEN HEBT. ";
280 INPUT A$
290 CLS
300 PRINT "** RIGEL-ZUIDERKWARTIER ** -KORTE INSTRUCTIE-"
310 PRINT "HET ENIGE DAT ER OP ZIT, IS DE XLUSSEL TE"
320 PRINT "VINDEN EN HEM TE DODEN MET EEN VAN DE"
330 PRINT "DRIE WERPMESSEN, DIE NAAST U LIGGEN ! TROCKS"
340 PRINT "HUIZEN OOK IN DEZE GROTTEN . HET GEZANG IS EEN"
350 PRINT "GROT VERDER TE HOREN. ALS ZE U GRIJPEN,"
360 PRINT "WORDT U IN EEN WILLEKEURIGE ANDERE"
370 PRINT "GROT GESMETEN. BOVENDIEN KAN HET ER OOK"
380 PRINT "ERG KOUD ZIJN. WANNEER, IN EEN NAASTLIGGENDE"
390 PRINT "GROT EEN SCHACHT ZIT, ONTSTAAT ER"
400 PRINT "IJSKOUDE TOCHT. DE SCHACHTEN REIKEN TOT"
410 PRINT "AAN DE WORTELS VAN DE BERG EN ER IS NOG"
420 PRINT "NIEMAND UIT GEKOMEN! DIT IS EEN TOTAAL"
430 PRINT "ANDERE BESCHRIJVING VAN HET RIGELZUIDER-"
440 PRINT "KWARTIER DAN IN DE REISFOLDER."
450 PRINT "DRUK <RETURN>, WANNEER U DIT GELEZEN HEBT. ";
460 INPUT A$
470 RETURN
480 :
490 REM *** INITIALIZEREN ***
500 REM ** INTERNE DATA FILE LEZEN + GROTNRS **
510 RESTORE
520 FOR I = 1 TO 22
530 G(I) = I :REM VOORLOPIG GROTNR
```

```

540 READ P(I,1),P(I,2),P(I,3)
550 NEXT I
560 REM ** GROTTABEL DOOR ELKAAR GOOIEN **
570 X = INT(RND(0) * 35 + 1) :REM AANTAL VERANDERINGEN
580 FOR I = 1 TO X
590 K = INT(RND(0) * 22 + 1)
600 P = INT(RND(0) * 22 + 1)
610 IF P = K THEN 590
620 REM ** WISSELEN **
630 M = G(K)
640 G(K) = G(P)
650 G(P) = M
660 NEXT I
670 REM ** PLAATSEN TROCKS-SCHACHTEN-XLUSSEL **
680 FOR I = 1 TO B
690 X = INT(RND(0) * 22 + 1) :REM RANDOM PLAATS
700 IF I = 1 THEN 800
710 H = 0 :REM STUURVARIABLE
720 FOR J = 1 TO I-1
730 IF X <> V(J) THEN 760
740 H = 1
750 J = I :REM UIT LUS
760 NEXT J
770 IF H <> 0 THEN 690
780 D = X
790 IF I = B THEN 810
800 V(I) = X
810 NEXT I
820 REM ** VARIABELEN **
830 B = 0 :REM BEURTTELLER
840 M = 3 :REM AANTAL WERPMESSEN
850 L = 1 :REM TEGEN HET LIJF GELOPEN
860 Q = G(D)
870 G$ = " GAAN "
880 RETURN
890 :
900 REM *** WISSEN STRINGS + BEURT ***
910 B = B + 1 :REM BEURT + 1
920 H = 0
930 A$ = ""
940 B$ = ""
950 C$ = ""
960 D$ = ""
970 RETURN
980 :
990 REM *** UITVOER ***
1000 CLS :REM SCHERM WISSEN
1010 PRINT "*** RIGEL-ZUIDERKWARTIER ***";
1020 PRINT TAB(35) "BEURT :";B
1030 PRINT TAB(35) "U HEEFT NOG";M;" MESSEN."
1040 PRINT
1050 PRINT "U KOOS GROTTNUMMER :";Q;"OM NAAR TOE TE ";G$
1060 PRINT "U BEVINDT ZICH IN GROTTNUMMER :";G(D)
1070 PRINT
1080 PRINT "ER ZIJN GANGEN NAAR ";G(P(D,1));";";

```



```

1090 PRINT G(P(D,2));"EN";G(P(D,3))
1100 PRINT A$
1110 PRINT B$
1120 PRINT C$
1130 PRINT D$
1140 RETURN
1150 :
1160 REM *** INVOER ***
1170 PRINT "WILT U VERPLAATSEN OF GOOIEN < V/G >";
1180 INPUT D$
1190 L = 0
1200 D$ = LEFT$(D$,1)
1210 IF D$ <> "V" THEN 1240
1220 G$ = " GAAN "
1230 L = 1 :REM VERPLAATSEN
1240 IF D$ <> "G" THEN 1280
1250 Y = D
1260 L = 2 :REM GOOIEN
1270 G$ = " GOOIEN "
1280 IF L = 0 THEN 1170
1290 PRINT "NAAR WELKE GROT WILT U";G$;
1300 INPUT @
1310 REM ** IS ER EEN DOORGANG **
1320 H = 0
1330 FOR J = 1 TO 3
1340 S = G(P(D,J))
1350 IF S <> @ THEN 1390
1360 H = 1
1370 D = P(D,J) :REM FEITELIJK GROTNR
1380 J = 3
1390 NEXT J
1400 IF H <> 0 THEN 1440
1410 D$ = "*** DIE GROT IS NIET BEREIKBAAR VIA DEZE TUNNEL ***"
1420 IF @ <> G(D) THEN 1440
1430 D$ = "*** U BEVINDT ZICH AL IN DIE GROT, LUIAARD ! ***"
1440 RETURN
1450 :
1460 REM *** VERPLAATSEN ***
1470 REM ** IETS TEGEN HET LIJF GELOPEN **
1480 H = 0
1490 FOR I = 7 TO 1 STEP -1
1500 IF D <> V(I) THEN 1650
1510 IF I > 3 THEN 1570
1520 H = 1
1530 D$ = "U WERD DOOR TROCKS GEGREPEN."
1540 V(I) = INT(RND(0) * 15 + 1) :REM NWE PLAATS TROCKS
1550 D = INT(RND(0) * 15 + 1) :REM NWE PLAATS SPELER
1560 IF V(I) = D THEN 1540
1570 IF (I < 4) OR (I > 6) THEN 1600
1580 H = 3 :REM STUURVARIABLE
1590 F$ ="HET IS AFGELOPEN.U BENT IN EEN SCHACHT GEVALLEN"
1600 IF I <> 7 THEN 1630
1610 H = 3
1620 F$ ="DE XLUSSEL HEEFT U MET HUID EN HAAR VERSLONDEN!"
1630 IF H <> 3 THEN 1650

```

```

1640 I = 1                                :REM UIT DE LUS
1650 NEXT I
1660 IF H = 1 THEN 1480
1670 IF H = 3 THEN 1890
1680 REM ** AANWIJZINGEN TROCKS & SCHACHTEN **
1690 FOR I = 1 TO 6
1700 FOR J = 1 TO 3
1710 IF P(D,J) <> V(I) THEN 1760
1720 IF I > 3 THEN 1740
1730 A$ = "ER IS GEZANG TE HOREN."
1740 IF I < 4 THEN 1760
1750 B$ = "HET IS HIER IJSKOUD. HET TOCHT !"
1760 NEXT J
1770 NEXT I
1780 REM ** AANWIJZING XLUSSEL **
1790 X = D
1800 FOR I = 1 TO 4
1810 IF I = 1 THEN 1830
1820 X = P(D,I-1)                          :REM GROTINGANG
1830 FOR J = 1 TO 3
1840 IF P(X,J) <> V(7) THEN 1870
1850 C$ = "HET STINKT NAAR GERONNEN BLOED !"
1860 I = 4
1870 NEXT J
1880 NEXT I
1890 RETURN
1900 :
1910 REM *** AANVAL OP XLUSSEL ***
1920 F$ = "ZONDER MESSEN, U BENT REDDELOOS VERLOREN !"
1930 H = 3
1940 IF M <= 0 THEN 2030
1950 F$ = "U HEEFT DE XLUSSEL GEDOOD ! JAMMER "
1960 IF D = V(7) THEN 2020
1970 C$ = "U HEEFT DE XLUSSEL VERSTOORD. HIJ IS GEVLUCHT !"
1980 V(7) = INT(RND(0) * 15 + 1)
1990 IF V(7) = Y THEN 1980
2000 D = Y
2010 H = 0
2020 M = M - 1                              :REM WERPME - 1
2030 RETURN
2040 :
2050 REM **** STUURMODULE ****
2060 REM *** INSTRUCTIES ***
2070 GOSUB 130
2080 REM *** INITIALIZEREN ***
2090 GOSUB 510
2100 IF B = 0 THEN 2170
2110 REM *** INVOER ***
2120 GOSUB 1170
2130 IF H = 0 THEN 2190
2140 REM *** WISSEN STRINGS ***
2150 GOSUB 910
2160 REM *** BEREKENING ***
2170 ON L GOSUB 1480 , 1920
2180 REM *** UITVOER ***

```

```
2190 GOSUB 1000
2200 IF H < 3 THEN 2120
2210 REM *** RESULTAAT ***
2220 PRINT F$
2230 END
```

Voorbeeld

```
*** RIGEL ZUIDERKWARTIER ***           BEURT : 0
                                           U HEEFT NOG 3  MESSEN.
```

```
U KOOS GROTNUMMER  8 OM NAAR TOE TE  GAAN
U BEVINDT ZICH IN GROTNUMMER : 8
```

```
ER ZIJN GANGEN NAAR  6 , 9 EN 14
HET IS HIER IJSKOUD. HET TOCHT !
HET STINKT NAAR GERONNEN BLOED !
WILT U VERPLAATSEN OF GOOIEN < V/G >VERPL
NAAR WELKE GROT WILT U GAAN  6
```

```
*** RIGEL ZUIDERKWARTIER ***           BEURT : 1
                                           U HEEFT NOG 3  MESSEN.
```

```
U KOOS GROTNUMMER  6 OM NAAR TOE TE  GAAN
U BEVINDT ZICH IN GROTNUMMER : 6
```

```
ER ZIJN GANGEN NAAR  8 , 1 EN 11
ER IS GEZANG TE HOREN.
WILT U VERPLAATSEN OF GOOIEN < V/G >VERPLAATS
NAAR WELKE GROT WILT U GAAN  1
```

```
*** RIGEL ZUIDERKWARTIER ***           BEURT : 2
                                           U HEEFT NOG 3  MESSEN.
```

```
U KOOS GROTNUMMER  1 OM NAAR TOE TE  GAAN
U BEVINDT ZICH IN GROTNUMMER : 15
```

```
ER ZIJN GANGEN NAAR  13 , 4 EN 11
U WERD DOOR TROCKS GEGREPEN.
HET IS AFGELOPEN. U BENT IN EEN SCHACHT GEVALLEN.
```

3.7 De slag bij de Shrixheuvels

Het gebied van de Shrixheuvels vormt de landengte tussen het Noordercontinent en het Zuidercontinent. De landengte was vroeger een stuk niemandsland, maar sinds de Schrammers het in bezit hebben genomen, hebben de handelskaravanen ontzettend veel last van de Schrammers.

Dat het gebied belangrijk is, laat zich gemakkelijk raden.

De Schrammers is niet hun werkelijke naam, maar ze worden zo genoemd vanwege hun uiterlijk. Ze lijken veel op bytosaurussen.

De taak van de strateeg van de Grauwduivels is het gebied van de Shrixheuvels te zuiveren van Schrammers. Hiertoe zullen eerst de strategische punten van de Schrammers dienen te worden gezocht en daarna aangevallen.

De Grauwduivels lijken wel wat op de Schrammers, hun ontwikkeling ligt echter iets hoger. Dat laatste klopt ook wel, omdat zij afstammen van de compusaurussen. Het bijzondere van deze Grauwduivels is, dat zij kunnen worden geprogrammeerd en wel op twee manieren: als Loerduivels of als Hellevegen. Eenmaal geprogrammeerd, zijn veranderingen niet meer mogelijk.

De Loerduivels kunnen alleen als verkenners worden gebruikt, de Hellevegen alleen als aanvallers op de Schrammers.

Om de strategische punten van de Schrammers te ontdekken, moeten Loerduivels worden ingezet. Dit kan alleen vanuit het zuiden (Y -coördinaat = 0). Wanneer de Loerduivels terugkomen, geven ze de strateeg een afstand op vanaf de ingezette lokatie tot aan het gevonden strategisch punt.



Mochten er nog meer strategische punten zijn dan blijkt dit uit de opmerking 'STRATEGISCH PUNT +1'. Wil men met voldoende zekerheid een strategisch punt bepalen dan moet men twee verkenningen van verschillende lokaties doen. Denkt de strateeg een punt te hebben gevonden dan kan hij de X- en de Y-coördinaat van het punt opgeven. Verschijnt er in het kaartje een 'O' dan klopt het en men zou een aanval kunnen wagen. Slaagt de aanval dan verandert de 'O' in '*'.

Wanneer de aanval niet slaagt, trekken de niet-vermiste Hellevegen zich direct terug. De Schrammers hebben natuurlijk ook verliezen geleden, wat dus de verdediging van het strategisch punt aantast. Slaagt een aanval op een strategisch punt dan vluchten de niet-vermiste Schrammers naar de overige nog niet bezette punten.

De strateeg volgt de strijd vanaf zijn commandopost met een blanco kaart, waarop de coördinaten zijn aangekend. Verder ligt er een liniaal en een passer. Het beeldscherm en de computer worden gebruikt om de Grauwduivels te programmeren. Tegelijkertijd levert de computer hem een aantal belangrijke gegevens, die de terugkerende troepen hem melden. Rechtsmidden wordt aangegeven hoeveel Grauwduivels en welke soort voor een bepaalde actie zijn ingezet. Wanneer Loerduivels zijn ingezet, verschijnt eronder de afstand tot een strategisch punt. Rechtsonder wordt bijgehouden hoeveel Schrammers en Grauwduivels er in totaal vermist zijn en hoeveel tijdens de uitgevoerde actie.

Suggestie 1: Het gedeelte van het programma dat de uitvoer regelt, neemt vrij veel ruimte in beslag. Met behulp van PEEK- en POKE-instructies kan het beeldscherm sneller en eenvoudiger worden opgebouwd. Men hoeft dan niet meer iedere keer het gehele scherm te wissen, men hoeft alleen maar de wijziging op het scherm aan te brengen. Het programma kan dan worden ingekort en zal daardoor sneller lopen.

Suggestie 2: Wanneer een strategisch punt eenmaal is bezet door de Hellevegen heeft de strateeg hier verder geen omkijken meer naar. Hij verliest een aantal Hellevegen in de strijd, maar verder niets. Om het spel reëler te doen lijken, moet de strateeg ook het punt verdedigen tegen de Schrammers. Dit laatste kan alleen wanneer hij er een aantal Hellevegen leert ter verdediging. Is zo'n punt slecht verdedigd, dan kan het weer door de Schrammers worden ingenomen. Het spel wordt er wel interessanter door.

Suggestie 3: Om het spel nog aantrekkelijker te maken, kan men de Loerduivels niet alleen de afstand tot een strategisch punt laten geven, maar ook een indicatie laten geven van de sterkte van het punt. De nauwkeurigheid van de gegeven 'sterkte' kan men afhankelijk stellen van het aantal ingezette Loerduivels.

Programma

```
10 REM *** DE SLAG BIJ DE SHRIX-HEUVELS ***
20 RANDOM
30 DIM A(5,4), B$(5), C$(3)
40 GOTO 2840
50 :
60 REM *** INSTRUCTIE ***
70 CLS :REM SCHERM WISSEN
80 PRINT "*** DE SLAG BIJ DE SHRIX-HEUVELS ***";
90 PRINT " -VERKORTE INSTRUCTIES-"
100 PRINT
110 PRINT "U BENT AANVOERDER VAN EEN HORDE GRAUWDUIVELS."
```

```

120 PRINT "ER ZIJN TWEE SOORTEN GRAUWDUIVELS : "
130 PRINT "-- LOERDUIVELS, DIE ALLEEN ALS VERKENNER GEBRUIKT"
140 PRINT "  KUNNEN WORDEN."
150 PRINT "-- HELLEVEGEN, DIE ALLEEN KUNNEN VECHTEN."
160 PRINT "U PROBEERT HET GEBIED VAN DE SHRIX-HEUVELS IN "
170 PRINT "BEZIT TE KRIJGEN DOOR DE STRATEGISCHE PUNTEN"
180 PRINT "TE ONTDEKKEN EN TE BEZETTEN."
190 PRINT
200 PRINT "DRUK <RETURN> IN ALS U DIT GELEZEN HEEFT." ;
210 INPUT A$
220 CLS
230 PRINT "*** DE SLAG BIJ DE SHRIX-HEUVELS ***";
240 PRINT " -VERKORTE INSTRUCTIES-"
250 PRINT
260 PRINT "DE PLAATS VAN DE STRATEGISCHE PUNTEN MOET U ZIEN"
270 PRINT "TE BEPALEN MET UW LOERDUIVELS. DE AANVAL OP ZO'N"
280 PRINT "STRATEGISCH PUNT MOET U MET DE HELLEVEGEN DOEN."
290 PRINT "DE STRATEGISCHE PUNTEN WORDEN VERDEDIGD DOOR DE"
300 PRINT "DE SCHRAMMERS, DE TEGENHANGERS VAN DE HELLEVEGEN."
310 PRINT "DE REST VAN DE INFORMATIE VINDT U IN HET BOEK."
320 PRINT
330 PRINT "DRUK <RETURN> IN ALS U DIT GELEZEN HEEFT." ;
340 INPUT A$
350 RETURN
360 :
370 REM *** INITIALIZEREN ***
380 REM ** STRATEGISCHE PUNTEN **
390 S = INT(RND(0)*4+2) :REM AANTAL STRATEGISCHE PUNTEN
400 T = 0 :REM AANTAL SCHRAMMERS
410 FOR Y = 1 TO S
420 I = INT(RND(0) * 6 + 2) :REM Y-COORD STRAT.PUNT
430 J = INT(RND(0) * 8) :REM X-COORD STRAT.PUNT
440 IF Y = 1 THEN 540
450 H = 0 :REM STUURVARIABELE
460 REM * CONTROLE OP VOORKOMEN *
470 FOR X = 1 TO Y-1
480 IF (I <> A(X,1)) OR (J <> A(X,2)) THEN 510
490 H = 1 :REM FOUT !
500 X = Y
510 NEXT X
520 IF H = 1 THEN 420
530 REM * WAARDEN OPNEMEN *
540 A(Y,1) = I
550 A(Y,2) = J
560 A(Y,3) = INT(RND(0) * 8+2)*10 :REM AANTAL SCHRAMMERS
570 A(Y,4) = 0 :REM NIET ONTDEKT.
580 T = T + A(Y,3) :REM TOTAAL SCHRAMMERS
590 NEXT Y
600 REM * SORTEREN *
610 FOR Q = S TO 2 STEP (-1)
620 FOR R = 1 TO Q
630 P = A(Q,1) * 100 - A(Q,2)
640 L = A(R,1) * 100 - A(R,2)
650 IF P > L THEN 710
660 FOR J = 1 TO 2

```

```

670 M = A(Q,J)
680 A(Q,J) = A(R,J)
690 A(R,J) = M
700 NEXT J
710 NEXT R
720 NEXT Q
730 REM ** STRINGS EN VARIABLEN **
740 A$ = "0--1--2--3--4--5--6--7"
750 B$(1) = "LOERDUIVELS"
760 B$(2) = "HELLEVEGEN"
770 B$(3) = ""
780 B$(4) = "O"
790 B$(5) = "*"
800 C$(1) = "VERKENNING "
810 C$(2) = "AANVAL "
820 C$(3) = "STATUS QUO "
830 L = 3 :REM STARTWAARDE
840 REM * GRAUWDUIVELS *
850 G = INT(RND(0) * T/2 + T/2) * 8 :REM START GRAUWD.
860 G1 = 0 :REM TOT.VERMIST
870 G2 = 0 :REM IN SLAG VERMIST
880 P(1) = 0 :REM LOERDUIVELS
890 P(2) = 0 :REM HELLEVEGEN
900 Q = 0 :REM INGEZET
910 REM * SCHRAMMERS *
920 T1 = 0 :REM TOT.VERMIST
930 T2 = 0 :REM IN SLAG VERMIST
940 S1 = 0 :REM GEVONDEN STRAT P.
950 N = T / S :REM GEM. AANTAL SCHR.
960 RETURN
970 :
980 REM *** VOORBEREIDING NIEUWE AKTIE ***
990 Q = 0 :REM INGEZET = 0
1000 T2 = 0 :REM SCHRAMMERS VERMIST = 0
1010 G2 = 0 :REM GRAUWDUIVELS VERMIST = 0
1020 D$ = ""
1030 REM ** WAT DOEN ? **
1040 PRINT "GAAT U AANVALLEN OF VERKENNEN ";
1050 INPUT Q$
1060 Q$ = LEFT$(Q$,1) :REM ALLEEN LINKER KARAKTER
1070 L = 1 :REM STUURVARIABLE
1080 IF Q$ = "V" THEN 1120
1090 L = 2
1100 IF Q$ = "A" THEN 1120
1110 L = 3
1120 RETURN
1130 :
1140 REM *** AANTAL IN TE ZETTEN GRAUWDUIVELS ***
1150 Q = 0
1160 GOSUB 1540
1170 PRINT"HOEVEEL";B$(L);"WILT U INZETTEN VOOR DE";C$(L);
1180 INPUT Q
1190 IF Q <= 0 THEN 1160
1200 H = 0 :REM STUURVARIABLE
1210 REM ** CONTROLE **

```

```

1220 R = Q - P(L) :REM HOEVEEL AL GEBRUIKT
1230 IF R <= 0 THEN 1380
1240 H = 1
1250 REM ** UIT GRAUWDUIVELS **
1260 IF R < G THEN 1290
1270 REM ** TE WEINIG **
1280 R = G
1290 Q = P(L) + R
1300 PRINT "U HEEFT ";Q;B$(L);" INGEZET ACCOORD <JA/NEE> ";
1310 INPUT Q$
1320 Q$ = LEFT$(Q$,1)
1330 IF Q$ = "N" THEN 1150
1340 IF H = 0 THEN 1380
1350 REM ** AANPASSING **
1360 G = G - R
1370 P(L) = P(L) + R
1380 RETURN
1390 :
1400 REM *** WELKE LOKATIE ***
1410 GOSUB 1540
1420 C$ = "VANAF "
1430 IF L = 1 THEN 1450
1440 C$ = "OP "
1450 PRINT C$;"WELKE LOKATIE WILT U EEN ";C$(L);"DOEN ?"
1460 PRINT "GEEF <X-COORD , Y-COORD> ";
1470 INPUT X,Y
1480 IF (X < 0) OR (X > 7) THEN 1410
1490 IF (Y < 0) OR (Y > 7) THEN 1410
1500 IF (L = 1) AND (Y <> 0) THEN 1410
1510 RETURN
1520 :
1530 REM *** UITVOER ***
1540 CLS :REM SCHERM WISSEN
1550 PRINT "*** DE SLAG BIJ DE SHRIX-HEUVELS ***";
1560 PRINT TAB(38) "--- ";D$;" ---"
1570 PRINT TAB(4) C$(L);
1580 PRINT TAB(29) "<BESCHIKBAAR>"
1590 PRINT TAB(4) A$;TAB(34)"GRAUWDUIVELS :";G
1600 FOR I = 8 TO 1 STEP (-1)
1610 PRINT I-1;TAB(3)"!";
1620 FOR M = 9 TO 1 STEP (-1)
1630 IF A(M,1) <> I-1 THEN 1680
1640 IF A(M,4) = 0 THEN 1680
1650 P = A(M,2) * 21 / 7 + 4 :REM TABLOKATIE
1660 R = A(M,4) + 3
1670 PRINT TAB(P) B$(R);
1680 NEXT M
1690 PRINT TAB(26)"!";
1700 IF I <> 8 THEN 1720
1710 PRINT TAB(34) "LOERDUIVELS :";P(1);
1720 IF I <> 7 THEN 1740
1730 PRINT TAB(34) "HELLEVEGEN :";P(2);
1740 IF I <> 5 THEN 1760
1750 PRINT TAB(29) "<INGEZET>";
1760 IF I <> 4 THEN 1780

```



```

1770 PRINT TAB(34) B$(L);TAB(47)":";Q;
1780 IF (H <> 3) OR (I <> 3) THEN 1800
1790 PRINT TAB(34)"AFSTAND";TAB(47)":";V;
1800 IF I <> 2 THEN 1820
1810 PRINT TAB(29) "<VERMIST>          TOT          ACTIE";
1820 IF I <> 1 THEN 1840
1830 PRINT TAB(34) "GRAUWDUIVELS :";G1;TAB(55)"<>";G2;
1840 PRINT
1850 NEXT I
1860 PRINT TAB(4) A$;TAB(34) "SCHRAMMERS   :";T1;
1870 PRINT TAB(55)"<>";T2
1880 RETURN
1890 :
1900 REM *** BEREKENING AFSTAND ***
1910 D$ = "NIETS TE VINDEN"
1920 IF S1 => S THEN 2050
1930 D$ = "STRATEGISCH PUNT"
1940 IF S1 + 1 = S THEN 1960
1950 D$ = "STRATEGISCH PUNT + 1"
1960 FOR I = 1 TO S
1970 IF A(I,4) <> 0 THEN 2000
1980 J = I
1990 I = S                               :REM UIT LUS
2000 NEXT I
2010 V = (A(J,2)-X)2 + (A(J,1)-Y)2 :REM BEPALING AFST.
2020 V = INT(SQR(V) * 10) / 10          :REM AFRONDEN
2030 REM ** VOLDOENDE LOERDUIVELS INGEZET **
2040 P = (A(J,3)*(-0.2)+N)/2           :REM BESPIEDINGSGRAAD
2050 R = INT(RND(0) * Q )
2060 IF Q > P THEN 2100
2070 R = Q
2080 V = 0
2090 D$ = "NIEMAND TERUG"
2100 GOSUB 2150
2110 GOSUB 2210
2120 RETURN
2130 :
2140 REM *** BIJWERKEN LOERDUIVELS VERMIST ***
2150 P(L) = P(L) - R
2160 G1 = G1 + R
2170 G2 = G2 + R
2180 RETURN
2190 :
2200 REM *** LOKATIE BEKEND ***
2210 IF S1 >= S THEN 2370
2220 H = 3
2230 GOSUB 1540
2240 H = 0
2250 D$ = ""
2260 PRINT "LOKATIE BEKEND <JA / NEE >";
2270 INPUT Q$
2280 IF LEFT$(Q$,1) <> "J" THEN 2380
2290 PRINT "GEEF X-COORD , Y-COORD ";
2300 INPUT X,Y
2310 R = 0

```

```

2320 GOSUB 2410
2330 IF R = 0 THEN 2380
2340 IF A(R,4) <> 0 THEN 2380
2350 A(R,4) = 1
2360 D$ = "GEVONDEN"
2370 S1 = S1 + 1 :REM WEER LOKATIE BEKEND
2380 RETURN
2390 :
2400 REM ** CONTROLE **
2410 FOR I = 1 TO S
2420 IF (X <> A(I,2)) OR (Y <> A(I,1)) THEN 2450
2430 R = I
2440 I = S
2450 NEXT I
2460 RETURN
2470 :
2480 REM *** BEREKENING AANVAL ***
2490 R = 0
2500 D$ = "HINDERLAAG"
2510 GOSUB 2410
2520 IF R = 0 THEN 2780
2530 D$ = "AFGESLAGEN"
2540 D = INT(RND(0) * Q/2 + Q/2)
2550 IF D <= A(R,3) THEN 2600
2560 REM ** STRATEGISCH PUNT IN BEZIT **
2570 D$ = "PUNT BEZET"
2580 A(R,4) = 2
2590 REM ** VERMISTE SCHRAMMERS **
2600 T2 = INT(RND(0) * A(R,3))
2610 T1 = T1 + T2
2620 A(R,3) = A(R,3) - T2
2630 IF A(R,4) <> 2 THEN 2780
2640 REM ** STRATEGISCHE PUNTEN IN HANDEN SCHRAMMERS **
2650 D = 0
2660 FOR I = 1 TO S
2670 IF A(I,4) = 2 THEN 2690
2680 D = D + 1
2690 NEXT I
2700 H = 9
2710 IF D = 0 THEN 2780
2720 H = 0
2730 REM ** TERUGTREKKEN **
2740 FOR I = 1 TO S
2750 IF A(I,4) = 2 THEN 2770
2760 A(I,3) = A(I,3) + INT(A(R,3) / D)
2770 NEXT I
2780 R = INT(RND(0) * Q)
2790 GOSUB 2150
2800 RETURN
2810 :
2820 REM **** STUURMODULE ****
2830 REM *** INSTRUCTIE ***
2840 GOSUB 70
2850 REM *** INITIALIZEREN ***
2860 GOSUB 390

```

```

2870 REM *** UITVOER ***
2880 GOSUB 1540
2890 REM *** INVOER ***
2900 GOSUB 990
2910 IF L = 3 THEN 2880
2920 REM *** AANTAL IN TE ZETTEN GRAUWDUIVELS ***
2930 GOSUB 1150
2940 REM *** WELKE LOKATIE ***
2950 GOSUB 1410
2960 REM *** VERKENNING = 1 AANVAL = 2 ***
2970 ON L GOSUB 1910 , 2490
2980 L = 3
2990 REM *** RESULTAAT ***
3000 R = P(2) + G
3010 IF (H <> 9) AND (R > 20) THEN 2880
3020 D$ = "VERLOREN"
3030 IF H <> 9 THEN 3060
3040 D$ = ">>GEWONNEN<<"
3050 REM *** UITVOER ***
3060 GOSUB 1540
3070 END

```

Voorbeeld

```

*** DE SLAG BIJ DE SHRIX-HEUVELS ***  --  --
      VERKENNING                <BESCHIKBAAR>
      0--1--2--3--4--5--6--7      GRAUWDUIVELS : 700
7 !                               !          LOERDUIVELS : 5
6 !                               !          HELLEVEGEN  : 0
5 !                               !
4 !                               !          <INGEZET>
3 !               *              0 !          LOERDUIVELS : 0
2 !                               !
1 !                               !          <VERMIST>          TOT      ACTIE
0 !                               !          GRAUWDUIVELS : 172  <> 0
      0--1--2--3--4--5--6--7      SCHRAMMERS   : 0      <> 0
HOEVEEL LOERDUIVELS WILT U INZETTEN VOOR DE VERKENNING 50
U HEEFT 50 LOERDUIVELS INGEZET ACCOORD <JA/NEE> JA

```

```

*** DE SLAG BIJ DE SHRIX-HEUVELS *** -- --
VERKENNING <BESCHIKBAAR>
0--1--2--3--4--5--6--7
7 ! : : : : : : : :
6 ! : : : : : : : :
5 ! : : : : : : : :
4 ! : : : : 0 ! <INGEZET>
3 ! : * : 0 ! LOERDUIVELS : 50
2 ! : : : : : : : :
1 ! : : : : : : : : <VERMIST> TOT ACTIE
0 ! : : : : : : : : GRAUWDUIVELS : 172 <> 0
0--1--2--3--4--5--6--7 SCHRAMMERS : 0 <> 0
VANAF WELKE LOKATIE WILT U EEN VERKENNING DOEN ?
GEEF <X-COORD , Y-COORD> 7 , 0

```

```

*** DE SLAG BIJ DE SHRIX-HEUVELS *** -- STRATEGISCH PUNT + 1
VERKENNING <BESCHIKBAAR>
0--1--2--3--4--5--6--7
7 ! : : : : : : : :
6 ! : : : : : : : :
5 ! : : : : : : : :
4 ! : : : : 0 ! <INGEZET>
3 ! : * : 0 ! LOERDUIVELS : 50
2 ! : : : : : : : : AFSTAND : 6
1 ! : : : : : : : : <VERMIST> TOT ACTIE
0 ! : : : : : : : : GRAUWDUIVELS : 214 <> 42
0--1--2--3--4--5--6--7 SCHRAMMERS : 0 <> 0

```

```

*** DE SLAG BIJ DE SHRIX-HEUVELS *** -- --
STATUS QUO <BESCHIKBAAR>
0--1--2--3--4--5--6--7
7 ! : : : : : : : :
6 ! : : : : : : : :
5 ! : : : : : : : :
4 ! : : : : 0 ! <INGEZET>
3 ! : * : 0 ! : 50
2 ! : : : : : : : :
1 ! : : : : : : : : <VERMIST> TOT ACTIE
0 ! : : : : : : : : GRAUWDUIVELS : 214 <> 42
0--1--2--3--4--5--6--7 SCHRAMMERS : 0 <> 0
GAAT U AANVALLEN OF VERKENNEN

```

4. Simulatiespellen

4.1 De kolenmijn

Men staat bij de ingang van een oude kolenmijn waarin een schat verborgen ligt. Men probeert de schat te zoeken en naar de buitenwereld te brengen. Tot de uitrusting hoort een lamp. Met het commando 'AAN' gaat de lamp aan; met het commando 'UIT' dooft de lamp. Tijdens de tocht door de mijn zullen aanwijzingen worden gegeven wat er te zien of te horen valt. Hoort men geiep dan weet men dat er een vleermuis in de buurt is. Wordt deze in zijn slaap gestoord dan grijpt hij de indringer en smijt hem in een willekeurige grot diep in de mijn. Men moet er wel rekening mee houden dat vleermuizen op het licht afkomen.

Wanneer men de lamp aan heeft, ziet men de luchtschachten. Is de lamp uit dan valt men er onherroepelijk in en komt men in één van de lager gelegen grotten terecht.

Heeft men ten slotte het goud gevonden dan is men er nog niet. Men moet ook nog naar buiten zien te komen door een van de uitgangen. Ook hierbij worden aanwijzingen gegeven. Andere schatzoekers zijn u voorgeweest, maar zij zijn er nooit levend uitgekomen.

Om zich te verplaatsen door het grottenstelsel zijn de volgende commando's beschikbaar: NOORD, ZUID, OOST, WEST, NAARBOVEN en OMLAAG. Heeft men eenmaal een uitgang gevonden dan kan men met het commando 'VERLATEN' weer in de buitenlucht komen.

Suggestie 1: Gezien de opbouw van het programma is het vrij eenvoudig een andere locatie te kiezen, bijvoorbeeld een eiland met piraten.



Suggestie 2: Beschikt men over veel geheugenruimte, dan kan men de grotten namen geven in plaats van nummers. Het spel wordt er zeker aantrekkelijker door.

Suggestie 3: In het programma wordt een lamp gebruikt. Als men valt, gaat de lamp niet uit en hij gaat ook niet kapot. Door nu de lamp als het ware met petroleum te vullen, kan worden gesuggereerd dat hij ook brandstof verbruikt. Elk beurt dat men de lamp aan heeft, wordt één eenheid brandstof verbruikt. Valt men tot overmaat van ramp in een schacht dan verliest men een gedeelte van de resterende brandstof.

Suggestie 4: Gezien de opbouw is het spel op een heel eenvoudige wijze drastisch uit te breiden, met allerlei gevaren die de schatzoeker bedreigen. Wat denkt u van een Balrog die plotseling verschijnt of een Trol ?

Programma

```
10 REM *** KOLENMIJN ***
20 RANDOM
30 DIM A(100,6),B(100),V(5),T$(9)
40 GOTO 3070
50 REM *** SPELREGELS ***
60 CLS
70 PRINT "*** KOLENMIJN ***           -VERKORTE SPELREGELS-"
80 PRINT
90 PRINT "DE BEDOELING VAN DIT SPEL IS EEN SCHAT"
100 PRINT "TE ZOEKEN EN VERVOLGENS MET DE SCHAT DE"
110 PRINT "BUITENWERELD TE BEREIKEN."
120 PRINT "VOOR HET ZICH VERPLAATSSEN VAN DE ENE GROT"
130 PRINT "NAAR DE ANDERE ZIJN DE VOLGENDE COMMANDO'S"
140 PRINT "BESCHIKBAAR: - NOORD, ZUID, OOST, WEST, NAAR"
150 PRINT "BOVEN EN OMLAAG -."
160 PRINT "JE HEBT EEN LAMP BIJ JE OM WAT TE KUNNEN ZIEN."
170 PRINT "MET HET COMMANDO 'AAN' GEEFT DE LAMP LICHT,"
180 PRINT "MET HET COMMANDO 'UIT' IS DE LAMP GEDOOFD. "
190 PRINT "LET OP ! DE VLEERMUIZEN KOMEN OP HET LICHT AF."
200 PRINT
210 PRINT "HEB JE DIT GELEZEN DRUK <RETURN> IN-";
220 INPUT K$
230 CLS
240 PRINT "*** KOLENMIJN ***           -VERKORTE SPELREGELS-"
250 PRINT
260 PRINT "HEB JE EEN UITGANG GEVONDEN DAN KUN JE DE MIJN"
270 PRINT "VERLATEN MET HET COMMANDO 'VERLATEN'."
280 PRINT "JE BENT NIET DE ENIGE SCHATZOEKER GEWEEST. "
290 PRINT "MISSCHIE N KOM JE ZE TEGEN ? DOOD OF LEVEND !"
300 PRINT
310 PRINT "RUSTIG ! ER IS EEN INSTORTING IN DE KOLENMIJN !"
320 RETURN
330 :
340 REM *** INITIALIZEREN ***
350 REM ** TEKSTEN + COMMANDO'S **
360 T$(1) = "WEST "
370 T$(2) = "OOST "
380 T$(3) = "NOORD "
```

```

390 T$(4) = "ZUID "
400 T$(5) = "NAAR BOVEN "
410 T$(6) = "OMLAAG "
420 T$(7) = "VERLATEN "
430 T$(8) = "AAN"
440 T$(9) = "UIT"
450 REM ** STUURVARIABELE WAARDE GEVEN **
460 Z = 0 :REM ZETTEN TELLER
470 Q = 0 :REM GOUD LIGT KLAAR
480 G = 1 :REM GROTNUMMER STARTPOSITIE
490 D = 1 :REM VORIG GROTNUMMER
500 REM ** STRING$ WISSEN **
510 GOSUB 1870
520 REM ** GROTTEN SCHOONMAKEN **
530 FOR I=1 TO 100
540 B(I) = 1
550 FOR J = 1 TO 6
560 A(I,J) = 0
570 NEXT J
580 NEXT I
590 REM ** GROTTENSTELSEL IN 3-D RUIMTE BOUWEN **
600 REM * WEST = KOLOM 1 OOST = KOLOM 2 *
610 FOR X = 5 TO 100 STEP 5
620 I = X - 4
630 A(I,1) = X :REM WEST
640 A(X,2) = I :REM OOST
650 FOR Y = I + 1 TO I + 4
660 A(Y,1) = Y - 1
670 A(Y-1,2) = Y
680 NEXT Y
690 NEXT X
700 REM * NAAR BOVEN = KOLOM 5 OMLAAG = KOLOM 6 *
710 FOR I = 21 TO 100
720 A(I,5) = I - 20 :REM NAAR BOVEN
730 A(I-20,6) = I :REM OMLAAG
740 IF I > 80 THEN 760
750 B(I) = 0 :REM CONTROLE
760 NEXT I
770 REM * NOORD = KOLOM 3 ZUID = KOLOM 4 *
780 FOR Y = 16 TO 96 STEP 20
790 FOR X = 0 TO 4
800 H = X + Y - 15
810 A(H,3) = Y + X :REM NOORD
820 A(H,4) = A(H,3) - 10 :REM ZUID
830 NEXT X
840 FOR X = 5 TO 19
850 H = X + Y - 15
860 A(H,3) = X + Y - 20 :REM NOORD
870 A(H,4) = A(H,3) + 10 :REM ZUID
880 IF X < 15 THEN 900
890 A(H,4) = A(H,3) - 10
900 NEXT X
910 NEXT Y
920 REM ** WILLEKEURIGE DOORGANGEN DICHTMAKEN **
930 Y = INT(RND(0)*100+1) :REM AANTAL DOORGANG = MAX.10

```

```

940 FOR X = 1 TO Y
950 J = INT(RND(0)*6+1) :REM KOLOMBEPALING
960 I = INT(RND(0)*100+1) :REM REGELBEPALING
970 IF B(I) >= 4 THEN 950
980 B(I) = B(I) + 1
990 A(I,J) = 0
1000 NEXT X
1010 FOR I = 1 TO 100
1020 B(I) = 0
1030 NEXT I
1040 REM ** SCHACHTEN PLAATSEN =1 **
1050 Y = INT(RND(0)*20+1) :REM AANTAL SCHACHTEN
1060 FOR X = 1 TO Y
1070 I = INT(RND(0)*100+1) :REM GROTBEPALING
1080 B(I) = 1
1090 NEXT X
1100 REM ** UITGANGEN MAKEN =2 **
1110 B(1) = 2 :REM HOOFDUITGANG
1120 A(1,5) = -10 :REM DOORGANG OMHOOG
1130 FOR X = 1 TO 5 :REM 5 UITGANGEN
1140 I = INT(RND(0)*18+2) :REM GROTBEPALING
1150 B(I) = 2 :REM UITGANG PLAATSEN
1160 A(I,5) = -10
1170 NEXT X
1180 REM ** GOUD PLAATSEN =3 **
1190 I = INT(RND(0)*90+11) :REM GROTBEPALING
1200 IF B(I) <> 0 THEN 1190 :REM IS GROT VRIJ ?
1210 B(I) = 3 :REM PLAATS VAN HET GOUD
1220 REM ** 5 VLEERMUIZEN PLAATSEN IN MAT V **
1230 FOR X = 1 TO 5
1240 V(X) = INT(RND(0)*80+21) :REM SLAAPPLAATS VLEERMUIS
1250 NEXT X
1260 RETURN
1270 :
1280 REM *** UITVOER ***
1290 CLS
1300 PRINT "*** KOLENMIJN ***";
1310 PRINT TAB(30)"ZET :";Z
1320 PRINT
1330 PRINT "JE BEVINDT JE IN GROTNR. :";G;
1340 PRINT ". JE WAS IN GROTNR.:";D
1350 PRINT A$ :REM AANWIJZING INVOER I
1360 PRINT B$ :REM AANWIJZING INVOER II
1370 PRINT
1380 PRINT "IN DE VOLGENDE RICHTINGEN ZIJN DOORGANGEN :";
1390 FOR J = 1 TO 6
1400 IF A(G,J) = 0 THEN 1420
1410 PRINT " - ";T$(J); :REM BESCHIKBARE V-COMMANDO
1420 NEXT J
1430 PRINT " - "
1440 PRINT C$ :REM AANWIJZING VLEERMUIS
1450 PRINT D$ :REM AANWIJZING SCHACHT
1460 PRINT E$ :REM AANWIJZING UITGANG
1470 PRINT F$ :REM AANWIJZING GOUD
1480 PRINT

```



```

1490 RETURN
1500 :
1510 REM *** INVOER + CONTROLE ***
1520 PRINT "WAT IS JE KEUZE ";
1530 INPUT K$
1540 K$ = LEFT$(K$,2) :REM EERSTE KARAKTER
1550 V = 0
1560 H = 0 :REM H=0 FOUT ANTWOORD
1570 FOR J = 1 TO 9 :REM MOGELIJKE ANTWOORDEN
1580 IF LEFT$( T$(J),2 ) <> K$ THEN 1810
1590 H = 1
1600 REM ** MOET LAMP UIT ? **
1610 IF J <> 9 THEN 1650
1620 A$ = "LAMP IS UIT !!"
1630 L = 0 :REM LAMP IS UIT
1640 REM ** MOET LAMP AAN ? **
1650 IF J <> 8 THEN 1690
1660 A$ = "LAMP IS AAN!!"
1670 L = 1 :REM LAMP IS AAN
1680 REM ** VERPLAATSEN ? **
1690 IF J <> 7 THEN 1740
1700 V = -10 :REM RELATIEF VERPLAATST
1710 IF B(G) = 2 THEN 1740
1720 A$ = "SUFFERD, ER IS HIER GEEN UITGANG !!"
1730 V = 0 :REM VERPLAATST. ONGEDAAN
1740 IF J > 6 THEN 1800
1750 V = A(G,J) :REM VERPLAATST ?
1760 IF A(G,J) <> 0 THEN 1800
1770 A$ = "JE BENT TEGEN EEN MUUR AANGEKNALD !!"
1780 B$ = "ER IS HIER GEEN DOORGANG."
1790 V = 0 :REM VERPLAATS. ONGEDAAN
1800 J = 9 :REM UIT LUS
1810 NEXT J
1820 IF H <> 0 THEN 1840
1830 A$ = "IK BEGRIJP JE NIET !! WAT BEDOEL JE ?"
1840 RETURN
1850 :
1860 REM *** OPHOGEN TELLER +STRING$ WISSEN ***
1870 Z = Z+1 :REM ZETTEN VERHOGEN
1880 D = G :REM VORIGE=HUIDIGE
1890 A$ = ""
1900 B$ = ""
1910 C$ = ""
1920 D$ = ""
1930 E$ = ""
1940 F$ = ""
1950 RETURN
1960 :
1970 REM *** VLEERMUIS ***
1980 REM ** AANWIJZING = C$ **
1990 REM ** OP LICHT AFKOMEN **
2000 IF L <> 1 THEN 2110 :REM L=1 LICHT AAN !
2010 FOR X = 1 TO 5
2020 H = ABS( V(X) - G ) :REM AFSTAND TOT VLEERM.
2030 IF H > 30 THEN 2090 :REM TE VER WEG ?

```

```

2040 REM ** VLEERMUIS HERKENT LICHT **
2050 IF V(X) => G THEN 2070
2060 V(X) = V(X) + 1 :REM VLEERM. ONDER JE
2070 IF V(X) <= G THEN 2090
2080 V(X) = V(X) - 1 :REM VLEERM. BOVEN JE
2090 NEXT X
2100 REM ** GRIJPEN VLEERMUIS **
2110 FOR J = 1 TO 6 :REM AANTAL MOGELIJKE DOORGANGEN = 6
2120 IF A(G,J) <= 0 THEN 2170
2130 FOR X = 1 TO 5
2140 IF A(G,J) <> V(X) THEN 2160 :REM VLEERMUIS ERBIJ ?
2150 C$ = "ER IS GEPIEP TE HOREN."
2160 NEXT X
2170 NEXT J
2180 REM ** GRIJPEN **
2190 FOR X = 1 TO 5
2200 IF V(X) <> G THEN 2250 :REM VLEERMUIS GRIJPT ?
2210 C$ = "JE WERD DOOR EEN VLEERMUIS GEGREPEN !!"
2220 G = INT(RND(0)*90+11) :REM WILLEKEURIGE GROT
2230 V(X) = INT(RND(0)*90+11) :REM NIEUWE SLAAPPLAATS
2240 X = 5 :REM UIT LUS
2250 NEXT X
2260 RETURN
2270 :
2280 REM *** SCHACHT ***
2290 REM ** AANWIJZING = D$ **
2300 IF B(G) <> 1 THEN 2360 :REM DOORGANG ?
2310 D$ = "ER IS HIER EEN SCHACHT, GELUKKIG WAS DE LAMP AAN"
2320 IF L <> 0 THEN 2360 :REM LAMP AAN ?
2330 REM ** LAMP WAS UIT **
2340 G = G + INT(RND(0)*(100-G)) :REM GROTTEN OMLAAG
2350 D$ = "JE BENT DOOR EEN LUCHTSCHACHT GEVALLEN"
2360 RETURN
2370 :
2380 REM *** UITGANG ***
2390 REM ** AANWIJZING = E$ **
2400 IF (G > 41) OR (Z < 11) THEN 2500
2410 FOR J = 1 TO 6
2420 IF A(G,J) <= 0 THEN 2460
2430 H = A(G,J) :REM GROT ERNAAST/BOVEN
2440 IF B(H) <> 2 THEN 2460
2450 E$ = "ER IS HIER EEN ZWAK LICHTSCHIJNSEL."
2460 NEXT J
2470 REM ** GROT MET UITGANG **
2480 IF B(G) <> 2 THEN 2500
2490 E$ = "JE KUNT DE ZON ZIEN SCHIJNEN."
2500 RETURN
2510 :
2520 REM *** GOUD ***
2530 REM ** AANWIJZING = F$ **
2540 IF Q = 1 THEN 2740 :REM HEEFT GOUD ?
2550 REM ** VLAKBIJ GOUD ? **
2560 FOR I = G - 1 TO G + 1
2570 IF (I < 1) OR (I > 100) THEN 2680
2580 FOR J = 1 TO 6

```

```

2590 IF A(I,J) <= 0 THEN 2670
2600 H = A(I,J) :REM GROT ERNAAST/BOVEN
2610 IF B(H) <> 3 THEN 2670
2620 F$ = "ER STAAN HIER VOETSTAPPEN."
2630 IF I <> G THEN 2670
2640 F$ = "ER LIGT HIER EEN GERAAMTE."
2650 J = 5 :REM UIT LUS
2660 I = G + 1
2670 NEXT J
2680 NEXT I
2690 REM ** GOUD GEVONDEN **
2700 IF B(G) <> 3 THEN 2740
2710 F$ = "JE HEBT HET GOUD GEVONDEN."
2720 Q = 1
2730 B(G) = 0 :REM GOUD MEEGENOMEN
2740 RETURN
2750 :
2760 REM *** AKTIE ***
2770 G = V :REM NIEUWE GROT
2780 IF V <> 0 THEN 2800
2790 G = D
2800 IF G <= 0 THEN 2890
2810 REM ** VLEERMUIS **
2820 GOSUB 2000
2830 REM ** SCHACHT **
2840 GOSUB 2300
2850 REM ** UITGANG **
2860 GOSUB 2400
2870 REM ** GOUD **
2880 GOSUB 2540
2890 RETURN
2900 :
2910 REM *** RESULTAAT ***
2920 CLS
2930 PRINT"*** KOLENMIJN ***"
2940 PRINT
2950 PRINT "EINDELIJK DE FRISSE LUCHT !"
2960 PRINT "JE BENT ER HEELHUIDS UITGEKOMEN, "
2970 A$ = "MAAR ZONDER HET GOUD STOMMELING."
2980 IF Q <> 1 THEN 3000
2990 A$ = "MET HET GOUD. FANTASTISCH !!"
3000 PRINT A$
3010 PRINT
3020 PRINT "IN TOTAAL HAD JE ER ";Z;"ZETTEN VOOR NODIG"
3030 RETURN
3040 :
3050 REM **** STUURMODULE ****
3060 REM ** SPELREGELS **
3070 GOSUB 60
3080 REM ** INITIALIZEREN **
3090 GOSUB 360
3100 REM ** UITVOER **
3110 GOSUB 1290
3120 REM ** OPHOGEN TELLER **
3130 GOSUB 1870

```

```
3140 REM ** INVOER **
3150 GOSUB 1520
3160 REM ** AKTIE **
3170 GOSUB 2770
3180 IF G > 0 THEN 3110
3190 REM ** RESULTAAT **
3200 GOSUB 2720
3210 END
```

Voorbeeld

```
*** KOLENMIJN ***           ZET : 1
JE BEVINDT JE IN GROTNR.   : 1 . JE WAS IN GROTNR.   : 1
IN DE VOLGENDE RICHTINGEN ZIJN DOORGANGEN :
- WEST - OOST - NAAR BOVEN - OMLAAG
WAT IS JE KEUZE  NOORD
```

```
*** KOLENMIJN ***           ZET : 2
JE BENT TEGEN EEN MUUR AANGEKNALD !!
ER IS HIER GEEN DOORGANG.
JE BEVINDT JE IN GROTNR.   : 1 . JE WAS IN GROTNR.   : 1
IN DE VOLGENDE RICHTINGEN ZIJN DOORGANGEN :
- WEST - OOST - NAAR BOVEN - OMLAAG
WAT IS JE KEUZE  OMLAAG
```

```
*** KOLENMIJN ***           ZET : 3
JE BEVINDT JE IN GROTNR.   : 21 . JE WAS IN GROTNR.   : 1
IN DE VOLGENDE RICHTINGEN ZIJN DOORGANGEN :
- WEST - OOST - NAAR BOVEN - OMLAAG ER IS GEPIEP TE HOREN.
WAT IS JE KEUZE  AAN
```

4.2 Boerderij Zeldenrust

In dit eenvoudige simulatiespel speelt men een boer die twee gewassen verbouwt: gruwelknollen, die overigens erg lekker zijn, en blauwe kool. Het doel van het spel is een zo groot mogelijke oogst binnen te halen.

Gruwelknollen sterven af wanneer de temperatuur voor langere tijd onder de 10 graden Celsius blijft of boven de 35 graden. Zij kunnen echter warm of koud worden gehouden door rode gruwelmieren die in grote kolonies op de gruwelknol nestelen. Aan het eind van de oogst moeten de gruwelknollen wel worden schoongemaakt. Een nadeel is dat de rode gruwelmieren verzot zijn op blauwe kool.

Kaalsprinkhanen jagen voornamelijk op gruwelmieren en blauwe hommels. De blauwe hommels zorgen voor de bestuiving van de blauwe kool. Wanneer er een sluipmuggenplaag is, vreten kaalsprinkhanen ook deze insecten.

Als het lange tijd droog is, moet men sluisjes openen om water in te laten. De sloten zijn tevens de broedplaatsen van de gevreesde sluipmug. De sluipmuggen zijn berucht vanwege hun vraatzucht. Beide gewassen hebben hier erg onder te lijden.

Om de insectenpopulaties onder controle te houden, zijn drie bestrijdingsmiddelen beschikbaar die echter ook een schadelijke uitwerking hebben op de gewassen. Men heeft de keuze uit:

APV : 10% van de insecten sterft en 4% van de gewassen.

SLEM : 40% van de insecten sterft en 25% van de gewassen.

ZWOP : 75% van de insecten sterft en 50% van de gewassen.

Men moet proberen de verschillende insectenpopulaties in de hand te houden om een zo groot mogelijke opbrengst te krijgen. Dit probleem kent zeer veel oplossingen. Dit komt onder andere doordat het weer een grote invloed uitoefent. Het betekent ook dat het spel steeds opnieuw kan worden gespeeld zonder te vervelen.

Suggestie: De groei van het gewas is niet te beïnvloeden, men is afhankelijk van het weer. In de moderne akkerbouwbedrijven wordt veel met kunstmest gewerkt. Dit element kan men ook nog aan het programma toevoegen. Men moet er wel op bedacht zijn dat er niet teveel kunstmest wordt gebruikt. Gebruikt men te veel dan verliest men een gedeelte van zijn oogst. Het is een kwestie van een extra module toevoegen.

Programma

```
10 REM *** BOERDERIJ ZELDENRUST ***
20 DIM A(3,2)
30 DATA 0.1,0.04, 0.4,0.25, 0.75,0.5
40 RANDOM
50 GOTO 2470
60 :
70 REM *** INITIALIZEREN ***
80 A = 0 :REM AANTAL DROGE DAGEN
90 D = 0 :REM DAGTELLER
100 G = 500 :REM GRUWELKNOLLEN
110 B = 500 :REM BLAUWE KOOL
120 S = 0 :REM SLUISJES
130 D2 = 0 :REM DAGEN > 20 C
```

```

140 L = 0 :REM SLUISJES
150 K = 0 :REM SPEL LOOPT
160 K1 = 0
170 REM ** BESTRIJDINGSMIDDELEN INLEZEN **
180 FOR I = 1 TO 3
190 READ A(I,1),A(I,2)
200 NEXT I
210 REM ** WEERSGESTELDHEID - STARTWAARDEN **
220 R = INT(RND(0) * 10) + 1 :REM KANS REGEN/DROOGTE
230 T = INT(RND(0) * 45) -5 :REM TEMPERATUUR
240 REM ** INSEKTENPOPULATIE - STARTWAARDEN **
250 H = INT(RND(0) * 500) + 1 :REM BLAUWE HOMMELS
260 P = INT(RND(0) * 500) + 1 :REM KAALSPRINKHANEN
270 M = INT(RND(0) * 500) + 1 :REM RODE GRUWELMIEREN
280 S = 0 :REM SLUIPMUGGEN
290 RETURN
300 :
310 REM *** UITVOER NAAR BEELDSCHERM ***
320 CLS :REM SCHERM WISSEN
330 PRINT "*** BOERDERIJ *** ";
340 PRINT "DAG :";D,"TEMPERATUUR :";T1;"C"
350 IF K1 <> 0 THEN PRINT ">> INSEKTENPLAAG !! <<"
360 PRINT TAB(32) "WEERSVERWACHTING :";A#
370 PRINT "-- GEWASSEN --";
380 PRINT TAB(16) "AANTAL GRUWELKNOLLEN :";G
390 PRINT TAB(16) "AANTAL BLAUWE KOOL :";B
400 PRINT
410 PRINT "-- INSEKTEN-POPULATIES --";
420 PRINT TAB(27) "AANTAL BLAUWE HOMMELS :";H
430 PRINT TAB(27) "AANTAL KAALSPRINKHANEN :";P
440 PRINT TAB(27) "AANTAL RODE GRUWELMIEREN :";M
450 PRINT TAB(27) "AANTAL SLUIPMUGGEN :";S
460 PRINT "AANTAL SLUISJES :";L
470 PRINT "+++++"

```



```

480 RETURN
490 :
500 REM *** CONTROLE OP < 0 ***
510 IF G < 0 THEN G = 0 :REM GRUWELKNOLLEN
520 IF B < 0 THEN B = 0 :REM BLAUWE KOOL
530 IF P < 0 THEN P = 0 :REM SPRINKHANEN
540 IF H < 0 THEN H = 0 :REM HOMMELS
550 IF M < 0 THEN M = 0 :REM MIEREN
560 IF S < 0 THEN S = 0 :REM SLUIPMUGGEN
570 RETURN
580 :
590 REM *** BEWERKINGSMODULE ***
600 D = D + 1 :REM DAGTELLER
610 REM ** WEERSGESTELDHEID **
620 Q = 0 :REM HULPVARIABLE
630 REM * TEMPERATUUR *
640 IF T <= 10 THEN Q = Q + 1
650 IF T < 3 THEN Q = Q + 0.5
660 IF T < 0 THEN Q = Q + 0.75
670 IF T < -3 THEN Q = Q + 1
680 IF T > 35 THEN Q = Q + 1
690 IF T > 37 THEN Q = Q + 1.75
700 R = INT(RND(0) * 10) + 1 :REM KANS REGEN/BUI/DROOGTE
710 A$ = "DROOG"
720 A = A + 1
730 IF R <= 4 THEN 800
740 A$ = "BUI"
750 A = A - 1
760 IF R < 7 THEN 800
770 A = A - 2
780 A$ = "REGEN"
790 REM ** NIEUWE TEMPERATUUR **
800 T1 = T :REM OUDE TEMPERATUUR
810 X = INT(RND(0) * 11) - 5 :REM TEMPERATUURSVERANDERING
820 IF X + T > 40 OR X + T < -5 THEN X = X * -1
830 T = T + X
840 IF T > 20 THEN D2 = D2 + 1 :REM WARME DAGEN + 1
850 RETURN
860 :
870 REM *** INVLOEDEN OP GROEI EN INSEKTENPOPULATIES ***
880 K1 = 0 :REM GEEN INSEKTENPLAAG
890 X = P + M + S + H :REM TOTALE INSEKTENPOPULATIE
900 Y = (G + B) * 3 + 4500 :REM GRENS INSEKTENPLAAG
910 IF X < Y THEN 950
920 K1 = 1 :REM INSEKTENPLAAG
930 B = INT(B - RND(0) * B/2 - B/2):REM VRAAT KOOL
940 G = INT(G - RND(0) * G/2 - G/2):REM VRAAT KNOL
950 X = RND(0) * H/1000 :REM EFFECTIVITEIT HOMMELS
960 B = B + INT(X * B)
970 G = G - INT(RND(0) * X * G)
980 REM ** EFFECTIVITEIT RODE GRUWELMIEREN **
990 X = INT(RND(0) * M/2 + M/2)
1000 Y = RND(0) * 2
1010 REM ** EFFECT VAN TEMPERATUUR OP KNOL **
1020 IF Q <= Y THEN 1070

```

```

1030 IF G > X/200 THEN 1060
1040 G = INT(G - (RND(0) * G / 6))
1050 GOTO 1070
1060 G = INT(RND(0) * X/200 + (2/3) * (X/200))
1070 X = X / 1500
1080 B = INT(B - (B * X))
1090 G = INT(G + (G * X))
1100 REM ** EFFECT KAALSPRINKHANEN **
1110 X = INT((RND(0) * P)/5) :REM FACTOR SPRINKHANEN
1120 H = H - X :REM UITROEI HOMMELS
1130 M = M - X :REM UITROEI MIEREN
1140 S = S - INT(X * RND(0))
1150 REM ** EFFECT SLUIPMUGGEN **
1160 S = S + (L * INT(RND(0) * 100 + 1)) :REM SLUIPMUGGEN
1170 IF S < 5 THEN 1230
1180 X = INT(((RND(0) * S) + 1)/20)
1190 G = G - X :REM VRAAT KNOLLEN
1200 X = INT(((RND(0) * S) + 1)/20)
1210 B = B - X :REM VRAAT KOOL
1220 REM ** DROOGTESHADE **
1230 IF A - L < 5 THEN 1260 :REM VOLDOENDE SLUIZEN
1240 G = INT(G - G * RND(0))
1250 B = INT(B - B * RND(0))
1260 GOSUB 510 :REM CONTROLE OP NUL
1270 RETURN
1280 :
1290 REM *** VRAAGSTELLING ALGEMEEN ***
1300 I = 0
1310 PRINT "0 = NIETS DOEN ";
1320 PRINT " 3 = SLUISJE OPENEN"
1330 PRINT "1 = INSEKTEN VRIJLATEN ";
1340 PRINT " 4 = SLUISJE SLUITEN"
1350 PRINT "2 = BESTRIJDINGSMIDDEL SPUITEN"
1360 PRINT "WAT IS UW KEUZE ";
1370 INPUT I
1380 IF I < 0 OR I > 4 THEN 1300
1390 IF I = 3 THEN L = L + 1
1400 IF I = 4 AND L > 0 THEN L = L - 1
1410 RETURN
1420 :
1430 REM *** INSEKTEN VRIJLATEN ***
1440 CLS :REM SCHERM WISSEN
1450 PRINT "DAG :";D;"TEMPERATUUR :";T1;"C"
1460 PRINT
1470 PRINT "1 = BLAUWE HOMMELS 3 = RODE GRUWELMIEREN"
1480 PRINT "2 = KAALSPRINKHANEN 4 = SLUIPMUGGEN"
1490 PRINT
1500 PRINT "WELKE SOORT MOET WORDEN UITGEZET ";
1510 INPUT X
1520 IF X < 1 OR X > 4 THEN 1440
1530 PRINT "HOEVEEL VAN DE GEKOZEN INSEKTEN ";
1540 INPUT Y
1550 Y = ABS(INT(Y))
1560 IF Y < 2500 THEN 1600
1570 K1 = 1 :REM INSEKTENPLAAG

```



```

1580 G = INT(G - (RND(0) * G/2 + G/2)
1590 B = INT(B - (RND(0) * B/2 + B/2)
1600 IF X = 1 THEN H = H + Y      :REM HOMMELS
1610 IF X = 2 THEN P = P + Y      :REM SPRINKHANEN
1620 IF X = 3 THEN M = M + Y      :REM MIEREN
1630 IF X <> 4 THEN 1680
1640 PRINT "WAAROM ";Y;"SLUIPMUGGEN !!"
1650 S = S + Y
1660 FOR X = 1 TO 200              :REM WACHTLUS
1670 NEXT X
1680 GOSUB 510
1690 RETURN
1700 :
1710 REM *** VRAAGSTELLING BESTRIJDINGSMIDDELEN ***
1720 CLS
1730 I=0
1740 PRINT"*** BOERDERIJ *** ";
1750 PRINT"DAG :";D,"TEMPERATUUR :";T1;"C."
1760 PRINT
1770 PRINT"1 = APV      10% INSEKTEN  4 % GEWASSEN"
1780 PRINT"2 = SLEM     40%      ''    25 %      ''"
1790 PRINT"3 = ZWOP     75%      ''    50 %      ''"
1800 PRINT
1810 PRINT"WELKE BESTRIJDINGSMIDDEL MOET GESPOTEN WORDEN ";
1820 INPUT I
1830 IF I < 1 OR I > 3 THEN 1720
1840 I=INT(I)
1850 REM ** EFFECTEN INSEKTICIDEN **
1860 Y=A(I,1)                      :REM EFFECT OP GEWAS
1870 X=A(I,2)                      :REM EFFECT OP INSECT
1880 G=INT(G-(G*X))                :REM EFFECT OP KNOL
1890 B=INT(B-(B*X))                :REM EFFECT OP KOOL
1900 H=INT(H-(H*Y))                :REM EFFECT OP HOMMEL
1910 M=INT(M-(M*Y))                :REM EFFECT OP MIER
1920 P=INT(P-(P*Y))                :REM EFFECT OP SPRINKHAAN
1930 S=INT(S-(S*Y))                :REM EFFECT OP SLUIPMUGGEN
1940 GOSUB 510
1950 RETURN
1960 :
1970 REM *** GROEI INSEKTEN-POPULATIE ***
1980 S = INT(S + S * 0.10)          :REM UITKOMEN VAN SLUIPMUG
1990 H = INT(H + H * 0.11)          :REM UITKOMEN VAN HOMMELS
2000 M = INT(M + M * 0.12)          :REM UITKOMEN VAN MIEREN
2010 P = INT(P + P * 0.15)          :REM UITKOMEN VAN SPRINKHANEN
2020 GOSUB 510
2030 RETURN
2040 :
2050 REM *** GROEI GEWASSEN ***
2060 X = D/5
2070 IF X <> INT(D/5) THEN 2170 :REM GROEISTOOT ELKE 5 DAGEN
2080 REM ** GROEI AFHANKELIJK VAN VOCHTIGHEID EN WARMTE **
2090 Y = INT(G*(.06+(RND(0)*(D2+L-A))/100))
2100 IF M < 5 THEN Y = 0
2110 G = G + Y
2120 Y = INT(B*(.06+(RND(0)*(D2+L-A))/100))

```

```

2130 IF H < 5 THEN Y = -INT(RND(0)*Y)
2140 B = B + Y
2150 D2 = 0
2160 GOSUB 510
2170 RETURN
2180 :
2190 REM *** VERLOREN ***
2200 CLS :REM SCHERM WISSEN
2210 PRINT"*** BOERDERIJ *** ":
2220 PRINT
2230 PRINT"JE BENT ER IN GESLAAGD HET GROOTSTE GEDEELTE"
2240 PRINT"VAN DE OOGST TE VERNIETIGEN."
2250 PRINT
2260 PRINT"JE HEBT NOG MAAR";G;"GRUWELKNOLLEN EN"
2270 PRINT B;"BLAUWE KOOL."
2280 PRINT
2290 PRINT"ZO'N LUIE BOER HEB IK NOG NOOIT MEEGEMAAKT !!"
2300 IF K1 = 0 THEN 2320
2310 PRINT"JE HEBT EEN INSEKTENPLAAG ONTKETEND !!"
2320 K = 1
2330 RETURN
2340 :
2350 REM *** GEWONNEN ***
2360 CLS :REM SCHERM WISSEN
2370 PRINT"*** BOERDERIJ *** ":
2380 PRINT
2390 PRINT"DE OOGST IS BINNEN ! ER ZIJN";G;"GRUWELKNOLLEN"
2400 PRINT"EN";B;"BLAUWE KOOL GEOOGST."
2410 PRINT
2420 PRINT"JE VERDIENT EEN PRIJS VOOR AKKERBOUW."
2430 K = 1
2440 RETURN
2450 :
2460 REM *** STUURMODULE ***
2470 GOSUB 80 :REM INITIALIZEREN
2480 REM *** SPEL ***
2490 GOSUB 600 :REM WEERSGESTELDHEID
2500 GOSUB 320 :REM UITVOER
2510 GOSUB 880 :REM INVLOED GROEI EN INSECT
2520 GOSUB 2050 :REM GROEI GEWAS
2530 GOSUB 1980 :REM GROEI INSEKTEN-POPULATIE
2540 GOSUB 1300 :REM VRAAGSTELLING ALGEMEEN
2550 IF I=1 THEN GOSUB 1440 :REM VRAAGSTELLING INSEKTEN
2560 IF I=2 THEN GOSUB 1720 :REM VRAAGSTELLING BESTRIJDINGS.
2570 X = G + B :REM BEPALING RESULT
2580 IF X > 4000 THEN GOSUB 2360 :REM > 4000 GEWONNEN !
2590 IF X < 50 THEN GOSUB 2200 :REM < 50 MISLUKT !
2600 IF K=0 THEN 2490
2610 REM *** SPEL AFGELOPEN ***
2620 END

```

Voorbeeld

*** BOERDERIJ *** DAG : 10 TEMPERATUUR : 28 C
 WEERSVERWACHTING : DROOG
-- GEWASSEN -- AANTAL GRUWELKNOLLEN : 441
 AANTAL BLAUWE KOOL : 530

-- INSEKTEN-POPULATIES -- AANTAL BLAUWE HOMMELS : 468
 AANTAL KAALSPRINKHANEN : 354
 AANTAL RODE GRUWELMIEREN : 297
 AANTAL SLUIPMUGGEN : 0

AANTAL SLUISJES : 0
+++++
0 = NIETS DOEN 3 = SLUISJE OPENEN
1 = INSEKTEN VRIJLATEN 4 = SLUISJE SLUITEN
2 = BESTRIJDINGSMIDDEL SPUITEN
WAT IS UW KEUZE 0

*** BOERDERIJ *** DAG : 11 TEMPERATUUR : 28 C
 WEERSVERWACHTING : REGEN
-- GEWASSEN -- AANTAL GRUWELKNOLLEN : 547
 AANTAL BLAUWE KOOL : 475

-- INSEKTEN-POPULATIES -- AANTAL BLAUWE HOMMELS : 519
 AANTAL KAALSPRINKHANEN : 407
 AANTAL RODE GRUWELMIEREN : 332
 AANTAL SLUIPMUGGEN : 0

AANTAL SLUISJES : 0
+++++
0 = NIETS DOEN 3 = SLUISJE OPENEN
1 = INSEKTEN VRIJLATEN 4 = SLUISJE SLUITEN
2 = BESTRIJDINGSMIDDEL SPUITEN
WAT IS UW KEUZE 0

4.3 De stadhouder van Flipflopland

Een managementspel zou de beste karakterisering zijn voor dit grote spel. In dit spel moet een stadhouder een land gedurende 14 achtereenvolgende jaren goed besturen. Daarbij is het de bedoeling dat het areaal grond wordt uitgebreid. Om dit te bereiken, kan men jaarlijks aan het eind van de oogst een aantal beslissingen nemen. Men regeert als feodaal vorst.

Het zou een paar hoofdstukken kosten om alle grappen die in het spel verwerkt zitten uit te leggen. Er zal slechts een tip van de sluier worden opgelicht. Wanneer men alles wil weten, zal men de bewerkingsmodule moeten doorlopen.

Als stadhouder kan men onder andere beslissen over de hoeveelheid van Flipflopers, de onderdanen. Men beslist verder over hoeveel hectaren grond ingezaaid wordt, hoeveel grond er gekocht of verkocht wordt, hoeveel marktplaatsen er worden gebouwd en hoeveel aan rattenbestrijding wordt uitgegeven.

Het spel is zodanig opgebouwd dat u eerst een aantal beslissingen moet nemen. De volgorde van de beslissingen bepaalt men zelf aan de hand van het menu. Denkt men geen beslissingen meer te hoeven nemen, dan geeft men in het menu een nul.

Men regeert voor een periode van 14 jaar. Na het verstrijken van de ambtstermijn wordt men beoordeeld op basis van de uitbreiding van de grond.

Om beslissingen te kunnen nemen, moet men op de hoogte zijn van een aantal feiten, anders wordt het een enorm gezoek. De prijzen van landbouwgrond en marktplaatsen zijn prijzen die op een markt van volkomen mededinging ontstaan. Ze zijn wel conjunctuurgevoelig, maar niet beïnvloedbaar door de speler. De prijzen zijn afhankelijk van de prijzen van het vorige jaar, het weer van het vorige jaar en een willekeurige component.

Elke Flipflopser dient aan het eind van een regeringsjaar zijn voedsel te krijgen. Men dient er rekening mee te houden, dat hij vijf tins graan per jaar verorbert.

Een marktplaats kan maximaal 8000 tins graan bevatten. Kunnen de marktplaatsen de opbrengst niet bergen, dan verrot de rest buiten. Een rattenplaag is dan niet denkbeeldig.

In de feodale tijd, waarin dit spel speelt, was er nog geen sprake van bemesting. Men moest dus uitkijken dat men de grond niet uitputte. Daarom maakte men in die tijd gebruik van het drieslagstelsel, d.w.z. een derde gedeelte van de grond liet men altijd braak liggen.

Het weer is volkomen willekeurig: onvoorspelbaar dus. De oogst is afhankelijk van het weer.

Voor elke ha grond die men inzaait, heeft men één tin graan nodig. Zoals men al heeft opgemerkt, wordt alles uitgedrukt in tins graan.

Het zal niet meevallen 14 jaar in het zadel te blijven terwijl misoogsten en epidemieën het stadhouderschap bedreigen.

Suggestie 1: Het programma is vrij groot en er wordt met vrij veel situaties rekening gehouden. In de module waar de Flipflopers hun voedsel krijgen, wordt een variabele gedefinieerd die de ondervoedingsgraad bijhoudt. Aan het begin van elk regeringsjaar wordt deze variabele in een vaste beginstand gebracht. Het gevolg is dat de ondervoeding alleen jaarlijks wordt bekeken en niet in de loop van de tijd. Men zou hier een wijziging in kunnen aanbrengen door de ondervoedingsgraad eveneens over de afgelopen jaren te bekijken.

Suggestie 2: In het programma wordt er zonder meer van uitgegaan dat men voldoende bewoners ter beschikking heeft om de grond in te zaaien. In werkelijkheid is dit natuurlijk

niet zo. Men zou dus kunnen stellen dat per ha grond die moet worden gezaaid een bepaald aantal bewoners nodig is.

Suggestie 3: De prijsontwikkelingen van landbouwgrond en marktplaatsen volgen samen min of meer hetzelfde patroon. Logisch zou het zijn om tussen deze twee prijsverlopen een tijdsvertraging in te bouwen, zodat de prijs van een marktplaats mede afhankelijk wordt van de prijs van de landbouwgrond.



Programma

```
10 REM *** FLIPFLOPLAND ***
20 RANDOM
30 DIM A(7)
40 GOTO 3050
50 REM *** SPELREGELS ***
60 CLS                                :REM SCHERM WISSEN
70 PRINT "*** STADHOUDER VAN FLIPFLOPLAND ***";
80 PRINT "    - VERKORTE INSTRUCTIE -"
90 PRINT
100 PRINT "U BENT VOOR EEN PERIODE VAN MAXIMAAL 14 JAAR "
110 PRINT "STADHOUDER VAN FLIP-FLOPLAND. DOOR GOED BELEID"
120 PRINT "MOET U PROBEREN UW VOLK OP UW HAND TE HOUDEN."
130 PRINT "DARBIJ DIENT U OOK UW BEZITTINGEN VERDER UIT-"
140 PRINT "TE BREIDEN.
150 PRINT "U DIENT ER REKENING MEE TE HOUDEN DAT ELKE"
160 PRINT "FLIP-FLOPPER ZO'N 5 TINS GRAAN VERORBERT."
170 PRINT "TEN OVERVLOEDE WIJS IK U EROP DAT DE PRIJZEN"
180 PRINT "VAN GROND EN MARKTPLAATSEN DOOR DE STADHOUDER"
190 PRINT "NIET BEINVLOEDBAAR ZIJN. DE PRIJZEN WORDEN"
200 PRINT "DOOR DE CONJUNCTUUR BEPAALD."
```

```

210 PRINT TAB(40)"UW DIENAAR"
220 PRINT
230 PRINT "HEEFT U DIT GELEZEN DRUK DAN <RETURN> IN";
240 INPUT D$
250 RETURN
260 :
270 REM *** INITIALIZEREN ***
280 A = 14 :REM TIJDSCYCLUS
290 A1 = 360 / A :REM EEN FASE CONJ
300 C = 0 :REM BEDRAG TOT. RATBESTR
310 H = 0
320 B1 = 0 :REM GROEIPERCENTAGE
330 D = 0 :REM DODEN
340 Q = 0 :REM OPBRENGST
350 B = INT(RND(0) * 500+500) :REM AANTAL FLIPFLOPPERS
360 X = INT(RND(0) * 5 + 7)
370 V = INT(B*X*(RND(0) + 1)) :REM BEGINVOORR. GRAAN
380 M = INT(V / 8000 + 1) :REM AANT. MARKTPLAATSEN
390 J = INT(RND(0) * 900 + 10) :REM STARTJAAR
400 T = INT(RND(0) * 7 + 1) :REM TELLER MAGERE/VETTE
410 R = INT(RND(0) * B) :REM AANTAL RATTEN
420 IF R < 5 THEN 410
430 G = INT(RND(0) * B/2 + B/2) * 4 :REM HECTAREN GROND
440 A(7) = G :REM BEGINSTAND ONTHOUDEN
450 W = INT(RND(0) * 100 + 1) :REM WEER JAAR N
460 W1 = INT(RND(0) * 100 + 1) :REM WEER JAAR N-1
470 P1 = INT(RND(0) * 50 + 50) :REM PRIJS GROND/HA
480 P2 = INT(RND(0)*600+300) :REM PRIJS MARKTPLAATS
490 GOSUB 550
500 D$ = ""
510 E$ = ""
520 RETURN
530 :
540 REM *** VARIABELEN MET EEN WAARDE VULLEN ***
550 G1 = 1 :REM 1 TIN GEZAAD
560 V1 = -1 :REM ONDERVOEDING
570 FOR I = 1 TO 6
580 A(I) = 0 :REM STUURVARIABELEN
590 NEXT I
600 RETURN
610 :
620 REM *** BEREKENING EFFECTEN ***
630 D$ = ""
640 E$ = ""
650 J = J + 1 :REM JAAR OPHOGEN
660 REM ** PRIJSONTWIKKELING **
670 REM * VETTE/MAGERE JAREN CYCLUS *
680 T = T + 1
690 IF T <= A THEN 710
700 T = 1 :REM BEGIN CYCLUS
710 X = A1 * T :REM WAARDE CYCLUS
720 F = SIN(X * (3.1457 / 180)) :REM FASE IN CYCLUS
730 REM * PRIJS-ONTWIKKELING LANDBOUWGROND *
740 X = 1 + 0.5*F+0.5*(W1-35)/100 :REM PRIJSWIJZIGING
750 P = P1 :REM OUDE P ONTHOUDEN

```

```

760 P1 = P1*0.45 + P1*X*0.35 + P1*(RND(0)+0.4)*0.20
770 P1 = INT((P + P1)/2) :REM NIEUWE PRIJS
780 IF P1 > 0 THEN 810
790 P1 = INT((P + (RND(0) * 50 + 1) ) / 2)
800 REM * PRIJSONTWIKKELING BOUW MARKT *
810 P = P2 :REM ONTHOUDEN OUDE P.
820 P2 = P2*0.45 + P2*X*0.35 + P2*(RND(0)+0.4)*0.2
830 X = RND(0) :REM RANDOM VARIABELE
840 P2 = INT(X * P + (1-X) * P2) :REM NIEUWE PRIJS
850 IF P2 > 0 THEN 880
860 P2 = INT(RND(0) * 500 + 300) :REM NIEUWE PRIJS
870 REM ** WEER IN JAAR J **
880 W1 = W :REM WEER VORIG JAAR
890 W = INT(RND(0) * 100 + 1) :REM WEER DIT JAAR
900 F$ = "DROOG"
910 IF W < 40 THEN 970
920 F$ = "REDELIJK"
930 IF W < 70 THEN 970
940 F$ = "GOED"
950 REM ** OPBRENGST GROND **
960 REM * DRIESLAGSTELSEL *
970 X = G1 - G * 2 / 3 :REM NIET TE VEEL GROND
980 IF X >= 0 THEN 1000
990 G1 = G1 - INT(RND(0) * G1 ) :REM GROND UITPUTTING
1000 Q = INT(RND(0) * G1/3 * 10) + G1 * 2/3 * 10 :REM OPBRENGST
1010 REM * INVLOED WEER *
1020 X = (W1 * 0.2 + 0.8 * W) / 100 :REM GROEIFACTOR
1030 IF X > 0.4 THEN 1050
1040 X = X - 0.4
1050 Q = INT( Q * (1 + (X - 0.5) ) ) :REM OPBRENGST
1060 REM ** RATTEN **
1070 REM * AANWAS RATTEN *
1080 X = (RND(0) + V/(M * 8000) + Q / 10 / G1) / 3
1090 REM * RATTEN EXPLOESIE *
1100 IF V = 0 THEN 1140
1110 IF C/V > 0.02 THEN 1140
1120 IF RND(0) > 0.25 THEN 1140
1130 X = X + RND(0)
1140 R1 = INT( R/2 * X ) :REM AANWAS RATTEN
1150 R = R + R1 / 2 :REM HELFT JONGE RATTEN
1160 REM * SCHADE VOORRAAD *
1170 X = (RND(0) * R) :REM % INVLOED
1180 V = V - INT(RND(0) * X * 3)
1190 REM * SCHADE GEWAS *
1200 X = R - X
1210 Q = Q - INT(RND(0) * X * 3)
1220 IF Q >= 0 THEN 1240
1230 Q = 0
1240 IF V >= 0 THEN 1260
1250 V = 0
1260 R = INT(R + R1 / 2) :REM POPULATIE RATTEN
1270 V = INT(V + Q) :REM VOORRAAD
1280 GOSUB 1570
1290 REM ** BEVOLKING-ONTWIKKELING **
1300 REM * OVERLEDEN OUDERDOM *

```

```

1310 X = INT(RND(0) * 0.1 * B)
1320 B1 = B1 - X
1330 B = B - X
1340 D = D + X
1350 REM * GEBOORTEN *
1360 X = INT((RND(0)*0.7+0.3)*B/5 * (1 + W1/100 - 0.4)*(1+V1))
1370 B = B + X :REM BEVOLKINGSAANWAS
1380 B1 = B1 + X
1390 REM * EPIDEMIE *
1400 IF V1 < -0.2 THEN 1430
1410 X = (W + W1)/2
1420 IF ( (R*RND(0))/B <= 1) OR ( X > 25 ) THEN 1490
1430 D$ = "ER IS EEN EPIDEMIE UITGEBROKEN !"
1440 X = INT(RND(0)*70 + 30)/100 :REM DOODSPERCENTAGE
1450 D = INT( X * B ) :REM AANTAL DODEN
1460 B1 = B1 - D
1470 B = B - D
1480 REM * BEPALING GROEIPERCENTAGE *
1490 B2 = INT(B1/(B-B1)*1000+0.5)/10 :REM GROEIPERC BEVOLK
1500 IF B2 >= -0.1 THEN 1530
1510 E$ = "DE BEVOLKING LOOPT TERUG !!"
1520 H = 2 :REM STUURVARIABELE
1530 GOSUB 550
1540 RETURN
1550 :
1560 REM *** VOLDOENDE RUIMTE MARKT ***
1570 IF V/M <= 8000 THEN 1620
1580 X = V - M * 8000
1590 V = M * 8000
1600 R = INT(R + RND(0) * X/5) :REM RATTEN EXTRA
1610 D$ = "EEN GEDEELTE VAN DE OPBRENGST VERROT BUITEN !"
1620 RETURN
1630 :
1640 REM *** GENOEG GRAAN ***
1650 H = 0 :REM STUURVARIABELE
1660 IF Y <= V THEN 1690
1670 PRINT "ZOVEEL GRAAN IS ER OP DE MARKTPLAATS NIET !"
1680 H = 1
1690 RETURN
1700 :
1710 REM *** GENOEG GROND ***
1720 H = 0 :REM STUURVARIABELE
1730 IF X <= G THEN 1760
1740 PRINT "ZOVEEL GROND BEZIT U NIET !"
1750 H = 1
1760 RETURN
1770 :
1780 REM *** RATTEN-CONTROLE = 5 ***
1790 PRINT "HOEVEEL TINS VOOR RATTENBESTRIJDING";
1800 INPUT Y
1810 IF Y < 0 THEN 1790
1820 GOSUB 1650
1830 IF H <> 0 THEN 1790
1840 V = V - Y :REM AANPASS GRAAN
1850 REM ** RATTENAFNAME I.V.M. BESTRIJDING **

```



```

1860 C = C + Y                               :REM TOT.BESTEEDT RAT
1870 R = R - INT((RND(0) * 0.3 + 0.7 * Y/V) * R)
1880 IF R/M > 60 THEN 1900
1890 R = INT(RND(0) * 60) * M                 :REM MIN. RATTEN
1900 A(5) = 1                                 :REM VRAAG GEBRUIKT
1910 RETURN
1920 :
1930 REM *** GROND KOPEN = 1 ***
1940 PRINT "HOEVEEL HA WILT U KOPEN ";
1950 INPUT X
1960 IF X < 0 THEN 1940
1970 IF X = 0 THEN 2080
1980 Y = X * P1
1990 GOSUB 1650
2000 IF H <> 0 THEN 1940
2010 REM ** TRANSACTIE **
2020 G = G + X                                 :REM GROND TOEVOEGEN
2030 V = V - Y                                 :REM GROND BETALEN
2040 REM ** AANTAL BEWONERS PER HA **
2050 Y = X * INT(RND(0) * 4 + 1)              :REM BEWONERS/HA
2060 B1 = B1 + Y                               :REM BEVOLK.TOENAME
2070 B = B + Y                                 :REM BEWONERS BIJ
2080 A(1) = 1                                 :REM VRAAG GEBRUIKT
2090 RETURN
2100 :
2110 REM *** GROND VERKOPEN = 2 ***
2120 PRINT "HOEVEEL HA GROND WILT U VERKOPEN ";
2130 INPUT X
2140 IF X < 0 THEN 2120
2150 IF X = 0 THEN 2220
2160 GOSUB 1720
2170 IF H <> 0 THEN 2120
2180 REM ** TRANSACTIE **
2190 G = G - X                                 :REM GRONDVERKOOP
2200 V = X * P1 + V
2210 GOSUB 1570
2220 A(2) = 1                                 :REM VRAAG GEBRUIKT
2230 RETURN
2240 :
2250 REM *** MARKTPLAATS = 3 ***
2260 PRINT "WILT U MARKTPLAATSEN BOUWEN 0-1-2-3 ";
2270 INPUT X
2280 IF (X < 0) OR (X > 3) THEN 2260
2290 REM ** GENOEG TINS JA/NEE **
2300 Y = X * P2 * 10
2310 GOSUB 1650
2320 IF H <> 0 THEN 2260
2330 REM ** TRANSACTIE **
2340 V = V - Y                                 :REM AANPASSEN VOORR
2350 M = M + X                                 :REM MARKT ERBIJ
2360 A(3) = 1                                 :REM VRAAG GEBRUIKT
2370 RETURN
2380 :
2390 REM *** VOEDSEL FLIPFLOPPERS = 4 ***
2400 PRINT "HOEVEEL TINNEN WILT U UW FLIPFLOPPERS GEVEN ";

```

```

2410 INPUT Y
2420 IF Y < 0 THEN 2400
2430 REM ** GENOEG GRAAN **
2440 GOSUB 1650
2450 IF H <> 0 THEN 2400
2460 REM ** ONDERVOEDING ? **
2470 V1 = (Y - B * 5) / (B * 5)      :REM % ONDERVOED
2480 IF V1 => 0 THEN 2520
2490 D$ = "DIT KAN ONDERVOEDING TOT GEVOLG HEBBEN."
2500 IF V1 > -0.3 THEN 2520
2510 V1 = V1 - RND(0)
2520 V = V - Y                        :REM VOORR AANP
2530 A(4) = 1                          :REM VRAAG BEANTWOORD
2540 RETURN
2550 :
2560 REM *** LAND INZAAIEN = 6 ***
2570 PRINT "HOEVEEL HA WILT U LATEN INZAAIEN ";
2580 INPUT X
2590 IF X < 0 THEN 2570
2600 GOSUB 1720
2610 IF H <> 0 THEN 2570
2620 REM ** VOLDOENDE ZAAIGOED **
2630 Y = INT(X)                        :REM 1 TIN PER HA
2640 GOSUB 1650
2650 IF H <> 0 THEN 2570
2660 G1 = X                             :REM IN TE ZAAIEN GROND
2670 V = V - Y
2680 A(6) = 1                          :REM VRAAG BEANTWOORD
2690 RETURN
2700 :
2710 REM *** BESLISSING ***
2720 PRINT "KOPEN = 1 VERKOPEN = 2 MARKT BOUWEN = 3"
2730 PRINT "VOEDSEL = 4 RATTENBESTRIJDING = 5 ZAAIEN = 6"
2740 PRINT "STOP = 0 UW KEUZE :";
2750 INPUT K
2760 IF (K <= 0) OR (K > 6) THEN 2790
2770 IF A(K) = 1 THEN 2790
2780 ON K GOSUB 1940 , 2120 , 2260 , 2400 , 1790 , 2570
2790 RETURN
2800 :
2810 REM *** UITVOER ***
2820 CLS                                :REM SCHERM WISSEN
2830 PRINT "*** STADHOUDER VAN FLIPFLOPLAND ***";
2840 PRINT TAB(40) "JAAR :";J
2850 PRINT
2860 PRINT "BEVOLKING                    :";B ;
2870 PRINT TAB(38) "GROEI-PERC. :";B2
2880 PRINT "AANTAL DODEN                :";D;
2890 PRINT TAB(38) "WEER                :";F$
2900 PRINT "RATTEN-POPULATIE           :";R
2910 PRINT
2920 PRINT "GRONDBEZIT IN HA             :";G;
2930 PRINT TAB(38) "PRIJS PER HA :";P1
2940 PRINT "VOORRAAD TINS GRAAN        :";V;
2950 PRINT TAB(38) "OPBRENGST         :";Q

```

```

2960 PRINT "AANTAL MARKTPLAATSEN      :";M;
2970 PRINT TAB(38) "BOUWPRIJS      :";P2*10
2980 PRINT
2990 PRINT D$
3000 PRINT E$
3010 RETURN
3020 :
3030 REM **** STUURMODULE ****
3040 REM *** INSTRUCTIES ***
3050 GOSUB 60
3060 REM *** INITIALIZEREN ***
3070 GOSUB 280
3080 FOR N = 1 TO 14
3090 REM *** UITVOER ***
3100 GOSUB 2820
3110 REM *** BESLISSING ***
3120 GOSUB 2720
3130 IF K <> 0 THEN 3100
3140 REM *** BEREKENING ***
3150 GOSUB 630
3160 IF H <> 2 THEN 3190
3170 N = 14                                :REM UIT LUS
3180 E$ = "HET VOLK AANVAARDT UW AUTORITEIT NIET LANGER !!"
3190 NEXT N
3200 REM *** UITVOER ***
3210 GOSUB 2820
3220 REM *** RESULTAAT ***
3230 IF H = 2 THEN 3310
3240 X = INT(G / A(7) * 1000 + 0.5) / 10
3250 D$="EEN KNAP HEERSER. HET GRONDBEZIT IS VERGROOT "
3260 IF X > 10 THEN 3300
3270 D$="EEN MATIG HEERSER. HET GRONDBEZIT IS VERGROOT "
3280 IF X > 0 THEN 3300
3290 D$="EEN SLECHT HEERSER. HET GRONDBEZIT IS VERKLEIND "
3300 PRINT D$;"MET "; ABS(X);" %"
3310 END

```

Voorbeeld

```

*** STADHOUDER VAN FLIPFLOPLAND ***      JAAR : 579

BEVOLKING                : 816           GROEI-PERC. : 0
AANTAL DODEN             : 0            WEER       :
RATTEN-POPULATIE        : 292

GRONDBEZIT IN HA        : 2496          PRIJS PER HA : 51
VOORRAAD TINS GRAAN     : 9055          OPBRENGST  : 0
                                   BOUWPRIJS   : 3180

```

KOPEN = 1 VERKOPEN = 2 MARKT BOUWEN = 3 VOEDSEL GEVEN = 4
RATTENBESTRIJDING = 5 ZAAIEN = 6 STOP = 0
UW KEUZE : 4
HOEVEEL TINNEN WILT U UW FLIPFLOPPERS GEVEN 4080

*** STADHOUDER VAN FLIPFLOPLAND *** JAAR : 579

BEVOLKING	: 816	GROEI-PERC. :	0
AANTAL DODEN	: 0	WEER	:
RATTEN-POPULATIE	: 292		
GRONDBEZIT IN HA	: 2496	PRIJS PER HA :	51
VOORRAAD TINS GRAAN	: 4975	OPBRENGST	: 0
		BOUWPRIJS	: 3180

KOPEN = 1 VERKOPEN = 2 MARKT BOUWEN = 3 VOEDSEL GEVEN = 4
RATTENBESTRIJDING = 5 ZAAIEN = 6 STOP = 0
UW KEUZE : 6
HOEVEEL HA WILT U LATEN INZAAIEN 1600

*** STADHOUDER VAN FLIPFLOPLAND *** JAAR : 579

BEVOLKING	: 816	GROEI-PERC. :	0
AANTAL DODEN	: 0	WEER	:
RATTEN-POPULATIE	: 292		
GRONDBEZIT IN HA	: 2496	PRIJS PER HA :	51
VOORRAAD TINS GRAAN	: 3375	OPBRENGST	: 0
		BOUWPRIJS	: 3180

KOPEN = 1 VERKOPEN = 2 MARKT BOUWEN = 3 VOEDSEL GEVEN = 4
RATTENBESTRIJDING = 5 ZAAIEN = 6 STOP = 0
UW KEUZE : 0

*** STADHOUDER VAN FLIPFLOPLAND *** JAAR : 580

BEVOLKING	: 949	GROEI-PERC. :	16.3
AANTAL DODEN	: 25	WEER	: GOED
RATTEN-POPULATIE	: 1177		
GRONDBEZIT IN HA	: 2496	PRIJS PER HA :	55
VOORRAAD TINS GRAAN	: 16000	OPBRENGST	: 18448
		BOUWPRIJS	: 3430

EEN GEDEELTE VAN DE OPBRENGST VERROT BUITEN !
KOPEN = 1 VERKOPEN = 2 MARKT BOUWEN = 3 VOEDSEL GEVEN = 4
RATTENBESTRIJDING = 5 ZAAIEN = 6 STOP = 0
UW KEUZE :

Appendix

In deze appendix zijn de volgende conversietabellen opgenomen:

Tabel 1: String-functies.

Tabel 2: Rekenkundige en logische functies.

Tabel 3: Statements.

De tabellen kunnen worden gebruikt bij het aanpassen van programma's die op een ander type computer zijn geschreven dan de in uw bezit zijnde computer.

Het gebruik van de tabellen gaat als volgt:

In het aan te passen programma ziet u een statement waarvan u denkt dat die niet op uw computer voorkomt. U zoekt de statement op in de tabellen, controleert op welke computers die statement voorkomt, en indien hij inderdaad op uw computer niet voorkomt dan kunt u in de kolom 'overeenkomstige statements' vinden of er vervangende statements zijn. Zo ja, dan kunt u de voor uw computer geldende vervangende statement in het aan te passen programma gebruiken.

Statements en functies die alleen op één computer voorkomen, en waarvoor geen vervangende statement of functie bestaat, zijn in deze tabellen meestal niet opgenomen.

De auteur van dit boek heeft getracht de programma's zoveel mogelijk in converteerbare statements te schrijven.

Tabel 1. String-functies.

string-functies	BBC-computer	Commodore 64	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige string-functies
\$ n											X	CHR\$(n)
ASC(str)	X	X	X	X	X	X	X	X		X		CH
CH											X	ASC(str)
CHR\$(n)	X	X	X	X	X	X	X	X	X	X		\$(n)
CODE									X			
FRE('')		X				X						FRE(n\$)
FRE(n\$)					X		X					FRE('')
GET	X	X	X	X		X				X		INKEY\$
INKEY\$	X						X	X				GET
LEFT\$(str,n)	X	X	X	X	X	X	X	X	X	X	X	
LEN(str)	X	X	X	X	X	X	X	X	X	X	X	
MID\$(str,m,n)	X	X	X	X	X	X	X	X	X	X	X	
RIGHT\$(str,n)	X	X	X	X	X	X	X	X	X	X	X	
slicing									X			geeft LEFT\$, MID\$ en RIGHT\$ functies
STR\$(n)	X	X	X	X	X	X	X	X	X	X	X	
STRING\$(n,kar)	X						X	X				
VAL(str)	X	X	X	X	X	X	X	X	X	X	X	
VARPTR n				X			X	X				

Tabel 2. Rekenkundige en logische functies.

functies	BBC-computer	Commodore 64	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige functies
& n	X										X	HEX\$(n)
%											X	CINT, FIX
?FRE(0)			X									FRE
ABS	X	X	X	X	X	X	X	X	X	X	X	
ACS	X								X			
AND	X		X	X	X	X	X	X	X	X	X	
ASN	X								X			
ATN	X	X	X	X	X	X	X	X	X	X	X	
CALLM..., (x)				X								LINK n, USR(x)
CDBL							X	X				
CINT							X	X				%
COS n	X	X	X	X	X	X	X	X	X	X	X	
CSGN							X	X				
CUR				X								POS
DEG	X										X	
ERL	X						X	X				
ERR	X						X	X				
EXP n	X	X	X	X	X	X	X	X	X	X	X	
FIX							X	X				%
FLT											X	
FRAC				X								
FRE		X	X	X	X	X	X	X		X		?FRE(0)
HEX\$(n)				X			X					& n
HTN											X	
INT	X	X	X	X	X	X	X	X	X	X	X	
LINK n											X	CALLM..., (x), USR(x)
LN	X								X			LOG
LOG	X	X	X	X	X	X	X	X		X	X	LN
LOGT				X								
MOD	X		X	X			X					x%y
NOT			X	X	X	X	X	X	X			
OCT\$(n)							X					
OR	X	X	X	X	X	X	X	X	X		X	
POS	X	X	X		X	X	X	X				CUR
RAD	X										X	
RND	X			X							X	RND(0), RND(neg)
RND%n											X	
RND(0)					X	X	X	X	X	X		RND(neg), RND
RND(n)		X					X	X				RND(pos)
RND(neg)			X									RND(0), RND
RND(pos)			X									RND(n)
SCRN(x,y)			X	X								
SGN	X	X	X	X	X	X	X	X	X	X	X	
SIN n	X	X	X	X	X	X	X	X	X	X	X	
SPC m	X	X	X	X	X	X	X					
SPC n						X				X		
SQR(n)	X	X	X	X	X	X	X	X	X	X	X	
SWAP							X					
TAN(n)	X	X	X	X	X	X	X	X	X	X	X	
USR(x)	X	X	X		X	X	X	X	X	X		CALLM..., (x), LINK n
x%y											X	MOD

Tabel 3. Statements.

statements	BBC-computer	Commodore 64	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige statements
\$											X	DEFSTR, IMPSTR
%											X	DEFDBL
?m=n	X										X	POKE m,n
?n	X										X	IN(poort), IN#n, INP, INP(n)
AT										X		CURSOR m,n
BEEP x,y									X		X	SOUND
CALL	X		X									CALLM, SYS, LINK
CALLM				X								CALL, SYS, LINK
CLEAR n	X										X	MODE n
CLEAR Ø	X										X	MODE Ø, TEXT, GR
CLOAD*						X						INPUT#n,m, RECALL, LOADA, LOAD, GET A
CLOSE	X	X			X					X		SHUT
CLS	X						X	X				HOME, PRINT \$12
COLOR n											X	HCOLOR=n
COLOR=n			X									COLOUR n
COLOUR n	X										X	COLOR=n
COUNT	X										X	
COUNT n	X										X	TAB
CSAVE*						X					X	SAVE A, STORE
CURSOR m,n				X						X		AT
DATA	X	X	X	X	X	X	X	X		X	X	
DEF FN (naam)	X	X	X		X	X				X		
DEFDBL						X	X					%
DEFINT						X	X					IMPINT
DEFSNG						X	X					IMPFPPT
DEFSTR							X	X				IMPSTR, \$
DIM var(n)	X	X	X	X	X	X	X	X	X	X	X	
DO..UNTIL											X	
DOT				X								SET (x,y), PLOT, HPLOT, PLOT k,x,y
DRAW	X			X								
DSP n			X									
END	X	X	X	X	X	X	X	X		X	X	
ERROR	X					X	X					
EXT	X										X	
FILL				X								
FIN											X	OPEN
FOR..TO..STEP	X	X	X	X	X	X	X	X	X	X	X	
GET	X										X	
GET A	X	X										INPUT#n,m, RECALL, LOADA, CLOAD*, LOAD
GOSUB n	X	X	X	X	X	X	X	X	X	X	X	
GOTO n	X	X	X	X	X	X	X	X	X	X	X	
GR			X									CLEAR Ø
HCOLOR=n			X									COLOR n
HOME			X									CLS, PRINT \$12
HPLOT			X									DOT, PLOT, PLOT k,x,y
HTAB (n)			X									
IF..GOSUB n	X		X	X	X	X	X				X	
IF..GOTO n	X		X	X	X	X	X			X	X	
IF..THEN..	X	X	X	X	X	X	X	X	X	X	X	
IF..THEN..ELSE	X					X	X				X	
IMPFPPT				X								DEFSNG
IMPINT				X								DEFINT
IMPSTR				X								DEFSTR, \$
IN (poort)					X		X					IN#n, INP, INP(n), ?n
IN#n			X									IN (poort), INP, INP(n), ?n
INP				X								IN (poort), IN#n, INP (n), ?n
INP (n)							X					IN (poort), IN#n, INP, ?n
INPUT n	X	X						X				INPUT "...",n
INPUT "...",n	X		X	X	X	X	X			X	X	INPUT n
INPUT #n,m	X	X			X	X	X			X	X	RECALL, LOADA, CLOAD*,LOAD, GET A
LET	X	X	X	X	X	X	X	X	X	X	X	

Tabel 3. Statements (vervolg).

statements	BBC-computer	Commodore 64	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige statements
LINK											X	CALL, CALLM, SYS
LOAD	X								X			INPUT #n,m, RECALL, LOADA, CLOAD*, GET A
LOADA				X								INPUT #n,m, RECALL, CLOAD*, LOAD, GET A
LPRINT						X	X	X				PRINT \$2, PRINT =
MODE n	X			X								CLEAR n
MODE 0				X								TEXT, CLEAR 0
MOVE	X										X	
NEXT	X	X	X	X	X	X	X	X	X	X	X	
ON..ERROR GOTO n	X					X	X				X	ONERR
ON..GOSUB n	X	X	X	X	X	X	X	X	X	X	X	
ON..GOTO n	X	X	X	X	X	X	X	X	X	X	X	
ONERR			X									ON..ERROR GOTO n
OPEN	X				X					X		FIN
OUT (poort, n)				X		X	X					PR #n, PRINT #n
PAUSE								X				WAIT
PDL			X	X								
PEEK (n)	X	X	X	X	X	X	X	X	X	X		PRINT ?n
PLOT			X					X				SET (x,y), HPLOT, DOT, PLOT k,x,y
PLOT k,x,y	X										X	RESET (x,y), UNPLOT, SET (x,y), HPLOT, DOT
POINT	X						X					
POKE m,n	X	X	X	X	X	X	X	X	X	X		?m,n
PR #n			X			X	X					OUT (poort, n), PRINT #n
PRINT	X	X	X	X	X	X	X	X	X	X	X	
PRINT \$12											X	CLS, HOME
PRINT \$2											X	PRINT =, LPRINT
PRINT =					X					X		SAVEA, STORE, LPRINT
PRINT ?n	X										X	PEEK (n)
PRINT @n							X					
PRINT USING						X	X					
PRINT #n	X	X			X					X		OUT (poort, n), PR #n
PTR (n)	X										X	
PUT											X	
READ	X	X	X	X	X	X	X	X	X	X	X	
RECALL			X									
REM	X	X	X	X	X	X	X	X	X	X	X	INPUT #n,m, LOADA, CLOAD*, LOAD, GET A
RESET (x,y)							X					UNPLOT, PLOT k,x,y
RESTORE	X	X	X	X	X	X	X	X	X	X	X	RSTORE n
RESUME			X			X	X					
RETURN	X	X	X	X	X	X	X	X	X	X	X	
RSTORE n						X						RESTORE
SAVEA				X								STORE, PRINT =
SET (x,y)							X					PLOT, HPLOT, DOT, PLOT k,x,y
SGET											X	
SHUT											X	CLOSE
SOUND	X			X								BEEP x,y
SPEED			X									
STOP	X	X	X	X	X	X	X	X	X	X	X	
STORE			X									SAVEA, PRINT =
SYS		X									X	CALL, CALLM, LINK
TAB	X		X	X	X	X	X	X	X	X		COUNT n
TEXT			X									MODE 0, CLEAR 0
TOP											X	
UNPLOT								X				RESET (x,y), PLOT k,x,y
VLIN..AT			X									
VTAB (n)			X									
WAIT	X	X	X	X	X	X				X	X	PAUSE

Ondanks alle serieuze en nuttige toepassingen is er voor de computer toch een belangrijke rol weggelegd als ontspannend element. Met andere woorden: de computer kan worden gebruikt voor spelletjes.

Dit boek bevat een verzameling origineel Nederlandstalige spelprogramma's variërend van eenvoudige korte programma's tot behoorlijk uitgebreide programma's met een hogere moeilijkheidsgraad.

Ondanks het luchtige karakter hebben deze programma's de serieuze programmeur meer te bieden. Er is door de auteur naar gestreefd om een zo duidelijk mogelijke structuur in de programma's aan te brengen. Men is daardoor in staat snel de werking te doorzien zodat de suggesties die worden gegeven voor uitbreiding of aanpassing, na enig oefenen best zijn uit te voeren.

In een appendix wordt aangegeven wat de verschillen zijn in de BASIC's van enkele populaire machines.

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

JUSTITIA MATHAMATICA

AMSTRAD

CPC



MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA



<https://acpc.me/>

[FRA] Ce document a été préservé numériquement à des fins éducatives et d'études, et non commerciales.

[ENG] This document has been digitally preserved for educational and study purposes, not for commercial purposes.

[ESP] Este documento se ha conservado digitalmente con fines educativos y de estudio, no con fines comerciales.